
Résumé

SIMOD est un progiciel de simulation de systèmes stochastiques à événements discrets, avec
vision par événements ou par processus. Il supporte aussi la simulation continue, pour laquelle
l’évolution de certaines variables du modèle se fait selon des équations différentielles. Le progiciel
est implanté sous forme de modules pré-compilés, écrits en Modula-2 (un langage qui ressemble à
Pascal). Ces modules comprennent des types de données prédéfinis, des fonctions, des procédures, et
supportent une partie de l’exécution des programmes utilisateurs. Ces outils permettent d’exprimer
un modèle de façon relativement concise, dans un langage lisible. Il ne s’agit pas d’un nouveau
langage; en réalité, un programme SIMOD est un programme Modula-2 qui utilise les modules de
SIMOD.

Le progiciel gère l’horloge et la liste des événements d’une simulation. Il fournit des outils
pour gérer des listes, générer des valeurs pseudo-aléatoires selon différentes lois de probabilité
et recueillir des statistiques durant la simulation. Les objets “actifs” dans le système peuvent
être représentés par des processus, et SIMOD s’occupe de la synchronisation entre les processus,
implantés sous forme de co-routines. Différentes formes de synchronisation entre les processus
sont disponibles, telles que l’exclusion mutuelle par des ressources à capacité limitée, la relation
de type producteur/consommateur par le biais d’un tampon (une pile de jetons), l’attente sur des
conditions, et les relations de type mâıtre/esclave.

La version courante de SIMOD est implantée sous SUN/Solaris et sous Linux.

1Denis Alain, Jean Bélanger, Michel Duclos, Nataly Giroux, Bruno Pelletier, Gaétan Perron, Francis
Picard et Jean-Sébastien Sénécal ont participé au développement de SIMOD.

3.6 Simulation discrète et continue: chauffage de lingots . . . . . . . . . . . . . . 24

3.7 Entretien d’un groupe de machines par un robot . . . . . . . . . . . . . . . . 28

4 Conclusion 42

ANNEXES 43

Avant-propos

La modélisation est une activité fondamentale, qui se situe au cœur de l’évolution de la
science et de l’humanité. C’est en construisant des modèles que nous pouvons mieux com-
prendre notre environnement, anticiper l’avenir, et développer des stratégies pour réagir à
cet environnement. Grâce à l’avènement d’ordinateurs de plus en plus puissants et d’outils
logiciels de haut niveau, nous pouvons développer et simuler sur ordinateur des modèles très
complexes à des coûts de plus en plus faibles.

Ce guide décrit SIMOD, un progiciel servant à faciliter l’écriture des programmes de
simulation de systèmes à événements discrets. SIMOD est basé sur Modula-2, un langage
élégant et puissant malgré sa relative simplicité. Ainsi, l’utilisateur a accès à une “bôıte à
outils” bien garnie pour la simulation, tout en conservant les avantages d’utiliser un langage
populaire et qui encourage l’application des principes modernes du génie logiciel. Ce progiciel
a été implanté avec un certain souci d’efficacité, de façon à ce qu’il soit vraiment utilisable
en pratique, même pour des applications de grande taille.

Je tiens à remercier tous ceux dont les commentaires et critiques ont permis d’améliorer
SIMOD et le présent guide, en particulier les étudiants des cours de simulation et les an-
ciens collègues Jules Desharnais, Marcel Dupras, Clermont Dupuis, Olivier Roux et Thien
Vo-Dai au département d’informatique de l’Université Laval de 1986 à 1990. Je remercie
également le CRSNG, le programme Défi 85 d’Emploi et Immigration Canada, de même
que le programme d’innovation pédagogique 1986-87 de la faculté des sciences et de génie de
l’Université Laval, dont les subventions ont permis de payer en partie les étudiants qui ont
travaillé à la conception et à l’implantation de SIMOD. La publication du présent document
a été supportée en partie par le fonds FCAR.

L’utilisation de SIMOD exige un compilateur Modula-2. Ce compilateur ne vient pas
avec le progiciel SIMOD et c’est la responsabilité de l’utilisateur de se le procurer.

1 Introduction

La simulation par ordinateur est un outil très utile aux preneurs de décisions. Il s’agit de
modéliser un système et de reproduire son comportement afin de mieux comprendre, analyser,
et éventuellement améliorer son fonctionnement. Les modèles suffisamment réalistes sont le
plus souvent dynamiques et stochastiques , c’est-à-dire que leur état évolue dans le temps et
que cette évolution peut se faire de façon aléatoire. SIMOD permet de simuler des modèles
de ce genre. Il a été conçu surtout pour la simulation à événements discrets [13, 19, 36, 37],
mais permet aussi la simulation continue.

Dans le passé, les programmes de simulation ont longtemps été (et sont encore souvent)
écrits dans des langages tout-usage, tels que C, Pascal ou FORTRAN [37, 13]. Dans le but
d’améliorer la productivité, on a développé, dans ces langages, des librairies de sous-routines
fournissant des outils de base pour la programmation de simulation. On a aussi créé, par
ailleurs, des langages spécialisés pour la simulation. Par exemple, les langages GPSS [52],
SIMSCRIPT [50], SLAM [48] et SIMAN [47] ont été créés pour la simulation des systèmes
à événements discrets. Plusieurs langages spécialisés ont aussi été créés pour la simulation
continue.

Les langages spécialisés fournissent des outils de plus haut niveau et permettent d’écrire
des programmes plus courts, mais offrent habituellement moins de flexibilité que les langages
tout-usage. Dans les langages de simulation les plus populaires, on doit souvent appeler des
routines FORTRAN, C, ou VisualBasic pour effectuer certaines opérations ou programmer
des aspects plus complexes du modèle. Les compilateurs des langages de simulation sont
souvent plus coûteux et moins répandus que ceux des langages tout-usage, et produisent
habituellement du code moins optimisé. Autre obstacle important aux langages spécialisés:
pour les utiliser, il faut les apprendre. Comme ces langages ont leur propre syntaxe, parfois un
peu excentrique, cela nécessite un investissement de temps non négligeable, surtout pour un
usage unique ou occasionnel. Plusieurs logiciels spécialisés pour la simulation proposent aussi
une approche dite “sans programmation”, où l’usager spécifie son modèle via des menus ou
en manipulant des objets graphiques à l’écran. Cette approche peut être intéressante pour
développer rapidement des petits modèles simples ou qui cadrent bien avec les structures
pré-programmées dans le logiciel, mais devient difficile à utiliser lorsque l’on veut modéliser
quelque chose qui sort du cadre fixé par le logiciel.

SIMOD est un progiciel de simulation implanté sous forme d’un ensemble structuré de
modules précompilés, dans le langage Modula-2. L’utilisateur programme en Modula-2, ce
qui lui donne toute la flexibilité d’un langage tout-usage. Il est aussi possible d’utiliser
SIMOD à partir d’un programme C. Les modules précompilés fournissent des procédures et
des types de données pré-définis, et contiennent un exécutif qui s’occupe de gérer l’horloge
de la simulation, de gérer la liste d’événements, de synchroniser les processus, de recueillir
des statistiques, etc. SIMOD supporte l’utilisation de la vision par processus: chaque objet
actif dans le système peut être représenté par un processus, qui possède ses propres données
(locales) et dont le comportement est décrit par une procédure (les processus de même type
ont la même procédure et les mêmes types de données). Durant la simulation, ces processus

interagissent. Il peut y avoir plusieurs types de processus dans un modèle, et plusieurs
instances d’un même type de processus à un instant donné. Dans SIMOD, les processus sont
implantés par des co-routines. La programmation avec vision par processus est reconnue
comme étant une façon très naturelle de décrire un système complexe [11, 9, 36, 6]. Les
systèmes réels sont souvent constitués d’objets actifs plus ou moins autonomes (tels que des
personnes, des avions, des véhicules, des robots, etc.) qui interagissent pour accomplir leurs
tâches, et on peut faire correspondre chaque type d’objet à un type de processus. Dans
SIMOD, on peut aussi prévoir des événements et donc mélanger, si on le désire, la vision
par processus avec la vision par événements. Dans plusieurs cas, on peut même n’utiliser
que des événements. En fait, il existe plusieurs types de systèmes qui se modélisent plus
facilement et plus simplement de cette façon. De plus, l’utilisation des co-routines ralentit
l’exécution de façon assez significative en pratique, et pour cette raison, on peut vouloir éviter
leur utilisation dans certain cas. SIMOD permet aussi la simulation continue ou “mixte”
(mélange de simulation continue et à événements discrets dans un même modèle). Dans ce
cas, les valeurs de certaines variables évoluent selon des équations différentielles.

Un effort a été fait afin de bien séparer la définition fonctionnelle du progiciel (la par-
tie visible à l’usager) de l’implantation proprement dite. L’usager peut utiliser les types de
données et les routines fournies par le progiciel, mais n’a pas à accéder directement aux struc-
tures de données utilisées par le progiciel. Il n’a pas à connâıtre les détails d’implantation
pour pouvoir utiliser le progiciel avec profit. Lorsqu’on a affaire à de gros modèles de simu-
lation, SIMOD encourage l’idée de décomposer le modèle et d’en implanter les parties dans
des modules compilés séparément. A la limite, on peut implanter indépendamment, dans
différents modules, des programmes de simulation pour différents sous-systèmes, puis faire
un petit module “chapeau” qui démarre une simulation intégrant les différents sous-systèmes
(on peut même faire plusieurs “copies” d’un même sous-système).

Pourquoi avoir choisi Modula-2 comme langage de base? Au moment de la conception
de SIMOD, en 1986, Pascal était le langage de programmation dominant dans les milieux
académiques. Conscient des limites de Pascal pour le grands projets de programmation, son
concepteur N. Wirth venait de proposer un langage amélioré pour le remplacer: Modula-2,
que plusieurs voyaient déjà comme le langage de l’avenir. On sait maintenant que ce ne fut
pas la cas, pour toutes sortes de raisons, en particulier l’absence de standard pour les librairies
pendant les 15 premières années et le fait que Modula-2 n’est pas un véritable langage à
objets. Par contre. Modula-2 est un langage simple, flexible, modulaire, lisible, qui donne
accès à des instructions de bas niveau, et qui peut produire du code aussi efficace à l’exécution
que C (et beaucoup plus efficace que pour des programmes écrits dans des langages à objets
comme C++ ou Java). On sait que la vitesse d’éxécution des programmes de simulation
stochastique est souvent une contrainte très importante en pratique. La disponibilité de
co-routines (ce qui n’est pas le cas en C) facilite l’implantation de la vision par processus.
Ce sont ces avantages qui ont motivé notre choix de Modula-2 en 1986.

Certaines des idées de SIMOD sont inspirées du progiciel DEMOS [10], basé sur le langage
Simula [9], de SIMSCRIPT II.5 [50], et de SIMPascal [38]. D’autres progiciels de simulation à
événements discrets, basés sur des langages qui supportent la quasi-concurrence, avaient déjà
été proposés dans les années 70 et 80. Sans prétendre être exhaustif, on peut mentionner des

progiciels basés sur ADA [6], C [52], Simula [11, 10], Modula-2 [43, 53], SMALLTALK [34],
SCHEME (un dialecte de Lisp) [58], PROLOG [59], etc. SIMOD a été implanté dans le passé
dans l’environnement Modula-2 de Logitech [24], sous MS-DOS dans l’environnement Top-
Speed Modula-2 [27], sous SUN/OS avec le compilateur Modula-2 de SUN, sous SUN/Solaris
avec le compilateur MCS Modula-2, puis (la version actuelle) sous SUN/Solaris et sous Linux
avec le compilateur-traducteur XDS [61].

La prochaine section donne une vue d’ensemble de SIMOD. Dans la section 3, on présente
une série d’exemples, d’abord simples puis plus complexes, dont l’objectif est d’illustrer
l’utilisation de SIMOD. L’examen de ces exemples est probablement la meilleure façon de se
familiariser avec le progiciel. La définition fonctionnelle de chacun des modules est donnée à
l’annexe A. On suggère au lecteur de s’y référer en étudiant les exemples. Dans la conclusion,
on discute de certains aspects positifs et négatifs de notre expérience avec Modula-2 et avec
SIMOD, et de quelques détails d’implantation. Tout au long de ce guide, nous supposons
que le lecteur est déjà familier avec le langage Modula-2. Pour une introduction à Modula-2,
voir [25, 26, 60].

2 Une vue d’ensemble du progiciel

Nous décrivons maintenant l’organisation de SIMOD et les principaux modules qui le con-
stituent. La description détaillée de tous les outils disponibles se trouve en Annexe A. Le
diagramme suivant illustre la structure hiérarchique qui existe entre les différents modules.
Lorsqu’on peut se rendre du module A au module B en suivant une ou plusieurs flèches, cela
indique que le module B utilise (directement ou indirectement) des objets du module A dans
sa définition ou dans son implantation. Par exemple, RES utilise PROCS, EVENT, SIM, LIST,
STAT, RAND1 et RAND. Ainsi, lorsqu’on utilise RES, on doit faire l’édition de liens avec tous
ces modules. Le plupart des modules utilisent aussi les modules NUM et MYINOUT, qui ne font
pas partie de SIMOD, mais qui fournissent certains outils d’usage général et qui servent à
améliorer la portabilité.

Les modules accessibles à l’utilisateur sont brièvement décrits ci-bas. Évidemment, outre
les modules mentionnés ici, l’utilisateur peut ajouter ses propres modules.

• RAND contient le générateur de valeurs pseudo-aléatoires. uniformes. C’est le générateur
récursif combiné (CMRG) MRG32k3a proposé par L’Ecuyer dans [39].

• RAND1 fournit des fonctions pour générer des valeurs aléatoires suivant différentes lois
de probabilité.

• RAND2 est une interface au module C--RAND implanté par Kremer et Stadlober [35, 55],
et permet aussi de générer des valeurs aléatoires suivant différentes lois de probabilité.

• RANDLIB est une interface au module Randlib, développé par Brown, Lovato, Russel
et Venier [14], qui sert aussi à générer des valeurs aléatoires.

• SIM fournit une fonction qui retourne l’instant courant de la simulation, et des
procédures pour initialiser, démarrer et stopper la simulation.

• EVENT offre des outils pour la prévision d’événements dans un délai donné, ou
l’annulation d’événements déjà prévus.

• CONT fournit les outils de base pour la simulation continue, pour laquelle certaines
variables évoluent selon des équations différentielles.

• STAT fournit des outils pour le recueil de statistiques.

• PROCS permet la programmation avec vision par processus. Chaque type de processus
est associé à une procédure définie par l’utilisateur, qui décrit le comportement de
l’objet représenté par ce processus.

• GENEP permet de démarrer et stopper des mécanismes automatiques pour générer une
suite d’événements ou de processus d’un certain type. Cette génération peut se faire
selon un processus de renouvellement quelconque, et en particulier selon un processus
de Poisson.

• LIST fournit un type qui correspond à une liste doublement châınée, et un ensemble
d’outils pour la gestion des listes. Les listes sont gérées sans connâıtre les types des
objets qu’elles contiennent. En fait, ces objets sont vues par le logiciel comme s’ils
étaient du type ADDRESS, qui est compatible avec tout type de pointeur. Ces listes
peuvent donc contenir à peu près n’importe quoi. En pratique, on s’en sert surtout
pour implanter les files d’attente. Le module offre aussi des outils pour la collecte
automatique de statistiques sur les listes.

• RES permet la synchronisation des processus par le biais de ressources à capacité limitée.
Une ressource correspond à un poste de service, avec une seule file d’attente, et dont
la politique de service est fixée lors de la création. La capacité de la ressource peut
correspondre, par exemple, au nombre de serveurs (identiques). Les processus doivent

demander un nombre d’unités de la ressource avant de pouvoir l’utiliser, et libérer ce
nombre d’unités lorsqu’ils en ont terminé. Un même processus peut détenir des unités
de différentes ressources, ou des unités d’une même ressource demandées à des instants
différents. Les requêtes peuvent aussi se faire avec des priorités. Enfin, le module peut
recueillir (automatiquement) des statistiques sur les ressources.

• BIN permet les relations de type producteur/consommateur entre les processus. Un
“bin” correspond essentiellement à une pile de jetons (un tampon) et une file d’attente
de processus. Un producteur peut ajouter des jetons à la pile, et un (processus) con-
sommateur peut demander à prendre des jetons de la pile. Dans ce dernier cas, s’il
n’y a pas assez de jetons pour satisfaire la demande du consommateur, ce dernier est
bloqué et placé dans la file d’attente pour cette pile. Il n’est réactivé que lorsque c’est
à son tour de se faire servir et qu’un nombre suffisant de jetons est disponible.

• COND offre des outils permettant de gérer l’attente de processus sur des conditions
booléennes. Lorsqu’un processus demande à attendre une condition et que cette
dernière est fausse, celui-ci est bloqué jusqu’à ce que la condition devienne vraie.

• MASLA implante le paradigme de synchronisation de type mâıtre/esclave entre les
processus. Un point de synchronisation (ou point de rencontre) possède deux files
d’attente: une pour les processus mâıtres et une pour les processus esclaves . Un pro-
cessus peut demander à devenir un mâıtre (et obtenir un esclave) pour un point de
rencontre donné. Si aucun esclave n’est disponible, ce processus est bloqué et placé
dans une file d’attente jusqu’à ce qu’il puisse obtenir l’esclave demandé. Un processus
peut aussi devenir un esclave pour un point de rencontre donné. Dans ce cas, si la file
d’attente des mâıtres est vide, ce processus est bloqué et placé dans la file d’attente
des esclaves, sinon il réveille le premier mâıtre de la file et devient son esclave.

3 Exemples de programmes SIMOD

Nous présentons dans cette section quelques exemples de programmes SIMOD. L’objectif est
de familiariser le lecteur avec le progiciel et d’illustrer son utilisation. Le premier exemple est
classique et très simple, il s’agit d’une file d’attente à plusieurs serveurs. Il illustre l’utilisation
simultanée d’événements et de processus. Les deux exemples qui suivent sont inspirés de
[37]. Il s’agit de programmes de simulation d’un atelier de production et d’un système à
temps partagé. Le quatrième exemple est une simulation de visites guidées, et illustre une
synchronisation des processus légèrement plus complexe. Le cinquième illustre l’utilisation
de SIMOD pour la simulation continue, et le sixième illustre une simulation mixte et plus
complexe. Le septième et dernier exemple montre comment SIMOD permet le développement
modulaire des programmes de simulation. Il s’agit d’un robot qui, dans une cellule de
production d’une usine, entretient une série de machines. La description du système, de la
politique de commande, et de l’expérience statistique que l’on veut faire avec le simulateur,
sont implantés dans trois modules séparés. Ainsi, on peut modifier par exemple la politique
de commande sans changer quoi que ce soit aux deux modules qui décrivent le système
et l’expérience statistique (et sans recompiler ces modules). Eventuellement, on pourrait
remplacer le module qui décrit le système par un mécanisme de recueil de données et de
commande en temps réel, branché sur le vrai système. De plus, le robot et la série de machines
peuvent être vus comme un sous-système, dans un système plus vaste. Lorsqu’on veut simuler
le système en entier, on peut vouloir définir plusieurs “copies” du sous-système, possiblement
avec des paramètres différents, et qui évolueront simultanément durant la simulation. Ces
sous-systèmes peuvent même interagir directement ou indirectement par le biais d’autres
sous-systèmes. Dans cet exemple, nous illustrons comment SIMOD permet de développer
des “simulateurs” pour différents types de sous-systèmes, puis d’intégrer ces “simulateurs”,
de façon modulaire, afin de former de plus gros sous-systèmes, etc.

En examinant ces exemples, le lecteur pourra se référer au besoin à la définition fonc-
tionnelle des objets fournis par les modules de SIMOD, à l’annexe A.



8 3 EXEMPLES DE PROGRAMMES SIMOD

3.1 Une file d’attente M/U/s

Considérons une file d’attente à s serveurs. Les clients arrivent selon un processus de
Poisson et les durées de service sont des variables aléatoires de loi uniforme. Un programme
SIMOD simulant un tel système est donné à la Figure 1. Pour simplifier, cette version
du programme ne lit aucune variable en entrée et les valeurs des paramètres du modèle
apparaissent directement dans le programme. Evidemment, en pratique (surtout dans le cas
des programmes réutilisés plusieurs fois), ces valeurs seront habituellement lues et placées
dans des variables.

Les énoncés IMPORT permettent d’importer des objets (types et procédures) des modules
de SIMOD. À remarquer qu’on peut importer explicitement des objets spécifiques, ou en-
core importer un module en entier pour utiliser un ou plusieurs des objets qui s’y trouvent.

MODULE Queue;

IMPORT SIM, EVENT, PROCS, RES, GENEP, RAND;
FROM RES IMPORT Resource, Request, Release, Report, ServicePolicy;
FROM PROCS IMPORT Delay, Terminate;
FROM RAND1 IMPORT Uniform;

VAR
Serveur : Resource;
Client : PROCS.ProcessType;
FinSim : EVENT.EventType;
Gener : GENEP.ArrivalProcess;
GenArr, GenServ : RAND.RngStream;

PROCEDURE ProcClient;
BEGIN
Request (1, Serveur);
Delay (Uniform (GenServ, 12.0, 16.0));
Release (1, Serveur);
Terminate;
END ProcClient;

PROCEDURE ProcFinSim;
BEGIN
Report (Serveur);
SIM.Stop;
END ProcFinSim;

BEGIN
RAND.CreateStream (GenArr, ’Generateur arrivees’);
RAND.CreateStream (GenServ, ’Generateur durees service’);
EVENT.Create (FinSim, ProcFinSim, ’Fin de la simulation’);
PROCS.Create (Client, ProcClient, 8000, ’Un client’);
RES.Create (Serveur, Fifo, 3, ’Poste de service’);
RES.CollectStat (Serveur);
GENEP.Create (Gener, "Processus d’arrivee");
SIM.Init;
EVENT.Schedule (FinSim, 1000.0, NIL);
GENEP.StartPoissonArrivalsP (GenArr, Gener, Client, NIL, 10.0, 0);
SIM.Start;
END Queue.

Figure 1: Simulation d’une file d’attente M/U/3.

Dans ce dernier cas, les identificateurs des objets utilisés doivent être “qualifiés” (préfixés
du nom de leur module) lors de leur utilisation. C’est au programmeur à décider du mode
d’importation pour chacun des objets importés. Son choix devra viser à faciliter la lisibilité
du programme. Une contrainte doit cependant être respectée: tous les identificateurs im-
portés explicitement doivent être uniques les uns par rapport aux autres, et par rapport aux
identificateurs déclarés localement. Par exemple, l’identificateur Create, disponible dans
plusieurs modules de SIMOD, doit être qualifié lorsqu’il est importé de plus d’un module.

Dans cet exemple, chaque client est vu comme un processus, dont les activités sont
décrites par la procédure ProcClient. Un client arrive, demande un serveur (Request),
attend si aucun serveur n’est disponible, occupe le serveur pendant un certain temps (Delay),
libère le serveur (Release), puis termine ses activités (Terminate). Lorsque le client appelle
Request pour demander un serveur, il obtient un serveur et poursuit son exécution s’il y en
a un de libre, sinon il est (automatiquement) placé dans une file d’attente et ne poursuit
son exécution que lorsqu’il obtient le serveur demandé. En appelant Delay, le processus
se bloque et doit attendre que l’horloge de la simulation avance d’un certain délai avant de
pouvoir poursuivre son exécution. Ici, ce délai correspond à sa durée de service, et est généré
aléatoirement (pour chaque client) selon une loi uniforme (continue) entre 12 et 16 unités de
temps, et en utilisant le générateur de valeurs pseudo-aléatoires GenServ.

La procédure ProcFinSim décrit un événement qui marque la fin de la simulation. Cette
procédure fait imprimer un rapport statistique complet sur l’utilisation de la ressource (le
serveur), puis arrête la simulation.

Dans le programme principal, on crée le type d’événement FinSim, associé à la procédure
ProcFinSim, et le type de processus Client, associé à la procédure ProcClient. Chaque pro-
cessus Client aura un espace-mémoire de 8000 octets pour son exécution. Pour chaque type
de processus, on doit ainsi déclarer l’espace-mémoire requis, et cet espace doit être suffisant
pour les variables locales de la procédure associée et de toutes les procédures qu’elle ap-
pelle directement ou indirectement. La variable Serveur a été déclarée comme représentant
un type de ressource. La procédure RES.Create crée cette ressource, qui correspond ici au
groupe de serveurs. La politique de service est Fifo (premier arrivé, premier servi), et le nom-
bre de serveurs est de 3. La châıne de caractères “Poste de service” sera utilisée par le progi-
ciel pour identifier cette ressource dans les rapports statistiques. L’appel à RES.CollectStat
sert à demander un recueil automatique de statistiques pour cette ressource. SIM.Init ini-
tialise le simulateur. Cela doit être fait avant toute prévision d’événement ou de processus.
On prévoit ensuite l’événement FinSim dans 1000 unités de temps, puis on démarre le pro-
cessus d’arrivée des clients. Gener est un processus d’arrivée, qui fera arriver les clients. La
procédure StartPoissonArrivalsP démarre ce processus. Les clients arrivent selon un pro-
cessus de Poisson de taux 0.1. Ainsi, les intervalles de temps entre les arrivées successives, de
même que le temps jusqu’à la première arrivée, sont des variables aléatoires exponentielles
de moyenne 10 (unités de temps). Leurs valeurs sont générées en utilisant le générateur
GenArr. Le dernier paramètre de StartPoissonArrivalsP indique habituellement le nom-
bre maximum de processus à générer. Ici, sa valeur est 0, ce qui indique que le processus de
Poisson générant les arrivées ne s’arrêtera que si on appelle StopArrivals (Gener), ou que

si la simulation se termine. L’appel à SIM.Start démarre la simulation, qui s’arrêtera après
1000 unités de temps pour imprimer le rapport apparaissant à la Figure 2.

Dans ce rapport, on voit que la capacité de la ressource est toujours demeurée à 3, le
nombre moyen de serveurs utilisés fut de 1.399, et la longueur moyenne de la file d’attente
fut de 0.042. Lorsque la simulation s’est arrêtée, 101 clients avaient débuté leur service, et
99 étaient sortis du système (2 étaient en train de se faire servir, et la file d’attente était
donc vide). Les temps moyens d’attente dans la file, de service, et de séjour dans le système
(attente + service) furent 0.416, 14.079 et 14.503 respectivement. À noter que la première
valeur est une moyenne sur 101 observations, tandis que les deux autres sont des moyennes
sur 99 observations, et c’est ce qui explique que la somme des deux premières valeurs n’est
pas exactement égale à la troisième valeur. Le rapport nous fournit également les écarts-
types échantillonnaux de ces observations. Cependant, il ne faut pas oublier qu’il s’agit
d’observations dépendantes , de sorte que ces écarts-types ne peuvent pas être utilisés pour
calculer des intervalles de confiance en utilisant les formules habituelles.

REPORT ON RESOURCE : Poste de service
From time 0.00 to time 1000.00

min max average standard dev. nb. Obs.
Capacity 3 3 3.000
Utilization 0 3 1.399
Queue Size 0 2 0.042
Wait 0.000 10.479 0.416 1.452 101
Service 12.029 15.958 14.079 1.266 99
Sojourn 12.029 24.474 14.503 2.012 99

Figure 2: Résultats de la simulation de la file M/U/3.

3.2 Un atelier de production

Cet exemple est inspiré de Law et Kelton [37], section 2.6, et de [38]. Un atelier de
production contient M groupes de machines. Pour m = 1, . . . ,M , le groupe m est constitué
de sm machines identiques. L’atelier est modélisé par un réseau de files d’attente. A chaque
groupe de machines est associée une file d’attente FIFO à plusieurs serveurs identiques
(sm serveurs pour le groupe m). Il y a N types de tâches qui arrivent à l’atelier, et pour
n = 1, . . . , N , les tâches de type n arrivent selon un processus de Poisson de taux λn (la
durée entre deux arrivées successives de tâches de type n est une variable aléatoire de loi
exponentielle de moyenne 1/λn).

Chaque type de tâche requiert une séquence d’opérations bien déterminée, chaque
opération devant être effectuée sur un type de machine bien précis. Les durées des opérations
sont déterministes. Une tâche de type n requiert pn opérations, qui doivent être ef-
fectuées dans l’ordre sur les machinesmn,1, mn,2, . . .mn,pn , et dont les durées respectives sont

dn,1, dn,2, . . . , dn,pn , en heures. À noter que rien n’empêche une tâche de repasser plusieurs
fois sur un même type de machine; il n’y a donc pas de limite supérieure sur la valeur de pn.

On veut simuler le fonctionnement de l’atelier pendant T heures. On suppose que l’atelier
est initialement vide, et on commencera à recueillir les statistiques après une période de
réchauffement de TR heures. On veut estimer : (a) le temps moyen passé dans le système
pour chaque type de tâche; (b) le taux moyen d’utilisation, le temps moyen d’attente, la
longueur moyenne de la file d’attente, etc. pour chaque type de machine.

Le programme SIMOD de la figure 3 permet d’effectuer cette simulation. Chaque groupe
de machines est vu comme une ressource, dont la capacité est égale au nombre de machines
(identiques) dans le groupe. Toutes les files d’attentes sont de type FIFO. Le programme
utilise un dossier (un “RECORD”) pour chaque type de tâche, afin de conserver l’information
associée: le taux d’arrivée, le nombre d’opérations à effectuer, les durées de ces opérations et
les groupes de machines sur lesquelles on doit les effectuer, de même qu’un bloc statistique
servant à recueillir des statistiques sur les temps passés dans le système pour les tâches de
ce type. Le chiffre 5 qui apparâıt dans la déclaration des types correspond à une borne
supérieure sur les nombres de groupes de machines, de types de tâches, et d’opérations par
tâche. La procédure LireDonnees lit les données dans un fichier dont le nom est fourni
interactivement par l’usager, et crée les types de machines et les types de tâches.

Chaque tâche dans le système est un processus dont la vie est décrite par la procédure
ProcTache. Les tâches sont générées par des processus de Poisson, un par type de
tâche, lesquels sont démarrés à partir du programme principal. Le second paramètre de
StartPoissonArrivalsP est le type de tâche à générer, et correspondra à la valeur prise
par l’attribut du processus tâche. La valeur de cet attribut est récupérée à l’intérieur de la
procédure ProcTache en appelant la fonction Attrib. Pour chaque opération à effectuer, la
tâche demande une machine du type approprié, se l’approprie pour la durée de l’opération,
puis la libère. Lorsque la tâche se termine, le temps qu’elle a passé dans l’atelier constitue
une nouvelle observation pour le bloc statistique DureesSejour, qui recueille des statistiques
sur les durées de séjour pour les tâches de ce type.



12 3 EXEMPLES DE PROGRAMMES SIMOD

Avant de démarrer la simulation, dans le programme principal, deux événements sont
prévus, l’un marquant la fin de la période de réchauffement et l’autre marquant la fin de
la simulation. Les procédures ProcFinRechauf et ProcFinSim, respectivement, s’exécutent
lors de l’occurence de ces événements. La première réinitialise tous les blocs statistiques,
tandis que la seconde fournit les rapports statistiques demandés et arrête la simulation.

MODULE Atelier;
IMPORT SIM, EVENT, PROCS, GENEP, STAT, RES, RAND;
FROM Storage IMPORT ALLOCATE;
FROM RES IMPORT Resource, Request, Release, ServicePolicy;
FROM MYINOUT IMPORT OpenInput, OpenOutput, CloseInput, ReadCard, ReadLn,

WriteString, ReadString, ReadLongReal;

TYPE
NumTypMachine = CARDINAL [1..5]; (* Un numero de groupe de machines. *)
NumTypTache = CARDINAL [1..5]; (* Un numero de type de tache. *)
NumOper = CARDINAL [1..5]; (* Un numero d’operation. *)
InfoTypTache = RECORD (* Information sur un type de tache. *)

TauxArrivee : LONGREAL; (* Taux d’arrivee. *)
DureesSejour : STAT.Block; (* Stats. sur les durees de sejour. *)
NOper : CARDINAL; (* Nb. d’operations a effectuer. *)
DureeOper : ARRAY NumOper OF LONGREAL; (* Durees des operations. *)
MachOper :ARRAY NumOper OF Resource;(*Machines requises pour ces oper.*)
END;

VAR
M, (* Nombre de groupes de machines. *)
N : CARDINAL; (* Nombre de types de taches. *)
j : CARDINAL; (* Indice d’un type de tache. *)
TypMachine : ARRAY NumTypMachine OF Resource; (* Groupes de machines. *)
TypTache : ARRAY NumTypTache OF POINTER TO InfoTypTache;

(* Tableau des types de taches. *)
ArrivTaches : ARRAY NumTypTache OF GENEP.ArrivalProcess;
Tache : PROCS.ProcessType;
FinRechauf, FinSim : EVENT.EventType;
DureeRechauf, DureeSim : LONGREAL;
GenArr : RAND.RngStream; (* Generateur d’arrivees. *)

PROCEDURE ProcTache;
VAR

TempsArrivee : LONGREAL; (* Instant d’arrivee de cette tache. *)
Typ : POINTER TO InfoTypTache; (* Type de cette tache. *)
Oper : NumOper; (* Numero de l’operation courante. *)

BEGIN
TempsArrivee := SIM.Time ();
Typ := PROCS.Attrib ();
WITH Typ^ DO

FOR Oper := 1 TO NOper DO
Request (1, MachOper [Oper]);
PROCS.Delay (DureeOper [Oper]);
Release (1, MachOper [Oper])
END;

STAT.Update (DureesSejour, SIM.Time () - TempsArrivee)
END;

PROCS.Terminate
END ProcTache;

Figure 3: Simulation d’un atelier de production.

PROCEDURE ProcFinRechauf;
VAR

m, j : CARDINAL;
BEGIN
FOR m := 1 TO M DO RES.InitStat (TypMachine [m]) END;
FOR j := 1 TO N DO STAT.Init (TypTache [j]^.DureesSejour) END
END ProcFinRechauf;

PROCEDURE ProcFinSim;
VAR

m, j : CARDINAL;
BEGIN
FOR m := 1 TO M DO RES.Report (TypMachine [m]) END;
FOR j := 1 TO N DO STAT.Report (TypTache [j]^.DureesSejour) END;
SIM.Stop
END ProcFinSim;

PROCEDURE LireDonnees;
VAR

Nom : ARRAY [0..31] OF CHAR;
Capacite : CARDINAL; (* Nb. de machines dans un groupe. *)
Oper : NumOper;
m, j : CARDINAL;

BEGIN
OpenInput (".dat"); OpenOutput (".res");
ReadLongReal (DureeRechauf); ReadLn ;
ReadLongReal (DureeSim); ReadLn ;
ReadCard (M); ReadCard (N); ReadLn;
FOR m := 1 TO M DO (* Donnees sur les machines. *)

ReadString (Nom); ReadCard (Capacite); ReadLn;
RES.Create (TypMachine [m], Fifo, Capacite, Nom);
RES.CollectStat (TypMachine [m])
END;

FOR j := 1 TO N DO (* Donnees sur les taches. *)
NEW (TypTache [j]);
WITH TypTache [j]^ DO

ReadString (Nom); ReadLongReal (TauxArrivee);
ReadCard (NOper);
FOR Oper := 1 TO NOper DO

ReadCard (m); ReadLongReal (DureeOper [Oper]);
MachOper [Oper] := TypMachine [m]
END;

STAT.Create (DureesSejour, STAT.Tally, Nom); ReadLn
END

END;
CloseInput
END LireDonnees;

BEGIN
RAND.CreateStream (GenArr, ’Generateur’);
EVENT.Create (FinSim, ProcFinSim, ’Fin de la simulation’);
EVENT.Create (FinRechauf, ProcFinRechauf, ’Fin du rechauffement’);
PROCS.Create (Tache, ProcTache, 5000, ’Une tache’);
LireDonnees;
EVENT.Schedule (FinSim, DureeSim, NIL);
EVENT.Schedule (FinRechauf, DureeRechauf, NIL);
FOR j := 1 TO N DO

GENEP.Create (ArrivTaches [j], " ");
GENEP.StartPoissonArrivalsP (GenArr, ArrivTaches [j], Tache, TypTache [j],

1.0 / (TypTache [j]^.TauxArrivee), 0)
END;

SIM.Start;
END Atelier.

Figure 3: Simulation d’un atelier de production (suite).

3.3 Un système à temps partagé

Cet exemple est inspiré de [37], section 2.4. Considérons un système informatique à temps
partagé, simplifié. Le système est constitué de T terminaux indépendants et identiques,
tous occupés, qui utilisent la même unité centrale (CPU). L’occupant d’un terminal soumet
une tâche au CPU, attend la réponse, puis réfléchit pendant une durée aléatoire avant de
soumettre une nouvelle tâche, etc. Le temps de réflexion est une variable aléatoire de loi
exponentielle de moyenne µ, tandis que le temps de CPU requis par une tâche est une variable
aléatoire de loi Weibull de paramètres α et λ. Les tâches en attente de service forment une file
d’attente pour le CPU, et sont servies selon une politique de “round-robin” avec quantum.
Lorsqu’une tâche s’empare du CPU, s’il lui reste moins de q secondes à exécuter (la valeur
du quantum), elle conserve le CPU jusqu’à la fin de son exécution. Sinon, elle s’exécute
pendant q secondes et retourne à la fin de la file. Dans les deux cas, il faut aussi h secondes
supplémentaires (“overhead”) pour l’enlever du CPU et, s’il y a lieu, y remettre une autre
tâche.

Le temps de réponse d’une tâche est défini comme étant le temps écoulé entre l’instant
où la tâche quitte son terminal et l’instant où se termine son enlèvement du CPU lorsqu’elle
a terminé son exécution. On s’intéresse au temps de réponse moyen, à l’état stationnaire.
On peut simuler le système jusqu’à ce que N tâches se soient terminées, en supposant
qu’initialement tous les terminaux sont dans l’état de réflexion. Afin d’alléger le biais ini-
tial causé par cette dernière hypothèse, on commence à recueillir les statistiques seulement
lorsque N0 tâches se seront terminées. De plus, on effectue R répétitions indépendantes de
cette simulation.

Supposons que l’on veut comparer deux configurations du système, du point de vue
du temps de réponse moyen. On définit ici une configuration du système comme étant
caractérisée par (T, q, h, µ, α, λ). Lorsque l’on veut comparer ainsi deux systèmes ou deux
configurations, on y gagne en général à utiliser des techniques de réduction de variance, dont
l’une des plus simples et des plus efficaces consiste à utiliser des valeurs aléatoires communes.

On voudrait, en fait, que les différences de performances observées entre les deux configu-
rations soient dues le plus exclusivement possible aux différences entre les configurations, et
le moins possible aux fluctuations aléatoires. On doit s’assurer que d’une configuration à
l’autre, la i-ième répétition utilise les mêmes séquences de valeurs pseudo-aléatoires U(0,1)
de base. On va coupler les répétitions deux à deux. Soient R1i et R2i les temps de réponse
moyens des i-ièmes répétitions pour les configurations 1 et 2, et soit

Di = R1i −R2i.

MODULE TempsPartage;

IMPORT SIM, STAT, PROCS, RES, RAND;
FROM RAND IMPORT ResetStream, SeedType;
FROM RAND1 IMPORT Weibull, Expon;
FROM RES IMPORT Resource, Request, Release, ServicePolicy;
FROM MYINOUT IMPORT OpenInput, OpenOutput, CloseInput, CloseOutput,

ReadCard, ReadLn, ReadLongReal;

VAR
TotTaches : CARDINAL; (* Nombre de fins de taches avant *)

(* la fin de la simulation. *)
PrelimTaches : CARDINAL; (* Nombre de fins de taches avant *)

(* la fin du rechauffement. *)
NTaches : CARDINAL; (* Nombre de taches deja terminees. *)
NTer : CARDINAL; (* Nombre de terminaux. *)
Quantum : LONGREAL; (* Duree d’un quantum. *)
Overhead : LONGREAL; (* Duree de l’overhead. *)
TReflexMoy : LONGREAL; (* Temps moyen de reflexion. *)
Alpha, Lambda : LONGREAL; (* Par. de la loi Weibull du temps *)

(* de CPU par tache. *)
CPU : Resource; (* Le CPU. *)
Terminal : PROCS.ProcessType; (* Un terminal. *)
NRep, Rep : CARDINAL; (* Nb. de repet. et numero de la *)

(* repet. courante. *)
MoyConf1 : ARRAY [1..1000] OF LONGREAL; (* Temps de reponse moy. des *)

(* repetitions pour la config. 1. *)
TRepMoy : STAT.Block; (* Stat. sur les temps de reponse *)

(* pour la repetition courante. *)
Diff : STAT.Block; (* Stat. sur les differences entre *)

(* les temps de reponse. *)
Conf : [1..2]; (* No. de la configation courante. *)
GenDelay,
GenTCPU : RAND.RngStream; (* Generateurs aleatoires. *)

PROCEDURE LireDonnees;
BEGIN
ReadCard (NTer); ReadLongReal (Quantum);
ReadLongReal (Overhead);
ReadLongReal (TReflexMoy);
ReadLongReal (Alpha);
ReadLongReal (Lambda);
ReadCard (TotTaches); ReadCard (PrelimTaches); ReadCard (NRep);
ReadLn;
END LireDonnees;

Figure 4: Simulation d’un système à temps partagé.

PROCEDURE ProcTerminal;
(* Decrit la suite des activites reliees a un terminal. *)
VAR

TArrivee : LONGREAL; (* Instant d’arrivee de la derniere *)
(* tache de ce terminal. *)

TCPURestant : LONGREAL; (* Temps cpu encore requis. *)
BEGIN
WHILE (NTaches <= TotTaches) DO

PROCS.Delay (Expon (GenDelay, TReflexMoy)); (* Temps de reflexion. *)
TArrivee := SIM.Time(); (* Arrivee d’une tache. *)
TCPURestant := Weibull (GenTCPU, Alpha, Lambda); (* Duree de la tache*)
WHILE (TCPURestant > Quantum) DO

Request (1, CPU); (* Il reste plus d’un quantum. *)
PROCS.Delay (Quantum + Overhead);
TCPURestant := TCPURestant - Quantum;
Release (1, CPU)
END;

Request (1, CPU); (* Il reste une fraction de quantum. *)
PROCS.Delay (TCPURestant + Overhead);
Release (1, CPU); (* Fin de la tache. *)
STAT.Update (TRepMoy, SIM.Time () - TArrivee);
IF (NTaches = PrelimTaches) THEN

STAT.Init (TRepMoy) END; (* Fin du rechauffement. *)
INC (NTaches)
END;

SIM.Stop; PROCS.Terminate (* Ici, on a NTaches = TotTaches. *)
END ProcTerminal;

PROCEDURE SimulerUneRepetition;
VAR

i : CARDINAL;
BEGIN
PROCS.KillAll; SIM.Init; STAT.Init (TRepMoy); RES.Init (CPU);
NTaches := 0;
FOR i := 1 TO NTer DO (* Activer les terminaux. *)

PROCS.Schedule (Terminal, 0.0, NIL) END;
SIM.Start
END SimulerUneRepetition;

BEGIN
RAND.CreateStream (GenDelay, ’Generateur aleatoire pour le delais’);
RAND.CreateStream (GenTCPU, ’Generateur pour le temps restant’);
PROCS.Create (Terminal, ProcTerminal, 1024*8, ’Un terminal’);
RES.Create (CPU, Fifo, 1, ’Le CPU’);
STAT.Create (TRepMoy, STAT.Tally, ’Temps reponse moyen’);
STAT.Create (Diff, STAT.Tally, ’Differences’);
OpenInput (".dat"); OpenOutput (".res");
FOR Conf := 1 TO 2 DO

LireDonnees;
IF Conf = 2 THEN

ResetStream (GenDelay, StartStream); ResetStream (GenTCPU, StartStream)
END;

FOR Rep := 1 TO NRep DO
SimulerUneRepetition;
IF Conf = 1 THEN

MoyConf1 [Rep] := STAT.Average (TRepMoy)
ELSE

STAT.Update (Diff, MoyConf1 [Rep] - STAT.Average (TRepMoy)) END;
ResetStream (GenDelay, NextBlock); ResetStream (GenTCPU, NextBlock)
END

END;
STAT.ReportConfidenceInterval (Diff, 0.9);
CloseInput; CloseOutput;
END TempsPartage.

Figure 4: Simulation d’un système à temps partagé (suite).

Les Di sont des variables aléatoires indépendantes et on peut utiliser leurs valeurs pour
obtenir un intervalle de confiance pour la différence d entre les temps de réponse des deux
systèmes. Le fait d’avoir obtenu R1i et R2i en utilisant des valeurs aléatoires communes
devrait, si R1i et R2i sont positivement corrélées, réduire la variance des Di et la largeur de
l’intervalle de confiance pour d.

Le programme SIMOD de la figure 4 permet d’effectuer cette simulation. La variable
Conf désigne le numéro de la configuration courante. Le tableau MoyConf1 sert à mémoriser
les valeurs des R1i, et le bloc statistique Diff sert à recueillir les statistiques sur les Di et à
calculer un intervalle de confiance pour d.

Pour chaque répétition, le bloc statistique TRepMoy cumule les statistiques sur les temps
de réponse des N −N0 tâches qui se terminent durant la période de recueil des statistiques.
On l’initialise au début de la répétition et y effectue une mise à jour chaque fois que l’une de
ces tâches se termine. Entre une répétition et sa suivante, on doit initialiser à zéro le nombre
de tâches terminées, réinitialiser la simulation et le bloc statistique TRepMoy, et fournir des
nouveaux germes aux deux générateurs utilisés, afin de s’assurer que d’une configuration
à l’autre, les répétitions correspondantes se fassent toutes à partir des mêmes germes et
utilisent des valeurs aléatoires communes. Avant de passer à la seconde configuration, les
générateurs sont réinitialisés à leurs germes initiaux.

Au début de chaque répétition, on active aussi un processus pour chaque terminal.
Lorsque la N -ième tâche se termine, le processus correspondant à cette tâche appelle
SIM.Stop, ce qui arrête la simulation et retourne le contrôle au programme principal, juste
après l’appel à SimulerUneRepetition. Lorsqu’on réinitialise la simulation, tous les proces-
sus de la répétition précédente (s’il y a lieu) sont détruits.

REPORT ON STATISTIC (Tally) : Differences
min max average standard dev. nb. Obs.
-1.401 0.479 0.129 0.198 1000

90% confidence interval for mean : ( 0.119, 0.140 )

Figure 5: Simulation d’un système à temps partagé.

Soient T = 20, h = .001, µ = 5 sec., α = 1/2 et λ = 1 pour les deux configurations. Pour
ces valeurs, la moyenne de la loi Weibull est de 2. On choisit q = 0.1 pour la configuration
1 et q = 0.2 pour la configuration 2. On choisit aussi N0 = 100, N = 1100 et R = 1000
répétitions. On obtient les résultats de la figure 5. L’intervalle de confiance ne contenant pas
zéro, on peut en conclure que les temps de réponse moyens avec q = 0.2 sont significativement
plus courts qu’avec q = 0.1.

A noter qu’on pourrait rendre le modèle plus réaliste en considérant par exemple
différents “types” de terminaux avec des paramètres différents, un nombre de terminaux
variable en fonction du temps, différentes classes de tâches avec des priorités, des demandes
d’entrées/sorties, etc. Le programme deviendrait plus élaboré, mais la structure de base
serait similaire. SIMOD permet très bien de simuler ce genre de modèle complexe.

3.4 Des visites guidées

On offre des visites guidées d’un site touristique. Trois hôtesses travaillent à cet endroit pour
guider les visiteurs. Le site ouvre à 10 heures et ferme à 16 heures. Les groupes de clients
arrivent selon un processus de Poisson à partir de 9:45 heures jusqu’à 16 heures, au taux
d’un groupe par 3 minutes. Le nombre de visiteurs dans un groupe est une variable aléatoire
X qui prend la valeur i avec une probabilité pi donnée au tableau suivant:

i 1 2 3 4
pi .2 .6 .1 .1

Les visiteurs qui arrivent avant 10 heures doivent attendre l’ouverture. Après 16 heures,
on termine les visites déjà amorcées, mais aucune nouvelle visite n’est entreprise. Ainsi, les
visiteurs qui sont encore dans la file d’attente après 16 heures doivent s’en retourner.

Une visite dure 45 minutes et nécessite une hôtesse. Une hôtesse ne peut pas faire visiter
plus de 15 personnes à la fois, et n’amorcera une visite que s’il y a au moins 8 personnes en
attente (sauf à 16 heures, où s’il y a une hôtesse de libre, celle-ci amorcera une visite avec
le dernier groupe de personnes même s’ils sont moins que 8). A midi, chaque hôtesse libre
prend 30 minutes pour manger. Les hôtesses qui sont occupées à ce moment prendront 30
minutes pour manger dès qu’elles termineront leur visite en cours. Lorsqu’une hôtesse se
libère (termine une visite sans devoir aller manger, revient de manger, ou débute sa journée),
s’il y a moins de 8 personnes en attente, elle attend, sinon elle entreprend une visite. S’il y
a plus de 15 personnes en attente, 15 d’entre elles iront avec l’hôtesse.

Les personnes qui veulent visiter forment une seule file d’attente pour les hôtesses. Par-
fois, lorsqu’un groupe arrive et qu’il y a trop de monde dans la file d’attente, celui-ci renonce à
attendre et s’en va. La probabilité d’une telle renonciation dépend du nombre n de personnes
déjà dans la file, et est donnée par:

R(n) =


0 pour n ≤ 10;
(n− 10)/30 pour 10 < n < 40;
1 pour n ≥ 40.

On veut estimer le nombre de visiteurs perdus par jour, en moyenne. Les visiteurs perdus
au cours d’une journée sont ceux qui renoncent à attendre, et ceux qui sont encore dans la
file après 16 heures.

Le programme SIMOD apparâıt à la figure 6. Pour simplifier les entrées/sorties, les
paramètres du modèle ont été placés directement dans le programme. En pratique, ces
valeurs seront habituellement lues dans un fichier et placées dans des variables. Le temps
est mesuré en heures et l’instant 0.0 correspond à minuit. Ainsi, par exemple, 9:45 heures
correspond à l’instant 9.75, et 30 minutes correspondent à 0.5 unités de temps. Le processus
Hotesse correspond à une hôtesse. Le processus GenArrivee génère les arrivées des groupes
de visiteurs, et l’événement Fermeture correspond à la fermeture du site, à 16 heures. Le
générateur d’arrivées utilise le “Bin” VisiteOK (qui contient toujours 0 ou 1 jeton), afin de

pouvoir “réveiller” une hôtesse lorsque le nombre de visiteurs en attente devient suffisamment
élevé pour démarrer une visite. Les résultats de la simulation apparaissent à la figure 7.

A remarquer que nous aurions pu générer les arrivées des groupes de façon différente, en
considérant chaque arrivée comme un événement, et en utilisant la procédure StartPois-

sonArrivalsE du module GENEP. On pourrait aussi penser utiliser une Condition du module
COND à la place du Bin. La Condition deviendrait vraie lorsqu’il y aurait au moins 8 per-
sonnes en attente. Malheureusement, cela poserait une difficulté importante, car en mettant
la Condition à vrai, on réveillerait toutes les hôtesses en attente, et non pas seulement la
première.

MODULE Visites;

IMPORT SIM, STAT, PROCS, LIST, BIN, EVENT, RAND;
FROM RAND IMPORT RngStream, RandU01;
FROM RAND1 IMPORT DiscreteDist, CreateDiscreteDist, Discrete, Expon;
FROM BIN IMPORT Bin, Take, Give, Avail, WaitList, ServicePolicy;
FROM PROCS IMPORT ProcessType, Delay;
FROM EVENT IMPORT EventType;
FROM MYINOUT IMPORT WriteString, WriteLn;

VAR
TailleQ : CARDINAL; (* Taille de la file d’attente. *)
NbPerdus : CARDINAL; (* Nb. de visiteurs perdus pour la *)

(* Repetition courante. *)
VisiteOK : Bin; (* Contient un jeton ssi il y a assez *)

(* de monde pour demarrer une visite. *)
TailleArriv : DiscreteDist; (* Fct. de repartition de la taille *)

(* groupes qui arrivent. *)
Hotesse : ProcessType; (* Une hotesse. *)
GenArrivee : ProcessType; (* Generateur d’arrivees. *)
Fermeture : EventType; (* La fermeture du site. *)
StatPertes : STAT.Block; (* Stat. sur les pertes pour les *)

(* differentes repetitions. *)
Rep : CARDINAL; (* No. de la repet. courante. *)
i : CARDINAL;
V, F : ARRAY [0..3] OF LONGREAL; (* Valeurs possibles de la taille X *)

(* d’un groupe, et fonction de *)
(* repartition de X. *)

GenDelay,
GenNbr,
GenDep : RngStream; (* Generateurs aleatoires. *)

PROCEDURE ProcFermeture;
BEGIN
IF LIST.Size (WaitList (VisiteOK)) = 0 THEN (* Aucune hotesse de libre. *)

NbPerdus := NbPerdus + TailleQ
END;

SIM.Stop
END ProcFermeture;

Figure 6: Simulation de visites guidées.

PROCEDURE ProcHotesse;
VAR

Lunch : BOOLEAN;
BEGIN
Lunch := FALSE; (* N’a pas encore mange. *)
WHILE TRUE DO

IF (NOT Lunch) AND (SIM.Time () > 12.0) THEN (* Aller manger. *)
Delay (0.5); Lunch := TRUE END;

Take (1, VisiteOK);
IF TailleQ > 15 THEN

TailleQ := TailleQ - 15
ELSE

TailleQ := 0 END;
IF (TailleQ >= 8) THEN (* Il y a encore assez de monde. *)

Give (1, VisiteOK) END;
Delay (0.75); (* Duree de la visite. *)
END

END ProcHotesse;

PROCEDURE ProcGenArrivee;
VAR

Temp : CARDINAL ;
BEGIN
WHILE TRUE DO

Delay (Expon (GenDelay, 0.05));
IF ((TailleQ > 10)

AND (RandU01 (GenDep) < (LFLOAT (TailleQ - 10)) / 30.0)
OR (TailleQ >= 40)) THEN (* Ces visiteurs renoncent. *)

Temp := TRUNC (Discrete (GenNbr, TailleArriv)) ;
NbPerdus := NbPerdus + Temp;

ELSE
Temp := TRUNC (Discrete (GenNbr, TailleArriv)) ;
TailleQ := TailleQ + Temp;
IF (TailleQ >= 8) AND (Avail (VisiteOK) = 0) THEN

Give (1, VisiteOK) END (* Il y a maintenant assez de monde. *)
END

END
END ProcGenArrivee;

PROCEDURE SimulerUneRepetition;
VAR

i : CARDINAL;
BEGIN
BIN.Init (VisiteOK); PROCS.KillAll; SIM.Init;
TailleQ := 0; NbPerdus := 0;
EVENT.Schedule (Fermeture, 16.0, NIL);
PROCS.Schedule (GenArrivee, 9.75, NIL);
FOR i := 1 TO 3 DO PROCS.Schedule (Hotesse, 10.0, NIL) END;
SIM.Start;
STAT.Update (StatPertes, LFLOAT (NbPerdus))
END SimulerUneRepetition;

Figure 6: Simulation de visites guidées (suite).

BEGIN
RAND.CreateStream (GenDelay, ’Generateur pour delais inter-arrivees’);
RAND.CreateStream (GenDep, ’Generateur pour les departs’);
RAND.CreateStream (GenNbr, ’Generateur pour le nombre de clients’);
EVENT.Create (Fermeture, ProcFermeture, ’Fermeture du site’);
PROCS.Create (GenArrivee, ProcGenArrivee, 1024*8, "Generateur d’arrivees");
PROCS.Create (Hotesse, ProcHotesse, 1024*8, ’Une hotesse’);
BIN.Create (VisiteOK, Fifo, ’Groupe’);
STAT.Create (StatPertes, STAT.Tally, ’Les clients perdus’);
FOR i := 0 TO 3 DO V[i] := LFLOAT (i + 1) END;
F[0] := 0.2; F[1] := 0.8; F[2] := 0.9; F[3] := 1.0;
CreateDiscreteDist (TailleArriv, 4, V, F);

FOR Rep := 1 TO 10 DO SimulerUneRepetition END;
STAT.ReportConfidenceInterval (StatPertes, 0.9);
RAND.DeleteStream (GenDelay);
RAND.DeleteStream (GenDep);
RAND.DeleteStream (GenNbr);
END Visites.

Figure 6: Simulation de visites guidées (suite).

REPORT ON STATISTIC (Tally) : Les clients perdus
min max average standard dev. nb. Obs.
10.000 58.000 26.200 14.498 10

90% confidence interval for mean : ( 17.796, 34.604 )

Figure 7: Résultats de la simulation des visites guidées.

3.5 Simulation continue: un système proies–prédateurs

Considérons un système classique constitué de deux types d’animaux: les proies et les
prédateurs, les premiers étant la nourriture des seconds. Soient x(t) et z(t) les nombres
respectifs de proies et de prédateurs au temps t. On suppose que l’évolution de ces variables
se fait selon les équations différentielles suivantes:

x′(t) = rx(t)− cx(t)z(t)

z′(t) = −sz(t) + dx(t)z(t)

avec pour valeurs initiales x(0) = x0 > 0 et z(0) = z0 > 0. A remarquer ici que nous
approximons un système à espace d’états discret (dans le monde réel, le nombre d’animaux
de chaque type est entier) par un système à espace d’états continu. Il s’agit ici d’un système
d’équations différentielles de Lotka-Volterra, que l’on pourrait en principe résoudre de façon
analytique. Cependant, pour illustrer l’utilisation du module CONT de SIMOD, nous allons
le résoudre numériquement. Cela est fait dans le programme de la figure 8.

Ce programme lit les paramètres du modèle dans un fichier d’extension “.DAT”, et im-
prime la trajectoire de (t, x(t), z(t)) dans un fichier d’extension “.RES”. Les noms de ces
fichiers sont fournis au terminal par l’usager. Le second fichier pourra être traité par la suite,
par exemple pour tracer le graphique de cette trajectoire. Les procédures Dx et Dz retournent
la dérivée en fonction du temps pour les valeurs de x et z respectivement. L’événement UnPas
s’exécute à chaque pas d’intégration, et permet d’imprimer la trajectoire. Cet événement
n’est associé qu’à la variable x, et non pas à x et z, car on veut qu’il ne soit exécuté qu’une
seule fois à chaque pas d’intégration. On utilise ici la méthode de Runge Kutta d’ordre 4.

MODULE ProiePred;
IMPORT SIM, EVENT, CONT;
FROM MYINOUT IMPORT WriteLn, WriteLongReal, WriteString, ReadLongReal,

OpenInput, OpenOutput, CloseInput, CloseOutput;
FROM EVENT IMPORT EventType;
FROM CONT IMPORT VarCont, Value, SelectMethod, Method, StartInteg;
VAR

r, c, s, d, x0, z0 : LONGREAL; (* Parametres du modele. *)
x, z : VarCont; (* Nb. de proies et de predateurs. *)
Duree : LONGREAL; (* Duree de la simulation. *)
FinSim : EventType; (* Fin de la simulation. *)
UnPas : EventType; (* Evenement execute a chaque pas. *)

PROCEDURE ProcFinSim;
BEGIN SIM.Stop END ProcFinSim;

PROCEDURE ProcUnPas;
BEGIN
WriteLongReal (SIM.Time (), 4, 12); WriteString(" ") ;
WriteLongReal (Value (x), 4, 12); WriteString(" ") ;
WriteLongReal (Value (z), 4, 12);
WriteLn;
END ProcUnPas;

PROCEDURE Dx (t : LONGREAL) : LONGREAL;
BEGIN RETURN r * Value (x) - c * Value (x) * Value (z) END Dx;

PROCEDURE Dz (t : LONGREAL) : LONGREAL;
BEGIN RETURN - s * Value (z) + d * Value (x) * Value (z) END Dz;

BEGIN
OpenInput (".dat"); OpenOutput (".res");
ReadLongReal (r); ReadLongReal (c);
ReadLongReal (s); ReadLongReal (d);
ReadLongReal (x0); ReadLongReal (z0);
ReadLongReal (Duree);
EVENT.Create (UnPas, ProcUnPas, "Un pas d’integration");
EVENT.Create (FinSim, ProcFinSim, "Fin de la simulation");
EVENT.Schedule (FinSim, Duree, NIL);
CONT.Create (x, x0, Dx, NIL); CONT.Create (z, z0, Dz, NIL);
SelectMethod (RungeKutta4, 5.0);
StartInteg (x, UnPas, NIL); StartInteg (z, NIL, NIL);
SIM.Start; CloseInput; CloseOutput
END ProiePred.

Figure 8: Simulation d’un système proies-prédateurs.

3.6 Simulation discrète et continue: chauffage de lingots

Cet exemple est inspiré de [47], page 210. Dans une aciérie, des lingots d’acier arrivent à une
fournaise de trempage pour être chauffés avant de poursuivre leur route vers l’étape suivante
du processus. Les lingots arrivent selon un processus de Poisson, et le temps moyen entre les
arrivées est de a heures. Il y a de la place pour N lingots dans la fournaise. Lorsqu’un lingot
arrive à la fournaise, on l’y insère s’il reste de la place. Sinon, le lingot est placé dans une
zone d’attente pour les lingots froids, et attend qu’il y ait de la place dans la fournaise. La
température initiale d’un lingot qui arrive est une variable aléatoire qui suit à peu près une
loi normale de moyenne µT et d’écart-type σT (en degrés celsius). On suppose cependant que
tous les lingots placés dans la zone d’attente ont une température égale à τT degrés celsius
lorsqu’ils sont insérés dans la fournaise.

La variation de température des lingots lorsqu’ils sont chauffés par la fournaise est décrite
par:

dTj

dt
= (T − Tj)hj

où Tj est la température du lingot j et hj est le coefficient de chauffage du lingot j. Pour
chaque lingot, le coefficient hj est la valeur d’une variable aléatoire normale de moyenne µC

et d’écart-type σC . Les lingots sont chauffés dans la fournaise jusqu’à ce que leur température
atteigne S degrés celsius. A ce moment, ils sont retirés du four. La fournaise est chauffée de
sorte que sa température T évolue dans le temps selon l’équation différentielle:

dT

dt
= (K − T )c−

∑
j∈Γ

(T − Tj)bhj,

où b et c sont des constantes, et Γ est l’ensemble des lingots qui sont dans le four.

Le programme SIMOD donné à la figure 3.6 permet de simuler un tel système. Il combine
à la fois la simulation à événements discrets, avec vision par événements et par processus, et
la simulation continue. Encore une fois, pour simplifier, les valeurs des paramètres du modèle
ont été placées directement dans le programme. En pratique, il est évidemment recommandé
de généraliser le programme en plaçant ces valeurs dans des variables. Ici, nous avons utilisé
les valeurs suivantes: N = 5, a = 15.0, µT = 230, σT = 12, τT = 200, K = 1425, µC = 0.15,
σC = 0.01, b = 0.5, c = 0.2, S = 1200. La température initiale de la fournaise est de 600
degrés celsius et cette dernière ne contient aucun lingot. Le système est simulé pour 1000
heures d’opération, et le programme produit le rapport de la figure 3.6 sur l’utilisation de la
fournaise.

Le premier lingot n’arrive qu’après une durée exponentielle à partir du début de la simu-
lation, mais on commence à chauffer la fournaise dès le début de la simulation. L’événement
FinSimmarque la fin de la simulation. Les places dans le four sont vues comme une ressource,
de capacité N , et pour laquelle on demande un recueil automatique des statistiques. Pour
la partie continue, les équations différentielles sont intégrées numériquement en utilisant
une méthode de Runge-Kutta d’ordre 2. Il y a une variable continue qui correspond à la
température du four, et à la température de chaque lingot. Chaque lingot est représenté par

MODULE Lingots;

IMPORT SIM, EVENT, STAT, PROCS, LIST, RES, CONT, RAND;
FROM SYSTEM IMPORT ADDRESS, CAST;
FROM Storage IMPORT ALLOCATE, DEALLOCATE;
FROM RAND1 IMPORT Normal, Expon;
FROM EVENT IMPORT EventType, EventNotice, NoticeAttrib;
FROM CONT IMPORT VarCont, Value, VarAttrib, CurrentVar, SelectMethod,

Method, StartInteg, Delete;
FROM PROCS IMPORT ProcessType, ProcessInstance, Delay, Suspend, Resume,

InstanceAttrib, CurrentProcess, Terminate;
FROM LIST IMPORT List, View, OutPosition;
FROM RES IMPORT Resource, Request, Release, ServList, User, UserDossier,

Avail, ServicePolicy, Report;
TYPE

Lingot = POINTER TO RECORD (* Un lingot. *)
Proc : ProcessInstance; (* Le processus associe a ce lingot. *)
Tj : VarCont; (* La temperature du lingot. *)
hj : LONGREAL (* Coefficient de rechauffement. *)
END;

VAR
PlaceDansFour : Resource; (* Les places dans le four. *)
UnLingot : ProcessType; (* Decrit le comportement d’un lingot. *)
GenArrivees : ProcessType; (* Genere les arrivees des lingots. *)
FinSim : EventType; (* Fin de la simulation. *)
TestS : EventType; (* Verifie la temperature d’un lingot. *)
TemperFour : VarCont; (* La temperature du four. *)
GenDelay,
GenRech,
GenTemp : RAND.RngStream; (* Generateurs aleatoires. *)

PROCEDURE ProcLingot;
VAR

L : Lingot;
BEGIN
L := PROCS.Attrib ();
WITH L^ DO

IF Avail (PlaceDansFour) = 0 THEN CONT.Init (Tj, 200.0) END;
Request (1, PlaceDansFour);
StartInteg (Tj, TestS, L); Suspend;
(* Le lingot vient maintenant d’atteindre la temperature S. *)
Release (1, PlaceDansFour); Delete (Tj); DISPOSE (L); Terminate
END

END ProcLingot;

PROCEDURE ProcTestS;
VAR

L : Lingot;
BEGIN
L := EVENT.Attrib ();
IF Value (L^.Tj) >= 1200.0 THEN Resume (L^.Proc) END
END ProcTestS;

PROCEDURE DerivLingot (t : LONGREAL) : LONGREAL;
VAR

L : Lingot;
BEGIN
L := VarAttrib (CurrentVar ());
RETURN L^.hj * (Value (TemperFour) - Value (CurrentVar ()))
END DerivLingot;

Figure 9: Chauffage de lingots dans un four.

PROCEDURE DerivFour (t : LONGREAL) : LONGREAL;
VAR

L : Lingot;
Sum : LONGREAL;
U : ADDRESS;
Serv : List;

BEGIN
Serv := ServList (PlaceDansFour);
Sum := 0.2 * (1425.0 - Value (TemperFour));
View (U, First, Serv);
WHILE U # NIL DO

L := InstanceAttrib (User (CAST (UserDossier, U)));
Sum := Sum - 0.5 * L^.hj * (Value (TemperFour) - Value (L^.Tj));
View (U, Next, Serv)
END;

RETURN Sum
END DerivFour;

PROCEDURE ProcFinSim;
BEGIN Report (PlaceDansFour); SIM.Stop END ProcFinSim;

PROCEDURE ProcGenArrivees;
VAR

L : Lingot;
BEGIN
WHILE TRUE DO

Delay (Expon (GenDelay, 15.0)); NEW (L);
WITH L^ DO

hj := Normal (GenRech, 0.15, 0.01);
CONT.Create (Tj, Normal (GenTemp, 230.0, 12.0), DerivLingot, L);
PROCS.ScheduleNamedNext (UnLingot, L, Proc)
END

END
END ProcGenArrivees;

BEGIN
RAND.CreateStream (GenDelay, ’Generateur de delais inter-arrivees’);
RAND.CreateStream (GenRech, ’Generateur pour le rechauffement’);
RAND.CreateStream (GenTemp, ’Generateur pour la temperature’);
RES.Create (PlaceDansFour, Fifo, 5, "Places dans le four");
RES.CollectStat (PlaceDansFour);
PROCS.Create (GenArrivees, ProcGenArrivees, 2048*4, "Generateur d’arrivees");
PROCS.Create (UnLingot, ProcLingot, 2048*4, "Un lingot");
EVENT.Create (FinSim, ProcFinSim, "Fin de la simulation");
EVENT.Create (TestS, ProcTestS, "Test si temper. >= S");
CONT.Create (TemperFour, 600.0, DerivFour, NIL);

EVENT.Schedule (FinSim, 1000.0, NIL);
PROCS.Schedule (GenArrivees, 0.0, NIL);
SelectMethod (RungeKutta2, 0.01);
StartInteg (TemperFour, NIL, NIL);
SIM.Start;
RAND.DeleteStream (GenDelay);
RAND.DeleteStream (GenRech);
RAND.DeleteStream (GenTemp);
END Lingots.

Figure 9: Chauffage de lingots dans un four (suite).

un pointeur sur un enregistrement contenant son coefficient hj, la variable continue qui décrit
sa température Tj, et le processus associé. Le comportement de ce processus est décrit par
la procédure ProcLingot, et les lingots sont générés par le processus GenArrivees, dont le
comportement est décrit dans ProcGenArrivees. Le paramètre (attribut) associé à chaque
“processus” lingot lors de sa création est en fait le lingot lui-même. Il en est de même pour
la variable continue qui décrit sa température. A noter que les variables hj et Tj ne peuvent
pas être simplement des variables locales de la procédure ProcLingot, car si tel était le cas,
on ne pourrait pas leur accéder de l’extérieur. Au début de l’exécution de sa procédure, le
lingot récupère le pointeur sur son enregistrement, et le place dans L. Il regarde ensuite si le
four est plein, auquel cas il réinitialise sa température à τT . Puis, il demande une place dans
le four. Dès qu’il obtient cette place, il démarre le processus d’intégration qui fera évoluer sa
température, et se place dans l’état Suspended. A chaque pas d’intégration, un événement
de type TestS sera exécuté, avec pour paramètre ce lingot. En exécutant cet événement, on
vérifie si la température du lingot a atteint le seuil S, et si oui, on réactive le lingot. Lors de
cette réactivation, le lingot libère sa place dans le four, détruit la variable continue qui lui
est associée (afin d’en arrêter l’intégration et d’en récupérer l’espace), puis disparâıt.

REPORT ON RESOURCE : Places dans le four
From time 0.00 to time 1000.00

min max average standard dev. nb. Obs.
Capacity 5 5 5.000
Utilization 0 5 2.042
Queue Size 0 2 0.088
Wait 0.000 21.353 1.251 4.020 70
Service 14.522 55.846 29.566 11.430 69
Sojourn 14.522 63.898 30.836 12.330 69

Figure 10: Statistiques sur l’utilisation du four.

3.7 Entretien d’un groupe de machines par un robot

Description du modèle. On retrouve le genre de système décrit ici dans certaines usines
de produits textiles [41]. Soit un groupe de N machines identiques placées en ligne droite,
équidistantes, numérotées 1, 2, . . . , N , de gauche à droite. Chaque machine fonctionne pour
une durée aléatoire, puis se bloque, attend que l’on vienne la réparer, se fait réparer, fonc-
tionne pour une autre période de durée aléatoire, etc. Un robot patrouille la ligne pour
réparer les machines qui se bloquent. La distance entre deux machines voisines est d’une
unité, et le robot voyage à vitesse constante, de sorte qu’il lui faut |j − i| unités de temps
pour voyager de la machine i à la machine j sans s’arrêter. Le robot peut changer de direc-
tion n’importe où, même entre les machines, et il peut aussi s’arrêter en face d’une machine.
Lorsqu’il se trouve en face d’une machine bloquée, il peut décider de la réparer et, dans ce
cas, il ne peut pas se déplacer tant que la réparation n’est pas terminée. Il arrive parfois
que le robot ne réussisse pas à compléter lui-même une réparation, et doive faire appel à
un réparateur pour la compléter manuellement. Dans ce cas, le robot demeure en face de
la machine brisée jusqu’à ce que le réparateur ait terminé la réparation. La durée d’une
réparation par le robot, celle d’une réparation manuelle par le réparateur, et le temps entre
la fin d’une réparation et le prochain bloquage, sont des variables aléatoires indépendantes
dont les lois de probabilité sont connues.

1 2 Ns s s
����
R

Les décisions concernant les mouvements du robot sont prises en suivant une politique
de commande, c’est-à-dire une règle qui, dépendant de l’état actuel du système, détermine
ce que le robot doit faire. Le robot peut prendre une décision lorsqu’il arrive à destination,
ou lors d’une fin de réparation, ou lorsqu’une machine se bloque pendant qu’il n’est pas en
train d’effectuer une réparation.

L’état du système, tel qu’observé par le robot, est un couple (p, E), où p ∈ [1, N ] indique
la position du robot sur la ligne, et E ⊆ {1, . . . , N} est l’ensemble des machines qui sont
en panne. Chaque décision d du robot est un élément de l’ensemble D = {0, 1, . . . , N}. Le
choix de d ∈ {1, . . . , N} indique que le robot se déplace en direction de la position d. Si une
machine se bloque avant qu’il atteigne cette position, il reconsidère alors sa décision, sinon
il doit prendre une nouvelle décision lorsqu’il atteint la position d (c’est-à-dire dans |p − d|
unités de temps). Si d = p, le robot s’arrête et attend la prochaine panne. Si d = 0, il décide
de réparer la machine qui se trouve devant lui. Cette dernière décision n’est admissible que
lorsque le robot se trouve devant une machine bloquée.

On veut estimer le taux de production moyen du système, c’est-à-dire le nombre moyen
de machines qui fonctionnent, en fonction du temps, sur horizon infini. La politique de
commande du robot n’est pas fixée d’avance. On pourrait par exemple utiliser la simulation
pour comparer plusieurs dizaines de politiques différentes sur un même système.

Structure du programme. Dans ce cas-ci, nous avons choisi d’implanter le programme
de simulation sous forme de plusieurs modules compilés séparément. Les modules GroupeMac,
Commande et Experience implantent respectivement la description des lois de comportement
d’un groupe de machines avec un robot, différentes politiques de commande du robot, et une
expérience statistique que l’on peut faire avec ce système. Les parties “définitions” des deux
premiers modules apparaissent aux figures 11 et 12. Ces définitions servent d’interfaces
entre l’intérieur et l’extérieur de ces modules. Les objets qui apparaissent dans la définition
du module GroupMac sont utilisés par le module Experience pour initialiser le système et
la position du robot, et par le module Commande pour connâıtre l’état du système lors de
la prise des décisions. L’implantation du module GroupMac utilise aussi les procédures du
module Commande pour déterminer les actions du robot. Un exemple de module Experience
apparâıt à la figure 13. L’implantation de GroupMac est donnée à la figure 16. A noter qu’il
n’est pas nécessaire de connâıtre ou d’avoir accès à cette implantation pour écrire les modules
Experience et Commande, de la même façon qu’on n’a pas besoin d’examiner l’implantation
des modules de SIMOD pour les utiliser. De même, il n’est pas nécessaire de connâıtre
l’implantation du module Commande pour pouvoir l’utiliser. Cette implantation est donnée
à la figure 17.

Dans la définition de GroupMac, un groupe de machines avec un robot est défini comme
un pointeur sur un bloc d’information contenant toutes les caractéristiques de ce groupe.
Cela peut permettre éventuellement de traiter plusieurs groupes de machines simultanément,
chaque groupe possédant ses propres caractéristiques. (Nous y reviendrons plus loin.) La
procédure Create permet d’initialiser (créer) un groupe, tandis que Delete permet de le
détruire. Nous avons opté pour un maximum de flexibilité: le nombre de machines, la position
initiale du robot, la règle de décision, le réparateur utilisé et toutes les lois de probabilité
sont passés en paramètres et peuvent varier d’un groupe à l’autre. Tel que le module est
implanté, l’usager a directement accès à tous les champs du “RECORD” associé à un groupe
de machines. Cependant, il ne devrait pas les modifier directement (sauf en appelant la
procédure Create), mais seulement observer leurs valeurs. Pour plus de sécurité, nous
aurions pu adopter une plus grande “abstraction des donnés”, selon les principes modernes
du génie logiciel, en définissant le type Groupe comme un type opaque. La définition de
InfoGroupe n’aurait été donnée que dans l’implantation du module, et on aurait fourni
quelques procédures permettant d’observer quelques-uns de ses champs. On peut noter,
d’ailleurs, que les modules de SIMOD sont implantés selon une telle approche. Par contre,
malgré ses dangers, l’approche de l’accès direct aux champs du “RECORD” permet de
réduire la longueur du code et (un peu) les temps d’exécution. Le programmeur utilisant le
module doit alors être suffisamment discipliné pour éviter les manoeuvres dangereuses.

Le module Commande implante quelques politiques de commande du robot. La politique
BalayageComplet consiste à patrouiller la ligne de gauche à droite jusqu’à la machine N ,

DEFINITION MODULE GroupMac;
IMPORT EVENT, PROCS, STAT, RES;

CONST
Nmax = 30; (* Nombre maximum de machines permis. *)

TYPE
Num = [1..Nmax]; (* Numéro de machine. *)
ExtNum = [0..Nmax];
EnsMach = SET OF ExtNum; (* Un ensemble de machines. *)
Direction = (Gauche, Droite);
Groupe = POINTER TO InfoGroupe; (* Un groupe de machines avec un robot.*)
Politique = PROCEDURE (Groupe) : ExtNum;

(* Une politique de commande du robot. *)
(* Retourne 0 : repare mach. devant lui; *)
(* Retourne 0 < i <= N : va vers i. *)

DureeFonc = PROCEDURE () : LONGREAL; (* Type de proc. pour retourner *)
(* une durée de fonctionnement. *)

DureeRepar = PROCEDURE (VAR BOOLEAN) : LONGREAL;
(* Retourne une durée de répar. et met *)
(* Succes a TRUE ssi la répar. a réussi. *)

InfoGroupe = RECORD
N : Num; (* Nb. de machines. *)
NbOK : ExtNum; (* Nb. de machines qui fonctionnent. *)
Pos : LONGREAL; (* Position courante du robot. *)
TDepart : LONGREAL; (* Instant de la dernière mise-a-jour *)

(* de Pos, si le robot voyage. *)
Dir : Direction; (* Dernière direction de déplacement. *)
Devant : ExtNum; (* Si > 0, le robot est en face de la *)

(* mach. i, sinon il est entre 2 mach. *)
Dest : ExtNum; (* Si > 0, le robot va vers cette machine *)

(* (ou s’y trouve), si = 0, il répare *)
(* la machine en face de lui. *)

EnPanne : EnsMach; (* Ensemble des machines en panne. *)
Decision : Politique; (* Règle de décision utilisée. *)
DFonc : DureeFonc; (* Génère une durée de fonctionnement. *)
DRepar : DureeRepar; (* Une durée de réparation par le robot. *)
DReparMan : DureeRepar; (* Une durée de rép. par le réparateur. *)
Reparateur : RES.Resource; (* Le réparateur affecté a ce groupe. *)
Prod : STAT.Block; (* Bloc stat. sur la productivité. *)
Robot : PROCS.ProcessInstance;
Machine : ARRAY Num OF PROCS.ProcessInstance;
AvisArrivee : EVENT.EventNotice (* Arriv. prévue du robot à une mach.*)
END;

PROCEDURE Create
( VAR G : Groupe;
N, Devant : Num;
Decision : Politique;
DFonc : DureeFonc;
DRepar : DureeRepar;
DReparMan : DureeRepar;
Reparateur : RES.Resource

);
(* Crée le groupe de machines G, et initialise les champs du RECORD *)
(* aux valeurs passées en paramètres. *)

PROCEDURE Delete (VAR G : Groupe);
(* Détruit le groupe de machines G et récupère l’espace. *)

END GroupMac.

Figure 11: Definition du module GroupMach.

DEFINITION MODULE Commande;

FROM GroupMac IMPORT ExtNum, Groupe;

PROCEDURE BalayageComplet (G : Groupe) : ExtNum;
(* Patrouille la ligne de 1 a N, puis de N a 1, etc., en reparant *)
(* les machines brisees que l’on rencontre. *)

PROCEDURE PlusPres (G : Groupe) : ExtNum;
(* Si toutes les machines fonctionnent, on va vers le centre puis on *)
(* attend, sinon on va vers la machine brisee la plus pres. *)
(* En cas d’egalite, on s’eloigne d’abord du centre. *)

END Commande.

Figure 12: Definition du module Commande.

puis de droite à gauche jusqu’à la machine 1, et ainsi de suite. Chaque fois qu’il rencontre
une machine bloquée, le robot s’arrête pour la réparer. Sous la politique PlusPres, le robot
entreprend aussi la réparation chaque fois qu’il se trouve en face d’une machine bloquée. Si
toutes les machines fonctionnent, il se dirige vers la machine centrale (ou la plus proche des
deux machines centrales s’il y a un nombre pair de machines). Sinon, s’il n’est pas déjà en
face d’une machine bloquée, il se dirige vers la machine bloquée qui se trouve le plus près de
lui. Ainsi, si une machine se bloque pendant que le robot se déplace, il est possible que ce
dernier change de direction. En cas d’égalité (deux machines bloquées à la même distance),
le robot choisit la direction de l’extrémité la plus proche (il s’éloigne du centre de la ligne), et
si l’égalité persiste, il choisit au hasard. On pourrait évidemment ajouter d’autres politiques
à ce module, et en particulier celles suggérées dans [41]. On doit fournir une telle politique
chaque fois que l’on initialise un groupe de machines. Evidemment, il n’est pas nécessaire
que la politique fournie provienne du module Commande, elle peut provenir d’ailleurs. En
pratique, on pourrait éventuellement remplacer l’implantation du module GroupMac par de
vrais groupes de machines, commandés en utilisant directement les procédures du module
Commande.

Exemple d’une expérience. Dans l’expérience décrite dans le module de la figure 13,
on ne considère qu’un seul groupe de machines. On examine différentes politiques pour
différentes configurations, et dans chaque cas, on calcule un intervalle de confiance sur la
productivité. Pour calculer cet intervalle de confiance, on utilise ici la méthode des lots
(“batch means”) [13, 37]. On simule pour une durée donnée, que l’on subdivise en NLots

intervalles de temps égaux (de durée DureeLot). Les résultats des Rechauf premiers in-
tervalles sont éliminés, et on traite les moyennes associées aux intervalles restants comme
(NLots-Rechauf) valeurs de variables aléatoires indépendantes, distribuées selon la loi nor-
male. Le bloc statistique ProdMoy sert à recueillir ces valeurs et à calculer l’intervalle de
confiance. Il est mis-à-jour à chaque fin de lot (sauf pour ceux du réchauffement) par
l’événement FinLot, qui exécute la procédure ProcFinLot. Cette procédure réinitialise
aussi le bloc statistique Prod associé au groupe de machines, et qui sert à calculer la produc-

MODULE Experience;

IMPORT SIM, STAT, EVENT, RAND, RAND1, RES;
FROM MYINOUT IMPORT WriteLn, ReadLn, WriteString, ReadString, ReadCard,

WriteCard, OpenInput, CloseInput, OpenOutput, CloseOutput,
WriteLongReal, ReadLongReal;

FROM CHRONO IMPORT Chrono, TimeUnit, InitChrono, WriteChrono, TimeFormat;
FROM EVENT IMPORT EventType, Schedule, Attrib;
FROM RAND IMPORT RngStream, CreateStream, DeleteStream, RandU01;
FROM STAT IMPORT Update, Block, Tally, Average;
FROM GroupMac IMPORT Groupe, Create, Delete, Politique, DureeRepar;
FROM Commande IMPORT BalayageComplet, PlusPres;

VAR
N : CARDINAL; (* Nb. de machines. *)
Lot : CARDINAL; (* No. du lot courant. *)
NLots : CARDINAL; (* Nb. de lots a effectuer au total. *)
Rechauf : CARDINAL; (* Nb. de lots pour le rechauffement. *)
DureeLot : LONGREAL; (* Duree de simulation par lot. *)
ProdMoy : Block; (* Stats. pour les differents lots. *)
FinLot : EventType; (* La fin d’un lot. *)
Alpha, Lambda, ProbSucces, DureeR, DureeRM : LONGREAL;

(* Parametres des lois ci-bas. *)
Ident : ARRAY [0..63] OF CHAR;
Reparateur : RES.Resource;
G : Groupe; C : Chrono; i : CARDINAL;
GenDFonc, GenGeo : RngStream; (* 2 Flots de valeurs aleatoires *)

PROCEDURE DFonc () : LONGREAL;
(* Duree de fonctionnement entre deux pannes. *)
BEGIN RETURN RAND1.Gamma (GenDFonc, Alpha, Lambda) END DFonc;

PROCEDURE DRepMan (VAR S : BOOLEAN) : LONGREAL;
(* Duree de reparation manuelle. *)
BEGIN S := TRUE; RETURN DureeRM END DRepMan;

PROCEDURE Const (VAR S : BOOLEAN) : LONGREAL;
BEGIN S := TRUE; RETURN DureeR END Const;

PROCEDURE Geo1 (VAR S : BOOLEAN) : LONGREAL;
BEGIN
IF RandU01 (GenGeo) <= ProbSucces THEN S := TRUE ELSE S := FALSE END;
RETURN DureeR
END Geo1;

PROCEDURE Geo2 (VAR S : BOOLEAN) : LONGREAL;
BEGIN
IF RandU01 (GenGeo) <= ProbSucces THEN S := TRUE; RETURN DureeR END;
IF RandU01 (GenGeo) <= ProbSucces THEN S := TRUE ELSE S := FALSE END;
RETURN (DureeR + DureeR)
END Geo2;

PROCEDURE Rapport;
(* Produit un rapport a la fin de la simulation d’une configuration. *)
VAR

MeanOp, MeanW, RadiusW, MeanProd, RadiusP, B1, B2 : LONGREAL;
BEGIN
WriteString ("Temps d’execution (CPU) : ");
WriteChrono (C, sec); WriteLn;
WriteString ("Temps de simulation (sec) : ");
WriteLongReal (SIM.Time (), 9, 1);
STAT.ReportConfidenceInterval (ProdMoy, 0.95); WriteLn;
Delete (G); SIM.Stop
END Rapport;

Figure 13: Implantation du module Experience.

PROCEDURE ProcFinLot;
BEGIN
IF Lot > Rechauf THEN Update (ProdMoy, Average (G^.Prod)) END;
STAT.Init (G^.Prod);
IF Lot >= NLots THEN Rapport
ELSE INC (Lot); Schedule (FinLot, DureeLot, NIL) END
END ProcFinLot;

PROCEDURE EssayerConfig
( Pol : Politique; DRep : DureeRepar; Nam : ARRAY OF CHAR
);
(* Simule pour la configuration courante avec la politique Pol. *)
(* La loi de prob. de la duree de repar. par le robot est DRep. *)
BEGIN
WriteString ("CONFIGURATION: "); WriteString (Ident);
WriteString (", "); WriteString (Nam); WriteLn;
SIM.Init; STAT.Init (ProdMoy); RES.Init (Reparateur); Lot := 1;
Create (G, N, (N + 1) DIV 2, Pol, DFonc, DRep, DRepMan, Reparateur);
Schedule (FinLot, DureeLot, NIL);
InitChrono (C);
SIM.Start
END EssayerConfig;

BEGIN
RAND.CreateStream (GenDFonc, "Generateur fonction DFonc");
RAND.CreateStream (GenGeo, "Generateur fonctions Geo1 et Geo2");
EVENT.Create (FinLot, ProcFinLot, "Fin d’un lot");
STAT.Create (ProdMoy, Tally, "Prod. moyenne");
RES.Create (Reparateur, RES.Fifo, 1, "Le reparateur");
OpenInput (".dat"); OpenOutput (".res");
ReadCard (NLots); ReadCard (Rechauf); ReadLongReal (DureeLot); ReadLn;
WriteString ("Simulation d’une serie de machines");
WriteString (" entretenues par un robot.");
WriteLn; WriteLn;
WriteString ("Nombre de lots : ");
WriteCard (NLots, 8); WriteLn;
WriteString ("Nombre de lots pour rechauffement : ");
WriteCard (Rechauf, 8); WriteLn;
WriteString ("Duree de la simulation pour chaque lot : ");
WriteLongReal (DureeLot, 10, 1); WriteLn; WriteLn;
ReadString (Ident); ReadLn;
WHILE Ident [0] # "0" DO

ReadCard (N);
ReadLongReal (Alpha); ReadLongReal (Lambda);
ReadLongReal (ProbSucces); ReadLongReal (DureeR);
ReadLongReal (DureeRM); ReadLn ;
EssayerConfig (BalayageComplet, Const, "Balayage complet, rep. const.");
EssayerConfig (PlusPres, Const, "Plus pres, rep. const");
EssayerConfig (PlusPres, Geo1, "Plus pres, rep. Geo1");
EssayerConfig (PlusPres, Geo2, "Plus pres, rep. Geo2");
ReadString (Ident); ReadLn;
END;

CloseInput; CloseOutput;
RAND.DeleteStream (GenGeo);
RAND.DeleteStream (GenDFonc);
END Experience.

Figure 13: Implantation du module Experience (suite).

tivité moyenne pour chaque lot. Lorsqu’on a terminé le dernier lot, on appelle la procédure
Rapport, qui imprime les résultats, détruit le groupe G, et arrête la simulation.

Dans le programme principal, on crée le type d’événement FinLot, le bloc statistique
et le réparateur. On lit ensuite le nombre de lots, le nombre de lots à rejeter, la durée de
la simulation pour chaque lot (la durée totale sera DureeLot * NLots), puis on imprime
une entête. Ensuite, dans le fichier de données, il y a deux lignes de données pour chaque
configuration que l’on doit traiter. La première ligne contient une châıne de caractères
décrivant cette configuration, et la seconde contient le nombre de machines et les valeurs
des paramètres à utiliser dans les lois de probabilité définies au début du module. Après
la dernière configuration, le fichier contient une ligne dont le premier caractère est un zéro.
La boucle WHILE du programme principal lit les configurations et les traite. Un appel à
EssayerConfig effectue une simulation en utilisant les paramètres que l’on vient de lire. La
politique à utiliser et la loi de probabilité de la durée de réparation par le robot sont passées
en paramètres. Les autres lois sont fixées, mais on a lu les valeurs de leurs paramètres. A
noter que cette façon de faire est tout à fait arbitraire et sert ici à illustrer la flexibilité
dont on dispose. On suppose ici que l’on ne veut varier que la politique et la loi DRep, mais
pas les autres lois. Dans EssayerConfig, on réinitialise la simulation, le bloc statistique,
le réparateur et le chronomètre. On crée et initialise ensuite un groupe de machines, et on
prévoit un événement qui marquera la fin du premier lot. A noter que comme chaque appel
à EssayerConfig crée un nouveau groupe de machines, il est important de détruire G à la fin
de la simulation. Sinon, tous les groupes de machines que l’on a crée vont continuer d’exister
et d’être simulés indéfiniment. Ici, toutes les configurations sont simulées avec des valeurs
aléatoires différentes. Si on voulait calculer un intervalle de confiance pour la différence entre
deux configurations, il vaudrait probablement mieux utiliser des valeurs aléatoires communes,
comme dans l’exemple de la section 3.3.

Les cinq premières procédures du module implantent les lois de probabilité utilisées. La
durée de fonctionnement d’une machine suit la loi Gamma de paramètres Alpha et Lambda.
Le temps requis pour compléter manuellement une réparation est constant et égal à DureeRM.
Les trois autres procédures servent à générer des durées de réparation par le robot. Dans
le cas de Const, le réparation réussit toujours et sa durée est constante et égale à DureeR.
Dans le cas de Geo1, elle a la même durée, mais ne réussit qu’avec une probabilité égale à
ProbSucces. Geo2 est semblable à Geo1, sauf que si la réparation échoue au premier essai,
le robot tente un second essai (avec la même probabilité de réussite) avant de faire venir le
réparateur. On pourrait évidemment ajouter d’autres procédures, implantant d’autres lois
de probabilité. Selon leur nombre ou leur taille, on pourrait aussi décider de les placer dans
un module à part, tout comme on l’a fait pour les politiques de commande.

Figure 14: Un fichier de données pour le module Experience.

Simulation d’une serie de machines entretenues par un robot.

Nombre de lots : 10
Nombre de lots pour rechauffement : 1
Duree de la simulation pour chaque lot : 1000.0

CONFIGURATION: Config1, Balayage complet, rep. const.
Temps d’execution (CPU) : 0.17 seconds
Temps de simulation (sec) : 10000.0
REPORT ON STATISTIC (Tally) : Prod. moyenne

min max average standard dev. nb. Obs.
8.071 8.836 8.547 0.295 9

95% confidence interval for mean : ( 8.321, 8.774 )

CONFIGURATION: Config1, Plus pres, rep. const
Temps d’execution (CPU) : 0.60 seconds
Temps de simulation (sec) : 10000.0
REPORT ON STATISTIC (Tally) : Prod. moyenne

min max average standard dev. nb. Obs.
8.848 9.109 9.005 0.077 9

95% confidence interval for mean : ( 8.946, 9.065 )

CONFIGURATION: Config1, Plus pres, rep. Geo1
Temps d’execution (CPU) : 0.60 seconds
Temps de simulation (sec) : 10000.0
REPORT ON STATISTIC (Tally) : Prod. moyenne

min max average standard dev. nb. Obs.
8.016 8.849 8.517 0.268 9

95% confidence interval for mean : ( 8.311, 8.723 )

CONFIGURATION: Config1, Plus pres, rep. Geo2
Temps d’execution (CPU) : 0.57 seconds
Temps de simulation (sec) : 10000.0
REPORT ON STATISTIC (Tally) : Prod. moyenne

min max average standard dev. nb. Obs.
8.083 8.894 8.580 0.268 9

95% confidence interval for mean : ( 8.374, 8.786 )

Figure 15: Les résultats produits à partir du fichier de la figure 14.

La figure 14 montre un exemple de fichier de données, et la figure 15 donne les résultats
imprimés par le programme en utilisant ce fichier de données. Dans cet exemple, il n’y a
qu’une seule configuration. Habituellement, en pratique, on imprimera un rapport plus com-
plet, indiquant par exemple, pour chaque configuration, le nombre de machines, la politique
utilisée, les paramètres des lois, etc.

Implantation de GroupMac. Dans l’implantation du module GroupMac, on voit que pour
chaque groupe de machines, le robot et chacune des machines correspondent à des proces-
sus. Ces derniers sont démarrés par la procédure Create, lors de la création du groupe,
et exécutent des boucles infinies. Ils ne se terminent donc que lorsqu’on détruit le groupe.
Ces processus sont prévus avec des paramètres dont on récupère la valeur au tout début de
leur exécution. Le paramètre d’un robot est le nom de son groupe, tandis que le paramètre
d’une machine est un pointeur sur un enregistrement contenant le nom de son groupe et son
numéro dans le groupe.

Chaque machine exécute une boucle infinie. Elle fonctionne jusqu’à ce qu’elle tombe en
panne puis, si le robot n’est pas occupé à réparer une machine (il se trouve alors dans l’état
Suspended), elle le réactive, afin qu’il prenne une nouvelle décision compte tenu du nouvel
état du système. Ensuite, la machine se “suspend” et attend d’être réactivée par le robot, à
la fin de la réparation, avant de recommencer la boucle.

Le robot exécute lui aussi une boucle infinie. Au début de la boucle, il prend une décision,
selon l’état du groupe de machines et de la politique adoptée. S’il décide de réparer (il ne
peut le faire que s’il se trouve exactement en face d’une machine brisée), il génère une durée
de réparation et attend pendant cette durée. Si la réparation n’a pas réussi, il fait ensuite
appel au réparateur, génère une durée de réparation manuelle, et utilise le réparateur pour
cette durée. Il réactive ensuite la machine qu’il vient de réparer. Le lecteur aura sans doute

IMPLEMENTATION MODULE GroupMac;

IMPORT SIM, STAT, EVENT, PROCS;
FROM SYSTEM IMPORT ADDRESS;
FROM Storage IMPORT ALLOCATE, DEALLOCATE;
FROM MYINOUT IMPORT WriteString, WriteLn;
FROM STAT IMPORT Update;
FROM RES IMPORT Resource, Request, Release;
FROM EVENT IMPORT EventType, EventNotice, ScheduleNotice, CancelNotice;
FROM PROCS IMPORT ProcessType, ScheduleNamed, Delay, Resume, Suspend,

CurrentProcess, InstanceState, ProcessState, Terminate;
FROM NUM IMPORT IToLR;

TYPE
ParamMach = POINTER TO RECORD (* Parametres d’une machine. *)

Gr : Groupe; (* Son Groupe. *)
No : Num (* Son numero. *)
END;

Figure 16: Implantation du module GroupMach.

VAR
UnRobot, UneMachine : ProcessType;
ArriveeRobot : EventType;

PROCEDURE Erreur (i : CARDINAL);
BEGIN
WriteString ("DECISION ILLEGALE DU ROBOT:");
CASE i OF

1 : WriteString (" Reparation sans etre en face d’une mach. fautive.")
| 2 : WriteString (" Destination < 1.")
| 3 : WriteString (" Destination > Nb. de mach.")
END;

WriteLn; SIM.Stop; Terminate
END Erreur;

PROCEDURE Create
( VAR G : Groupe; n, dev : Num; d : Politique;

df : DureeFonc; dr : DureeRepar; drm : DureeRepar; r : Resource
);
VAR

i : Num; Par : ParamMach;
BEGIN
NEW (G);
WITH G^ DO

EnPanne := EnsMach {};
FOR i := 1 TO n DO

NEW (Par); Par^.Gr := G; Par^.No := i;
ScheduleNamed (UneMachine, 0.0, Par, Machine [i]);
END;

N := n; NbOK := n; Devant := dev; Pos := IToLR (Devant);
Dir := Droite; Decision := d; DFonc := df; DRepar := dr;
DReparMan := drm; Reparateur := r;
ScheduleNamed (UnRobot, 0.0, G, Robot);
STAT.Create (Prod, STAT.Accumulate, "Taux de production");
STAT.Init (Prod); Update (Prod, VAL(LONGREAL,N)) ;
END

END Create;

PROCEDURE Delete (VAR G : Groupe);
(* Recupere tout l’espace occupe par G. *)
VAR

i : Num; Par : ParamMach;
BEGIN
WITH G^ DO

FOR i := 1 TO N DO
Par := PROCS.InstanceAttrib (Machine [i]);
DISPOSE (Par); PROCS.Kill (Machine [i])
END;

STAT.Delete (Prod); PROCS.Kill (Robot)
END;

DISPOSE (G)
END Delete;

PROCEDURE ProcArrivee;
(* Evenement : Le robot atteint sa destination. *)
VAR

G : Groupe;
BEGIN
G := EVENT.Attrib ();
WITH G^ DO Devant := Dest; Resume (Robot) END;
END ProcArrivee;

Figure 16: Implantation du module GroupMach (suite).

PROCEDURE ProcMachine;
(* Decrit le comportement d’une machine. *)
VAR

No : Num; (* Son numero. *)
Par : ParamMach; (* Pointeur sur ses parametres. *)

BEGIN
Par := PROCS.Attrib (); No := Par^.No;
WITH Par^.Gr^ DO (* Groupe auquel elle appartient. *)

WHILE TRUE DO
Delay (DFonc ()); (* Fin du delai: La mach. tombe en panne. *)
DEC (NbOK); INCL (EnPanne, No);
Update (Prod, VAL(LONGREAL,NbOK));
IF InstanceState (Robot) = Suspended THEN Resume (Robot) END;
Suspend
END (* Reactivation: fin de la reparation. *)

END
END ProcMachine;

PROCEDURE ProcRobot;
VAR

S : BOOLEAN; G : Groupe;
BEGIN
G := PROCS.Attrib (); (* Groupe auquel il appartient. *)
WITH G^ DO

WHILE TRUE DO
Dest := Decision (G);
IF Dest = 0 THEN (* Repare la machine devant lui. *)

IF (Devant = 0) OR (NOT (Devant IN EnPanne)) THEN Erreur (1) END;
Delay (DRepar (S));
IF (NOT S) THEN (* La reparation a echoue. *)

Request (1, Reparateur);
Delay (DReparMan (S));
Release (1, Reparateur)
END;

Resume (Machine [Devant]);
(* Doit s’executer avant la prochaine prise de decision: *)
INC (NbOK); EXCL (EnPanne, Devant);
Update (Prod, VAL(LONGREAL,NbOK))

ELSIF Dest = Devant THEN Suspend (* Le robot attend. *)
ELSE (* Le robot se deplace. *)

IF Dest > N THEN Erreur (3) END;
IF VAL(LONGREAL,Dest) > Pos THEN Dir := Droite
ELSE Dir := Gauche END;
Devant := 0; TDepart := SIM.Time ();
ScheduleNotice (ArriveeRobot, ABS (Pos - VAL(LONGREAL,Dest)),

G, AvisArrivee);
Suspend;
IF Devant = Dest THEN (* Le robot arrive a destination. *)

Pos := VAL(LONGREAL,Dest)
ELSE (* Une machine tombe en panne. *)

IF Dir = Droite THEN Pos := Pos + SIM.Time () - TDepart
ELSE Pos := Pos - SIM.Time () + TDepart END;
CancelNotice (AvisArrivee)
END

END
END

END
END ProcRobot;

BEGIN
EVENT.Create (ArriveeRobot, ProcArrivee, "Arrivee du robot");
PROCS.Create (UneMachine, ProcMachine, 2*4096, "Une machine");
PROCS.Create (UnRobot, ProcRobot, 2*4096, "Un robot")
END GroupMac.

Figure 16: Implantation du module GroupMach (suite).

IMPLEMENTATION MODULE Commande;

FROM RAND IMPORT RngStream, CreateStream, RandU01;
FROM GroupMac IMPORT Num, ExtNum, EnsMach, Direction, Groupe;
FROM MYINOUT IMPORT WriteString, WriteLn, WriteCard ;
FROM NUM IMPORT IToLR;

PROCEDURE BalayageComplet (G : Groupe) : ExtNum;
VAR

m : ExtNum;
BEGIN
WITH G^ DO

IF (Devant IN EnPanne) THEN RETURN 0 END; (* Repare. *)
IF N = 1 THEN RETURN 1 END;
IF Devant = 1 THEN Dir := Droite END;
IF Devant = N THEN Dir := Gauche END;
IF Devant > 0 THEN (* En face d’une machine. *)

m := Devant
ELSE (* Entre deux machines. *)

m := TRUNC (Pos);
IF ((Dir = Droite) AND (m < N)) OR (m < 1) THEN INC (m) END
END;

IF Dir = Droite THEN
WHILE ((m < N) AND (NOT (m IN EnPanne))) DO INC (m) END

ELSE
WHILE ((m > 1) AND (NOT (m IN EnPanne))) DO DEC (m) END
END;

RETURN m
END

END BalayageComplet;

PROCEDURE PlusPres (G : Groupe) : ExtNum;
VAR

i, j : ExtNum; D, Dmin : LONGREAL;
g : RngStream;

BEGIN
CreateStream(g, "Generateur aleatoire");
WITH G^ DO

IF (Devant IN EnPanne) THEN RETURN 0 END; (* Repare. *)
IF N = 1 THEN RETURN 1 END;
IF (EnPanne = EnsMach{}) THEN (* Pas de panne, va vers le centre. *)

IF (Pos < IToLR (N + 1) / 2.0) THEN RETURN (N + 1) DIV 2
ELSE RETURN (N + 2) DIV 2 END
END;

(* Trouve la mach. en panne la plus proche: la mach. j. *)
Dmin := IToLR (N);
FOR i := 1 TO N DO

IF i IN EnPanne THEN
D := ABS (IToLR (i) - Pos);
IF D < Dmin THEN j := i; Dmin := D END
END

END;
(* Verifie les egalites (si devant une machine). *)
IF Devant > 0 THEN

i := ABS(VAL(INTEGER,2 * Devant) - VAL(INTEGER,j));
(* La mach. i est a la meme dist. que j. *)
IF ((1 <= i) AND (i <= N) AND (i IN EnPanne)) THEN

IF (((i < j) AND (i - 1 < N - j)) OR ((i > j) AND (N - i < j - 1))
(* i est plus pres d’une extremite. *)

OR ((i - 1 = N - j) AND (RandU01 (g) < 0.5))) THEN
RETURN i
END

END
END

END;
RETURN j ;
END PlusPres;

END Commande.

Figure 17: Implantation du module Commande.

remarqué que l’ensemble des machines en panne, leur nombre, et le bloc statistique sur la
productivité, sont mis-à-jour par le processus Machine lors d’une panne, mais par le processus
Robot à la fin de la réparation. La raison est qu’à la fin d’une réparation, cette mise-à-jour
doit se faire avant que le robot ne prenne sa prochaine décision. Comme la machine ne prend
pas le contrôle immédiatement dès que le robot appelle Resume, mais seulement après qu’il
ait appelé Suspend, le robot prendra sa prochaine décision avant de passer le contrôle à la
machine. Une autre approche consisterait à forcer le robot à passer le contrôle à la machine
avant la prochaine prise de décision, par exemple en plaçant l’instruction Delay (0.0) juste
après l’appel à Resume.

Dans le cas où le robot est exactement en face d’une machine et décide d’y rester, il
se suspend et attend d’être réactivé par la prochaine machine qui tombera en panne. S’il
décide de se déplacer, il prévoit un événement qui correspond à son arrivée à destination, et
se suspend. Si aucune panne ne survient d’ici là, cet événement réactive le robot. Sinon, ce
dernier est réactivé par une machine qui tombe en panne, et annule l’événement qui prévoyait
sa réactivation. Dans les deux cas, il met à jour sa position et reprend une nouvelle décision.

Discussion. Souvent, comme le laisse entrevoir l’implantation du module Commande, la
politique de commande d’un système peut s’exprimer assez facilement sous forme de règles.
Dans certains cas, il pourrait être intéressant d’implanter cette politique sous forme d’une
base de règles, par exemple sous forme d’un programme PROLOG. Le système Modula-
Prolog décrit dans [46] permet d’utiliser Modula-2 et PROLOG simultanément, dans un
même programme, et pourrait donc permettre de le faire.

A remarquer que nous ne présentons pas ici le programme de simulation le plus efficace
pour estimer la productivité d’un groupe de machines. Pour diminuer les temps d’exécution,
on pourrait d’abord facilement précalculer certaines expressions. Ensuite, l’adoption de
techniques de réduction de la variance, telles que celles utilisées dans [41], permettrait de
réduire par un facteur de 5 à 10 le temps de simulation requis pour un intervalle de confiance
de rayon donné. Enfin, si on utilisait une approche uniquement par événements au lieu
d’utiliser des processus, on réduirait encore les temps d’exécution de façon substantielle.
Les événements correspondraient alors aux instants où le robot arrive à destination, où une
machine tombe en panne, et où une réparation se termine. A noter que dans ce cas-ci, le
code ne serait pas beaucoup plus long ni plus compliqué.

Un système composé de plusieurs groupes de machines. Pour effectuer l’expérience
décrite à la figure 13, il n’était pas vraiment nécessaire de définir un module GroupMac aussi
élaboré. Mais l’intérêt de la définition adoptée est qu’elle permet de créer autant de groupes
de machines que l’on veut, et de les faire évoluer en même temps. De plus, ces groupes
de machines ne sont pas nécessairement indépendants: ils peuvent par exemple partager un
ou plusieurs réparateur(s). Ainsi, notre programme de simulation d’un groupe de machines
peut servir de “composante” d’un programme de simulation d’un plus gros système, qui
peut lui-même servir de composante d’un programme encore plus gros, etc. Le programme
de la figure 18 illustre cette idée. On crée quatre groupes de machines, dont trois groupes

identiques qui partagent un même réparateur, et un quatrième groupe qui possède son propre
réparateur. On simule durant 2000 unités de temps, puis on fait imprimer un rapport sur
chacun des groupes de machines.

MODULE Systeme;

IMPORT SIM, STAT, EVENT, RAND, RAND1, RES;
FROM EVENT IMPORT EventType, Schedule;
FROM GroupMac IMPORT Groupe, Create;
FROM Commande IMPORT BalayageComplet, PlusPres;

VAR
FinSim : EventType; (* La fin de la simulation. *)
Guy, Bob : RES.Resource;
G : ARRAY [1..4] OF Groupe;
i : CARDINAL;
GenDFonc,
GenGeo : RAND.RngStream;

PROCEDURE DFonc () : LONGREAL;
BEGIN RETURN RAND1.Expon (GenDFonc,100.0) END DFonc;

PROCEDURE DRepMan (VAR S : BOOLEAN) : LONGREAL;
BEGIN S := TRUE; RETURN 5.0 END DRepMan;

PROCEDURE Geo2 (VAR S : BOOLEAN) : LONGREAL;
BEGIN
IF RAND.RandU01 (GenGeo) <= 0.8 THEN S := TRUE; RETURN 5.0 END;
IF RAND.RandU01 (GenGeo) <= 0.8 THEN S := TRUE ELSE S := FALSE END;
RETURN 10.0
END Geo2;

PROCEDURE ProcFinSim;
VAR

i: CARDINAL;
BEGIN
FOR i := 1 TO 4 DO STAT.Report (G [i]^.Prod) END;
SIM.Stop
END ProcFinSim;

BEGIN
RAND.CreateStream (GenDFonc, "Generateur pour la fonction DFonc");
RAND.CreateStream (GenGeo, "Generateur pour la fonction Geo2");
EVENT.Create (FinSim, ProcFinSim, "Fin de la simulation");
RES.Create (Guy, RES.Fifo, 1, "Le reparateur Guy");
RES.Create (Bob, RES.Fifo, 1, "Le reparateur Bob");
FOR i := 1 TO 3 DO

Create (G [i], 10, 5, PlusPres, DFonc, Geo2, DRepMan, Guy)
END;

Create (G [4], 15, 1, BalayageComplet, DFonc, Geo2, DRepMan, Bob);
Schedule (FinSim, 2000.0, NIL);
SIM.Start;
RAND.DeleteStream (GenDFonc);
RAND.DeleteStream (GenGeo);
END Systeme.

Figure 18: Un système comprenant plusieurs groupes de machines.



42 4 CONCLUSION

4 Conclusion

Le langage Modula-2 possède plusieurs avantages pour la simulation à événements discrets
(e.g., il est relativement simple, permet la vérification des types et la conception modulaire,
sépare la définition de l’implantation, offre des variables de type PROCEDURE, et supporte les
co-routines), mais il a aussi plusieurs faiblesses. A ce sujet, on peut lire la critique de Moffat
[44] et la section 5 de [40].

Parmi les aspects qui nous ont agacés dans le développement et l’utilisation de SIMOD,
on peut mentionner d’abord l’absence de portabilité. Les programmes Modula-2 doivent faire
appel largement à des librairies, même pour les procédures d’entrée-sortie. Jusqu’à la fin des
années 90, il n’y avait pas de standard pour les librairies, qui variaient beaucoup d’une im-
plantation à l’autre. Modula-2 n’est pas non plus un langage véritablement “orienté-objet”.
En particulier, il ne supporte pas la notion d’extension de type (définition d’un sous-type
qui hérite des propriétés du type parent, et qui peut posséder des propriétés additionnelles).
On peut mentionner aussi le traitement des entrées/sorties (qui ne fait pas vraiment partie
du langage), l’impossibilité d’avoir des procédures qui ont des paramètres optionnels ou qui
varient en nombre, l’impossibilité de surcharger les identificateurs (plusieurs procédures qui
ont le même nom et qui se distinguent par les types de leurs paramètres), l’impossibilité
d’avoir des co-routines dont les procédures associées ont des paramètres, l’obligation de cal-
culer la taille de l’espace mémoire alloué à chaque co-routine, le fait que les pointeurs ne
sont pas initialisés à NIL (ce qui nous empêche de faire vérifier par SIMOD si un objet a déjà
été créé ou pas), le traitement des exceptions, le traitement des châınes de caractères, etc.
Bien sûr, ajouter au langage toutes ces facilités augmenterait la difficulté d’implantation et
la complexité des compilateurs.

Modula-2 demeure malgré tout efficace et pratique pour implanter des logiciels de sim-
ulation. Les principaux points forts de SIMOD sont (a) sa grande flexibilité (accès à un
langage de base tout-usage, possibilité d’appeler des routines dans d’autres langages), (b)
l’efficacité du code produit, (c) l’utilisation d’une syntaxe qui donne lieu à des programmes
lisibles, et (d) la puissance des outils disponibles. Comme nous l’avons vu dans les exem-
ples de la section 3, SIMOD est au moins aussi puissant, pour la simulation à événements
discrets, que les langages spécialisés tels que GPSS, SLAM, SIMSCRIPT II.5, et SIMAN.
Il est aussi relativement facile d’y ajouter d’autres modules au besoin. On pourrait par ex-
emple y ajouter des outils de haut niveau pour l’animation graphique, etc. On peut utiliser
SIMOD conjointement avec des progiciels existants pour la gestion de bases de données, les
statistiques, le graphisme par ordinateur, etc.

Nous avons vu, à la section 3.7, comment SIMOD permet et encourage l’implantation
modulaire d’un programme de simulation, et comment cette modularité facilite la modi-
fication ultérieure d’une partie du modèle, ou des règles de commande, ou encore du plan
d’expérience, etc. L’avantage de cette approche sera d’autant plus marqué que le modèle sera
complexe et/ou de grande taille. Ainsi, SIMOD n’est pas seulement un outil pédagogique ser-
vant à traiter de petits exemples académiques, mais on peut aussi l’utiliser pour la simulation
des grands systèmes rencontrés en pratique.

ANNEXES

A. Définition fonctionnelle

des modules de SIMOD

RAND

Ce module fournit les générateurs (pseudo)aléatoires uniformes pour la simulation. On
peut l’utiliser directement et il sert aussi de base aux autres modules (RAND1, RAND2,

RANDLIB) qui servent à générer des valeurs aléatoires suivant d’autres lois de probabilité.

Le générateur de base utilisé est celui proposé par L’Ecuyer [39], dont la période est
d’environ 2191. Son état peut être représenté par 6 entiers de 32 bits. Le cycle du générateur
est découpé en sous-suites disjointes et contiguës (appelées “streams”) de longueur 2127. Cha-
cune de ces sous-suites correspond à un générateur virtuel (“RngStream”). A tout moment
durant l’exécution du programme, l’état de chaque générateur virtuel est un vecteur de 6
entiers positifs et correspond à une position dans la sous-suite associée. Chaque fois que ce
générateur produit une valeur, son état avance d’une position. Chaque sous-suite de 2127

valeurs est aussi partitionnée en 251 blocs (ou sous-sous-suites) de longueur 276.

L’état initial du générateur de base, que l’on peut changer en appelant SetPackageSeed,
est le point de départ de la première sous-suite créée. Chaque fois que l’on crée une nouvelle
sous-suite (un object de type RngStream), son point de départ est calculé automatiquement,
2127 positions plus loin que la position de départ de la sous-suite précédente.

On dénote par Cg l’état courant de la sous-suite (i.e., du générateur virtuel) g, Ig son
état initial, Bg létat au début du bloc courant, et Ng l’état au début du prochain bloc. Le
diagramme suivant donne l’exemple d’une sous-suite dont l’état courant est à la sixième
position du troisième bloc, i.e., 2 × 251 + 5 positions plus loin que son état initial Ig et 5
positions plus loin que Bg.

Cg
⇓. . . . . . . . . . . . . . . . . .

Ig Bg Ng

La procédure ResetStream permet de remettre une sous-suite soit à son état initial
(Cg ← Ig et Bg ← Ig), soit au début du bloc dans lequel on se trouve (Cg ← Bg), ou soit
au début du prochain bloc (Cg ← Ng and Bg ← Ng). La procédure GetState retourne létat
courant d’une sous-suite. On peut changer l’état d’une sous-suite sans modifier l’état des
autres en appelant SetSeed ou AdvanceState. Toutefois, après avoir appelé SetSeed pour
une sous-suite donnée, les états initiaux des différentes sous-suites ne seront plus espacés
de 2127 valeurs. Il est donc recommandé d’utiliser cette procédure le moins possible. La
procédure ResetStream suffit pour la plupart des applications. L’exemple de la section 3.3
illustre l’utilité pratique de cette procédure.

Les procédures RandU01 et RandInt génèrent des valeurs (pseudo-)aléatoires uniformes.
Chaque sous-suite peut produire, si désiré, des nombres aléatoires antithetiques à ceux nor-
malement produits, soit 1 − U au lieu de U , en appelant SetAntithetic pour changer la

valeur d’un interrupteur. Par ailleurs, on peut utiliser la procédure DoubleGenerator pour
augmenter le nombre de bits de précision fournis lors des appels à RandU01.

DEFINITION MODULE RAND;

TYPE

SeedType = (StartStream, StartBlock, NextBlock);

RngStream;

Un générateur virtuel (i.e., une sous-suite de valeurs aléatoires).

PROCEDURE CreateStream
( VAR g : RngStream;

Name : ARRAY OF CHAR
);

Creates a new stream g. This procedure reserves space to keep the information relative to g,
initializes its seed Ig, sets Bg and Cg equal to Ig, sets its antithetic switch to 0. The seed Ig
is equal to the initial seed of the package given by SetPackageSeed if this is the first stream
created, otherwise it is Z steps ahead of that of the most recently created stream.

PROCEDURE DeleteStream
( VAR g : RngStream
);

Deletes the stream g and recover its memory, if it has been created previously by CreateStream.
Otherwise does nothing.

PROCEDURE ResetStream
( g : RngStream;

Where : SeedType
);

Reinitialize the current state of stream g, according to the value of where:
If where = StartStream, Cg and Bg are set to Ig;
if where = StartBlock, Cg is set to Bg;
if where = NextBlock, Ng is computed, and Cg and Bg are set to Ng.

PROCEDURE DoubleGenerator
( g : RngStream ;

D : BOOLEAN
);

The RngStream may be set to return a more precise number when calling RandU01. If so,
every call advances the state by 2 steps and returns a number with 53 bits of precision. In
the non-antithetic case, the instruction “x = RandU01(g)” when generator is toggled to give
double values is equivalent to “x = (RandU01(g) + RandU01(g) * fact) % 1.0” where the
constant fact is equal to 2−24. Note that this also applies when calling RandInt. By default
the generator is set to return a value with 32 bits of precision. When this procedure is called
with D = TRUE, g will always return values with more precision until it is called again with D =

FALSE.

PROCEDURE SetAntithetic
( g : RngStream;

A : BOOLEAN
);

When this procedure is called with A = TRUE, the stream g starts generating antithetic variates,
i.e., 1− U instead of U , until the procedure is called again with A = FALSE.

PROCEDURE SetPackageSeed
( Seed : ARRAY OF LONGCARD
);

Sets the initial seed of the package to the six integers in the vector seed. This will be the
seed (initial state) of the first stream. If this procedure is not called, the default initial seed
is (12345, 12345, 12345, 12345, 12345, 12345). If it is called, the first 3 values of the seed must
all be less than m1 = 4294967087, and not all 0; and the last 3 values must all be less than
m2 = 4294944443, and not all 0.

PROCEDURE SetSeed
( g : RngStream;

Seed : ARRAY OF LONGCARD
);

Sets the initial seed Ig of stream g to the vector seed. This vector must satisfy the same
conditions as in SetPackageSeed. The stream g is then reset to this initial seed. The states
and seeds of the other streams are not modified. As a result, after calling this procedure, the
initial seeds of the streams are no longer spaced Z values apart. We discourage the use of this
procedure, proper use of ResetStream is preferable.

PROCEDURE AdvanceState
( g : RngStream;

e, c : LONGINT
);

Advances the state of stream g by k values, without modifying the states of other streams (as
in SetSeed), nor the values of Bg and Ig associated with g. If e > 0, then k = 2e + c; if e < 0,
then k = −2e + c; and if e = 0, then k = c. Note: c is allowed to take negative values.

PROCEDURE GetState
( g : RngStream;

VAR Seed : ARRAY OF LONGCARD
);

Returns the current state Cg of stream g. This is convenient if we want to save the state for
subsequent use.

PROCEDURE WriteState
( g : RngStream
);

Prints (to standard output) the current state Cg of stream g.

PROCEDURE RandU01
( g : RngStream
) : LONGREAL;

Returns a (pseudo)random number from the uniform distribution over the interval (0, 1), using
stream g, after advancing the state by one step. The returned number has 32 bits of precision.

PROCEDURE RandInt
( g : RngStream;

i, j : LONGINT
) : LONGINT;

Returns a (pseudo)random number from the discrete uniform distribution over the integers
{i, i+ 1, . . . , j}, using stream g, after advancing the state by one step.

END RAND.



48

RAND1

Ce module sert à générer des valeurs aléatoires selon différentes lois de probabilité. Des
fonctions sont disponibles pour générer des valeurs selon les lois uniforme discrète, uniforme
continue, exponentielle, Weibull, gamma, bêta, normale et Student. Elles utilisent toutes la
méthode d’inversion [13, 37], sauf pour les lois gamma et bêta. L’usager peut aussi définir sa
propre loi discrète, en donnant l’ensemble des valeurs qui peuvent être prises, et la fonction
de répartition. Pour faire cela, on déclare une variable de type DiscreteDist, on définit
la distribution en question en appelant CreateDiscreteDist, puis on utilise la fonction
Discrete pour générer des valeurs.

DEFINITION MODULE RAND1;

FROM RAND IMPORT RngStream;

TYPE
DiscreteDist;

Une loi de probabilité discrète, que l’usager définira lui-même.

PROCEDURE DiscreteUniform
( g : RngStream;

N1, N2 : LONGINT
): LONGINT;

Fournit une valeur aléatoire suivant la loi uniforme discrète sur l’ensemble des entiers {N1, . . . , N2},
en utilisant le générateur g. On doit avoir N2 ≥ N1 et la différence N1 - N2 ne doit pas dépasser
2147483561.

PROCEDURE CreateDiscreteDist
( VAR D : DiscreteDist;

n : CARDINAL;
VAR V, F : ARRAY OF LONGREAL

);

Permet de définir la loi de probabilité discrète D. Le nombre de valeurs qui peuvent être prises
est n, ces valeurs (distinctes) doivent être placées dans le tableau V en ordre croissant (dans
V[0..n-1]), et le tableau F doit contenir la valeur de la fonction de répartition à chacune de
ces valeurs. Ainsi, F[i] est la probabilité que la valeur de la variable aléatoire soit inférieure
ou égale à V[i]. Les valeurs placées dans F[0..n-1] doivent être entre 0.0 et 1.0, en ordre
croissant, et on doit avoir F[n-1]= 1.0.

PROCEDURE Discrete
( g : RngStream;

D : DiscreteDist
): LONGREAL;

Fournit une valeur aléatoire suivant la loi discrète D, définie précédemment par l’usager, en
utilisant le générateur g.

PROCEDURE Uniform
( g : RngStream;

A, B : LONGREAL
) : LONGREAL;

Fournit une valeur aléatoire suivant la loi uniforme continue de paramètres A et B, en utilisant
le générateur g. La valeur produite est comprise strictement entre A et B.

PROCEDURE Normal
( g : RngStream;

Mean, StdDeviation : LONGREAL
): LONGREAL;

Fournit une valeur aléatoire suivant la loi normale de moyenne µ = Mean et d’écart-type σ =
StdDeviation. Utilise l’inversion via la fonction InvNormalDist.

PROCEDURE InvNormalDist
( U : LONGREAL
) : LONGREAL;

Retourne y = F−1(u), où F est la fonction de répartition de la loi de normale de moyenne 0 et
de variance 1. Utilise une approximation rationnelle donnant au moins 7 décimales de précision
pour 10−20 <U< 1− 10−20.

PROCEDURE Student
( g : RngStream;

DegFreedom : LONGINT
): LONGREAL;

Fournit une valeur aléatoire suivant la loi de Student avec DegFreedom degrés de liberté. Utilise
l’inversion et une approximation rationnelle.

PROCEDURE InvStudentDist
( n : LONGINT;

u : LONGREAL
) : LONGREAL;

Retourne y = F−1(u), où F est la fonction de répartition de la loi de Student à n degrés de
liberté.

PROCEDURE Expon
( g : RngStream;

Mean : LONGREAL
) : LONGREAL;

Fournit une valeur aléatoire suivant la loi exponentielle de moyenne µ = Mean, c’est-à-dire, dont
la fonction de répartition est

F (x) = 1− e−x/µ, x > 0.

Appelle la fonction Uniform01 et utilise l’inversion.

PROCEDURE Weibull
( g : RngStream;

Alpha, Lambda : LONGREAL
): LONGREAL;

Fournit une valeur aléatoire suivant la loi Weibull de paramètres α = Alpha et λ = Lambda,
c’est-à-dire dont la fonction de répartition est

F (x) = 1− e−λxα
, x > 0.

On doit avoir α > 0 et λ > 0. La moyenne est égale à

Γ(1 + 1/α)

λ1/α
, α > 0.

Appelle la fonction Uniform01 et utilise l’inversion.

PROCEDURE Gamma
( g : RngStream;

Alpha, Lambda : LONGREAL
): LONGREAL;

Fournit une valeur aléatoire suivant la loi Gamma de paramètres α = Alpha et λ = Lambda,
c’est-à-dire dont la fonction de densité est

f(x) =
λαxα−1e−λx

Γ(α)
, x > 0.

On doit avoir α > 0 et λ > 0. La moyenne et la variance sont α/λ et α/λ2 respectivement.
Utilise les fonctions RGS et RGKM3 des figures L.12 et L.13 de [13]. Il s’agit d’une méthode
d’acceptation-rejet, de sorte que le nombre d’appels à la fonction Uniform01 est variable.

PROCEDURE Beta
( g : RngStream;

a, b : LONGREAL
): LONGREAL;

Fournit une valeur aléatoire suivant la loi Beta de paramètres α = a et β = b, c’est-à-dire dont
la fonction de densité est

f(x) =
Γ(α+ β)xα−1(1− x)β−1

Γ(α)Γ(β)
, 0 < x < 1.

On doit avoir α > 0 et β > 0. Utilise l’algorithme de Cheng donné à la figure L.2 de [13]. Il s’agit
d’une méthode d’acceptation-rejet, de sorte que le nombre d’appels à la fonction Uniform01 est
variable.

END RAND1.

This module is simply an interface to the module C-RAND designed by Stadlober and
Niederl [56], written in C, which is in turn an updated and improved version of a C-package
developed by Kremer and Stadlober [35, 55].

The package of [56] has been adapted to provide compatibility with the object-based
approach of the RAND module, which provides the underlying uniform generator for RAND2,
but it has not been altered otherwise. The description of what the procedures do has been
taken directly from the documentation [56], with minor ajustments.

Note: The procedures implemented in C-RAND have undefined behavior when the input
parameters are out of range; they may either halt, stall or provide wrong return values.

DEFINITION MODULE ["C"] RAND2;

FROM RAND IMPORT RngStream;

PROCEDURE bbbc
( gen : RngStream;

a, b : LONGREAL
) : LONGREAL;

Beta Distribution - Acceptance Rejection with log-logistic hats for the Beta Prime Distribution.
Procedure bbbc samples a random number from the beta distribution with parameters a,b

(a > 0, b > 0). Random number generator gen is used to provide randomness. Reference: R.
C. H. Cheng [18].

PROCEDURE bsprc
( gen : RngStream;

a, b : LONGREAL
) : LONGREAL;

Beta Distribution - Stratified Rejection/Patchwork Rejection. Procedure bsprc samples a ran-
dom variate from the beta distribution with parameters a > 0, b > 0. Reference: H. Sakasegawa
[51], E. Stadlober and H. Zechner [57].

PROCEDURE bprsc
( gen : RngStream;

n, p : LONGREAL
) : LONGINT;

Binomial Distribution - Patchwork Rejection/Inversion. Procedure bprsc samples a random
number from the Binomial distribution with parameters n and p . It is valid for n∗min(p, 1−p) >
0 where n ≥ 1 and 0 < p < 1. Note that parameter n must be of type LONGREAL in this function.
Reference: V. Kachitvichyanukul and B. W. Schmeiser [30].

PROCEDURE bruec
( gen : RngStream;

n : LONGINT;
p : LONGREAL

) : LONGINT;

Binomial Distribution - Ratio of Uniforms/Inversion. Procedure bruec samples a random num-
ber from the Binomial distribution as bprsc, but using a Ratio of Uniforms method combined
with Inversion instead of a Patchwork of Rejection and Inversion method. Note that parameter
n must be a LONGINT in this function. Reference: E. Stadlober [54].

PROCEDURE btpec
( gen : RngStream;

n : LONGINT;
p : LONGREAL

) : LONGINT;

Binomial-Distribution - Acceptance Rejection/Inversion. Procedure btpec combines an Accep-
tance Rejection method with Inversion for generating Binomial random numbers. Note that
parameter n must be a LONGINT in this function. Reference: C. D. Kemp [32] and H. Zechner
[62].

PROCEDURE burr1
( gen : RngStream;

r : LONGREAL;
nr : LONGINT

) : LONGREAL;

Burr II, VII, VIII, X Distributions - Inversion. Procedure burr1 samples a random number
from one of the Burr II, VII, VIII, X distributions with parameter r > 0 , where the number of
the distribution is indicated by the variable nr (which must so be either 2, 7, 8 or 10 in order
for the function to work properly). Reference: L. Devroye [22].

PROCEDURE burr2
( gen : RngStream;

r, k : LONGREAL;
nr : LONGINT

) : LONGREAL;

Burr III, IV, V, VI, IX, XII Distribution - Inversion. Procedure burr2 samples a random
number from one of the Burr III, IV, V, VI, IX, XII distributions with parameter r > 0 and
k > 0, where the number of the distribution is indicated by the variable nr (which must so be
either 3, 4, 5, 6, 9 or 12 in order for the function to work properly). Reference: L. Devroye [22].

PROCEDURE cin
( gen : RngStream
) : LONGREAL;

Cauchy Distribution - Inversion. Procedure cin samples a random number from the standard
Cauchy distribution C(0, 1).

PROCEDURE chru
( gen : RngStream;

a : LONGREAL
) : LONGREAL;

Chi Distribution - Ratio of Uniforms with shift. Procedure chru samples a random number
from the Chi distribution with parameter a > 1. Reference: J. F. Monahan [45].

PROCEDURE eln
( gen : RngStream
) : LONGREAL;

Exponential Distribution - Inversion. Procedure eln samples a random number from the stan-
dard Exponential distribution Exp(0, 1).

PROCEDURE epd
( gen : RngStream;

tau : LONGREAL
) : LONGREAL;

Exponential Power Distribution - Acceptance Rejection. Procedure epd samples a random
number from the Exponential Power distribution with parameter tau ≥ 1 by using the non-
universal rejection method for logconcave densities. Reference: L. Devroye [22].

PROCEDURE ev_i
( gen : RngStream
) : LONGREAL;

Extreme value I Distribution - Inversion. Procedure ev_i samples a random number from the
Extreme Value I distribution.

PROCEDURE ev_ii
( gen : RngStream;

a : LONGREAL
) : LONGREAL;

Extreme Value II Distribution - Inversion. Procedure ev_ii samples a random number from
the Extreme Value II distribution with parameter a > 0.

PROCEDURE gds
( gen : RngStream;

a : LONGREAL
) : LONGREAL;

Gamma Distribution - Acceptance Rejection combined with Acceptance Complement. Proce-
dure gds samples a random number from the standard gamma distribution with parameter
a > 0. Acceptance Rejection algorithm is used for a < 1 and Acceptance Complement for
a ≥ 1. Note: calls the nsc Standard Normal distribution function of this package. Reference:
J. H. Ahrens and U. Dieter [5].

PROCEDURE gll
( gen : RngStream;

a : LONGREAL
) : LONGREAL;

Gamma Distribution - Acceptance Rejection with log-logistic envelopes. Procedure gll pro-

duces a sample from the standard gamma distribution with parameter a > 0. Reference: R. C.
H. Cheng [17].

PROCEDURE gpoic
( gen : RngStream;

my, lambda : LONGREAL
) : LONGINT;

Generalized Poisson Distribution - Inversion/Ratio of Uniforms/Rejection. Procedure gpoic

samples a random number from the Generalized Poisson distribution with the two parameters
my > 0 and lambda < 1. Reference: P. Busswald [15] and L. Devroye [23].

PROCEDURE gigru
( gen : RngStream;

h, b : LONGREAL
) : LONGREAL;

Generalized Inverse Gaussian Distribution - Ratio of Uniforms. Procedure gigru samples a
random number from the reparameterised Generalized Inverse Gaussian distribution with pa-
rameters h > 0 and b > 0 using Ratio of Uniforms method without shift for h ≤ 1 and b ≤ 1
and shift at mode m otherwise. Reference: J. S. Dagpunar [21] and K. Lehner [42].

PROCEDURE geo
( gen : RngStream;

p : LONGREAL
) : LONGINT;

Geometric Distribution - Inversion. Procedure geo samples a random number from the Geo-
metric distribution with parameter 0 < p < 1.

PROCEDURE hyplc
( gen : RngStream;

a, b : LONGREAL
) : LONGREAL;

Hyperbolic Distribution - Non-Universal Rejection. Procedure hyplc samples a random number
from the hyperbolic distribution with shape parameter a and b valid for a > 0 and |b| < a using
the non-universal rejection method for log-concave densities. Reference: L. Devroye [22].

PROCEDURE h2pec
( gen : RngStream;

N, M, n : LONGINT
) : LONGINT;

Hypergeometric Distribution - Acceptance Rejection/Inversion. Procedure h2pec samples a
random number from the Hypergeometric distribution with parameters N (number of red and
black balls), M (number of red balls) and n (number of trials) valid for N ≥ 2, M, n ≤ N, M ≥ 0,
n > 0. Note that parameters are of type LONGINT. Reference: V. Kachitvichyanukul and B. W.
Schmeiser [29].

PROCEDURE hprsc
( gen : RngStream;

N, M, n : LONGREAL
) : LONGINT;

Hypergeometric Distribution - Patchwork Rejection/Inversion. Procedure hprsc samples a

random number from the Hypergeometric distribution as h2pec does, but using a different
algorithm. Note that this time, parameters are of type LONGREAL (but should be whole numbers
for the expression to make sense.) Reference: H. Zechner [62].

PROCEDURE hruec
( gen : RngStream;

N, M, n : LONGINT
) : LONGINT;

Hypergeometric Distribution - Ratio of Uniforms/Inversion. Procedure hruec uses another
algorithm (Ratio of Uniforms combined with Inversion) to sample random numbers form an
Hypergeometric distribution. Note that parameters are of type LONGINT. Reference: E. Stad-
lober [54].

PROCEDURE john
( gen : RngStream;

my, sigma : LONGREAL;
nr : LONGINT

) : LONGREAL;

Johnson Distribution - Transformation of one Normal random number. Procedure john samples
a random number from the Johnson S(B), S(L), or S(U) distribution with parameters my and
sigma > 0, where the type of the distribution is indicated by a pointer variable nr, which must
be either 1, 2 or 3. Note: this procedure uses the nsc Standard Normal distribution function.
Reference: N. L. Johnson [28].

PROCEDURE lamin
( gen : RngStream;

l3, l4 : LONGREAL
) : LONGREAL;

Lambda Distribution - Inversion. Procedure lamin samples a random number from the Lambda
distribution with parameter l3 and l4. Reference: J. S. Ramberg and B. W. Schmeiser [49].

PROCEDURE lapin
( gen : RngStream
) : LONGREAL;

Laplace (Double Exponential) Distribution - Inversion. Procedure lapin samples a random
number from the standard Laplace distribution Lap(0, 1).

PROCEDURE login
( gen : RngStream
) : LONGREAL;

Logistic Distribution - Inversion. Procedure login samples a random number from the standard
Logistic distribution Log(0, 1).

PROCEDURE lsk
( gen : RngStream;

p : LONGREAL
) : LONGINT;

Logarithmic Distribution - Inversion/Transformation. Procedure lsk samples a random number
from the Logarithmic distribution with parameter 0 < p < 1. Reference: A. W. Kemp [31].

PROCEDURE nbp
( gen : RngStream;

r, p : LONGREAL
) : LONGINT;

Negative Binomial Distribution - Compound method. Procedure nbp samples a random num-
ber from the Negative Binomial distribution with parameters r (no. of failures given) and p

(probability of success) valid for r > 0, 0 < p < 1. Note: this procedure uses both procedures
gds and pd in its implementation. Reference: J. H. Ahrens and U. Dieter [1].

PROCEDURE nsc
( gen : RngStream
) : LONGREAL;

Normal Distribution - Sinus-Cosinus or Box/Muller Method. Procdeure nsc samples a random
number from the standard Normal distribution N(0, 1). Reference: G. E. P. Box and M. E.
Muller [12].

PROCEDURE pd
( gen : RngStream;

my : LONGREAL
) : LONGINT;

Poisson Distribution - Tabulated Inversion combined with Acceptance Complement. Procedure
pd samples a random number from the Poisson distribution with parameter my > 0. Tabulated
Inversion for my < 10, Acceptance Complement for my ≥ 10. Uses nsc Standard Normal
distribution random number generator in its implementation. Reference: J. H. Ahrens and U.
Dieter [4].

PROCEDURE pruec
( gen : RngStream;

my : LONGREAL
) : LONGINT;

Poisson Distribution - Ratio of Uniforms/Inversion. Procedure pruec samples a random number
from the Poisson distribution as function pd does, but using an Inversion method for my < 5.5
and a Ratio of Uniforms for my ≥ 5.5. Reference: E. Stadlober [54].

PROCEDURE pprsc
( gen : RngStream;

my : LONGREAL
) : LONGINT;

Poisson Distribution - Patchwork Rejection/Inversion. Procedure pprsc samples a random
number from the Poisson distribution. For parameter my < 10, Tabulated Inversion is applied
while for my ≥ 10 Patchwork Rejection is employed. Reference: H. Zechner [62].

PROCEDURE slash
( gen : RngStream
) : LONGREAL;

Slash Distribution - z/u (z from N(0, 1), u from U(0, 1)). Procedure slash samples a random
number from the Slash distribution. Note: makes use of the Standard Normal distribution
provided by the nsc function.

PROCEDURE trouo
( gen : RngStream;

a : LONGREAL
) : LONGREAL;

Student-t Distribution - Ratio of Uniforms. Procedure trouo samples a random number from
the Student-t distribution with parameter a ≥ 1. Reference: A. J. Kinderman and J. F.
Monahan [33].

PROCEDURE tpol
( gen : RngStream;

a : LONGREAL
) : LONGREAL;

Student-t Distribution - Polar Method. Procedure tpol uses the polar method of Box/Muller
for generating Normal variates in order to produce Student-t distributed random numbers, using
parameter a > 0. Reference: R. W. Bailey [7].

PROCEDURE stab
( gen : RngStream;

alpha, delta : LONGREAL
) : LONGREAL;

Stable Distribution - Integral Representations. Procedure stab samples a random number from
the Stable distribution with parameters 0 < alpha ≤ 2 and −1 ≤ delta ≤ 1. Reference: J. M.
Chambers et al. [16].

PROCEDURE tra
( gen : RngStream
) : LONGREAL;

Triangular Distribution - Inversion: x = + − (1 − sqrt(u)). Procedure tra samples a random
number from the standard Triangular distribution in (−1, 1).

PROCEDURE mwc
( gen : RngStream;

k : LONGREAL
) : LONGREAL;

Von Mises Distribution - Acceptance Rejection. Procedure mwc samples a random number from
the von Mises distribution (−π ≤ x ≤ π) with parameter k > 0 via rejection from the wrapped
Cauchy distibution. Reference: D. J. Best and N. I. Fisher [8].

PROCEDURE win
( gen : RngStream;

c : LONGREAL
) : LONGREAL;

Weibull Distribution - Inversion. Procedure win samples a random number from the Weibull
distribution with parameter c > 0.

PROCEDURE zeta
( gen : RngStream;

ro, pk : LONGREAL
) : LONGINT;

Zeta Distribution - Acceptance Rejection. Procedute zeta samples a random number from the
Zeta distribution with parameters ro > 0 and pk ≥ 0. Reference: J. Dagpunar [20].

END RAND2.

RANDLIB

This module is an interface to the module RANDLIB [14], written in C, and available
at http://odin.mdacc.tmc.edu/anonftp/page_2.html#RANDLIB. The original module has
been adapted to provide full compatibility with the object-based approach of the RAND mod-
ule, which provides the underlying uniform generator for RANDLIB, but it has not been altered
otherwise.

DEFINITION MODULE ["C"] RANDLIB;

FROM RAND IMPORT RngStream;

PROCEDURE genbet
( gen : RngStream;

a, b : LONGREAL
) : LONGREAL;

Returns a single random deviate from the beta distribution with parameters a and b. The
density of the beta is x(a−1) ∗ (1−x)(b−1)/B(a, b) for 0 < x < 1. Method used: R. C. H. Cheng
[18].

PROCEDURE genchi
( gen : RngStream;

DF : LONGREAL
) : LONGREAL;

Returns a single random deviate from the distribution of a chisquare with DF degrees of freedom
random variable. DF must be > 0.

PROCEDURE genexp
( gen : RngStream;

AV : LONGREAL
) : LONGREAL;

Returns a single random deviate from an exponential distribution with mean AV, which must
be ≥ 0. For details about algorithm, see J. H. Ahrens and U. Dieter [2].

PROCEDURE genf
( gen : RngStream;

DFN, DFD : LONGREAL
) : LONGREAL;

Returns a single random deviate from the F (variance ratio) distribution with DFN degrees of
freedom in the numerator and DFD degrees of freedom in the denominator. DFN and DFD must
be > 0.

PROCEDURE gengam
( gen : RngStream;

A, R : LONGREAL
) : LONGREAL;

Returns a single random deviate from the gamma distribution with location parameter A and
shape parameter R. A and R must be > 0. Uses methods from Ahrens and Dieter [5, 1].

PROCEDURE genmn
( gen : RngStream;

Parm : ARRAY OF LONGREAL;
X : ARRAY OF LONGREAL;
Work : ARRAY OF LONGREAL

);

Generates a multivariate normal random deviate. Return in X the generated vector deviate. In
Parm are the parameters needed. Parm must be set by a previous call to setgmn. Work is a
scratch array.

PROCEDURE setgmn
( Meanv : ARRAY OF LONGREAL;

Covm : ARRAY OF ARRAY OF LONGREAL;
P : LONGINT;
Parm : ARRAY OF LONGREAL

);

Returns in Parm the needed information to call genmn, i.e. P, Meanv and the Cholesky fac-
torization of Covm. Meanv is the mean vector of multivariate normal distribution. Covm is the
covariance matrix of the multivariate of normal distribution and is returned destroyed. P is the
dimension of the normal, or length of Meanv. Parm dimension must be at least P*(P+3)/2 + 1.

PROCEDURE genmul
( gen : RngStream;

N : LONGINT;
P : ARRAY OF LONGREAL;
NCAT : LONGINT;
IX : ARRAY OF LONGINT

);

Generates an observation from the multinomial distribution. N is the number of events that will
be classified into one of the categories 1..NCAT (N ≥ 0). P is the vector of probabilities. P(i) is
the probability that an event will be classified into category i. Thus, P(i)must be [0,1]. Only
the first NCAT-1 P(i) must be defined since P(NCAT) is 1.0 minus the sum of the first NCAT-1 P(i).
NCAT is the number of categories. Length of P and IX. (NCAT > 1). Result will be put in IX.
All IX(i), the number of events of the ith category that occured, will be nonnegative and their
sum will be N. Uses algorithm from Devroye [22].

PROCEDURE gennch
( gen : RngStream;

DF, Nonc : LONGREAL
) : LONGREAL;

Returns a single random deviate from the distribution of a noncentral chisquare with DF degrees
of freedom and noncentrality parameter Nonc. DF must be ≥ 1.0 and Nonc, ≥ 0.

PROCEDURE gennf
( gen : RngStream;

DFN, DFD, Nonc : LONGREAL
) : LONGREAL;

Returns a random deviate from the noncentral F (variance ratio) distribution with DFN degrees
of freedom in the numerator, and DFD degrees of freedom in the denominator, and noncentrality
parameter Nonc. DFN must be ≥ 1.0, DFD must be > 0 and Nonc must be ≥ 0.

PROCEDURE gennor
( gen : RngStream;

AV, SD : LONGREAL
) : LONGREAL;

Returns a single random deviate from a normal distribution with mean, AV, and standard
deviation, SD. SD must be ≥ 0. Method: Ahrens and Dieter [3].

PROCEDURE genprm
( gen : RngStream;

IArray : ARRAY OF LONGINT;
LArray : LONGINT

);

Generates a random permutation of the elements of IArray. LArray is the length of IArray.

PROCEDURE ignbin
( gen : RngStream;

N : LONGINT;
P : LONGREAL

) : LONGINT;

Returns a single random deviate from a binomial distribution whose number of trials is N

and whose probability of an event in each trial is P. Method used: algorithm BTPE from
Kachitvichyanukul and Schmeiser [30].

PROCEDURE ignnbn
( gen : RngStream;

N : LONGINT;
P : LONGREAL

) : LONGINT;

Returns a single random deviate from a negative binomial distribution. N is the number of
events wanted and P is the probability of an event in each trial. Algorithm from Luc Devroye
[22].

PROCEDURE ignpoi
( gen : RngStream;

AV : LONGREAL
) : LONGINT;

Returns a single random deviate from a Poisson distribution with mean AV. AV must be > 0.0.
For details about algorithm, see Ahrens and Dieter [4].

PROCEDURE ignuin
( gen : RngStream;

Low, High : LONGINT
) : LONGINT;

Returns a single random deviate uniformly distributed between Low and High. (High - Low)
must be < 2,147,483,561.

END RANDLIB.

SIM

Ce module fait l’interface avec l’exécutif du progiciel de simulation. Il fournit des
procédures permettant d’initialiser, de démarrer ou d’arrêter la simulation, de même qu’une
fonction qui retourne la valeur de l’horloge.

DEFINITION MODULE SIM;

PROCEDURE Init;

Réinitialise la simulation en vidant la liste des événements, si elle ne l’est pas déjà, et en
remettant l’horloge à zéro. S’il y a des processus existants (voir module PROCS), on doit tous
les détruire avant d’appeler cette procédure.

PROCEDURE Start;

Démarre l’exécutif de la simulation. Lors de son appel, il doit déjà y avoir au moins un événement
ou un processus de prévu.

PROCEDURE Stop;

Arrête l’exécutif de la simulation et retourne le contrôle au programme principal (ou à l’une de
ses procédures), juste après l’instruction SIM.Start qui a démarré cette simulation. A noter
cependant que l’exécutif ne s’arrêtera que lorsqu’on lui aura tranféré le contrôle. Si on est en
train d’exécuter un événement, ce sera dès la fin de la procédure d’événement, et si on est dans
un processus, ce sera dès que ce processus appelle Terminate pour terminer son exécution ou
passe le contrôle à l’exécutif en appelant Delay ou Suspend ou Request ou Take, etc.

PROCEDURE Time () : LONGREAL;

Retourne la valeur de l’instant courant de la simulation.

END SIM.

EVENT

Ce module offre les outils nécessaires pour effectuer une simulation avec vision par
événements. Le type EventType, prédéfini dans ce module, correspond à un type d’événe-
ment qui est associé à une procédure sans paramètre. Pour chaque type d’événement, l’usager
doit déclarer une variable de type EventType, puis appeler Create pour lui associer sa
procédure correspondante, qui sera exécutée à chaque occurence d’un événement de ce type.
On peut prévoir un événement dans un délai donné, par la procédure Schedule. On donne
alors le type de l’événement, le délai, et un pointeur sur les attributs (paramètres) associés
à cet événement. On peut récupérer la valeur de ce pointeur, à l’intérieur de la procédure
qui exécute l’événement en question, en appelant la fonction Attrib. La procédure Cancel
permet d’annuler un événement déjà prévu, selon un attribut particulier. Pour chaque
événement prévu, le système crée un avis d’événement (EventNotice) qui contient le type
de l’événement (le EventType), son instant d’occurence (supérieur ou égal à l’instant courant)
et un pointeur sur ses attributs, s’il en a.

Lorsqu’on veut prévoir un événement pour l’instant courant, et que l’on veut qu’il soit
placé au tout début de la liste des événements, avant les autres déjà prévus pour le même in-
stant s’il y a lieu, on appelle plutôt ScheduleNext au lieu de Schedule. Les procédures
ScheduleNotice et ScheduleNoticeNext permettent de prévoir un événement tout en
récupérant le pointeur sur l’avis d’événement. Cela peut permettre par exemple d’annuler
ultérieurement cet avis d’événement en appelant CancelNotice. On peut aussi placer un
avis d’événement juste avant ou juste après un autre avis qui se trouve déjà dans la liste
(voir ScheduleNoticeBefore et ScheduleNoticeAfter). D’autres procédures permettent
d’obtenir le type, le pointeur sur les attributs, et l’instant d’occurence d’un événement prévu,
de retrouver un avis d’événement satisfaisant une condition particulière, ou de faire imprimer
la liste des événements prévus.

DEFINITION MODULE EVENT;

FROM SYSTEM IMPORT ADDRESS;

TYPE
EventType;

Un type d’événement. Pour chaque type d’événement qu’il veut utiliser, l’usager doit déclarer
une variable de ce type, puis appeler la procédure Create.

EventNotice;

Un avis d’événement.

CondProc = PROCEDURE (ADDRESS) : BOOLEAN;

Type de procédure utilisé comme paramètre dans la procédure FindNoticeSuchThat.

PROCEDURE Create
( VAR E : EventType;

P : PROC;
Name : ARRAY OF CHAR

);

Associe la procédure P et l’identificateur Name au type d’événement E. Cette procédure doit
être obligatoirement appelée pour chaque type d’événement. Name servira à identifier ce type
d’événement si on fait imprimer la liste des événements. On recommande un maximum de 32
caractères.

PROCEDURE Schedule
( E : EventType;

D : LONGREAL;
Par : ADDRESS

);

Prévoit un événement de type E, en insérant un avis dans la liste des événements. Son occurence
est prévue pour l’instant “SIM.Time() + D”, c’est-à-dire dans D unités de temps. L’usager peut
associer à cet événement un bloc de paramètres (un attribut), sous forme d’un pointeur dont
la valeur est donnée au paramètre formel Par. Il pourra récupérer la valeur de ce pointeur lors
de l’exécution de l’événement, en utilisant la fonction Attrib(). D doit être non négatif, sinon
une erreur est signalée et rien d’autre n’est effectué. Lorsque deux ou plusieurs événements sont
prévus pour le même instant, ils sont placés dans la liste (et exécutés) dans l’ordre selon lequel
ils ont été prévus.

PROCEDURE ScheduleNext
( E : EventType;

Par : ADDRESS
);

Tout comme Schedule, sauf que l’événement prévu est placé au tout début de la liste
d’événements, et doit s’exécuter à l’instant courant. Si d’autres événements sont aussi prévus
pour l’instant courant, l’événement que l’on prévoit maintenant s’exécutera avant eux.

PROCEDURE Cancel
( E : EventType;

Par : ADDRESS
);

Annule (enlève de la liste d’événements) le premier avis d’événement de type E et qui a été
prévu avec la valeur Par pour ses attributs. La recherche s’effectue à partir du début de la liste
des événements (par ordre chronologique). Si la liste d’événements est vide, ou que l’événement
recherché est inexistant, rien n’est effectué.

PROCEDURE Attrib () : ADDRESS;

Retourne le pointeur sur les attributs de l’événement en cours d’exécution. Ce pointeur est
la valeur du paramètre Par fourni lors de sa prévision. C’est à l’usager de s’assurer que ce
résultat est affecté à une variable du même type (c’est-à-dire un pointeur sur le même type de
structure) que la variable passée en paramètre lors de la prévision de cet événement. Attention :
si l’usager appelle cette procédure ailleurs que dans une procédure exécutant un événement, les

résultats sont imprévisibles (il est possible que l’on soit en train d’exécuter un événement défini
par SIMOD. . . ).

PROCEDURE ScheduleNotice
( E : EventType;

D : LONGREAL;
Par : ADDRESS;
VAR N : EventNotice

);

Tout comme Schedule, mais retourne l’avis d’événement associé.

PROCEDURE ScheduleNoticeNext
( E : EventType;

Par : ADDRESS;
VAR N : EventNotice

);

Tout comme ScheduleNext, mais retourne l’avis d’événement associé.

PROCEDURE ScheduleNoticeBefore
( E : EventType;

Par : ADDRESS;
VAR N : EventNotice;
N1 : EventNotice

);

Tout comme ScheduleNotice, sauf que l’avis d’événement N est placé immédiatement avant
l’avis N1 dans la liste. N1 doit être un avis d’événement déjà prévu. Cette procédure peut
être utile par exemple pour placer un événement “espion” qui s’exécutera juste avant un autre
événement donné.

PROCEDURE ScheduleNoticeAfter
( E : EventType;

Par : ADDRESS;
VAR N : EventNotice;
N1 : EventNotice

);

Tout comme ScheduleNoticeBefore, sauf que l’avis d’événement N est placé immédiatement
après l’avis N1 dans la liste.

PROCEDURE FindNoticeSuchThat
( E : EventType;

Cond : CondProc
) : EventNotice;

Retourne l’avis d’événement associé au premier événement de type E dans la liste tel que la
procédure Cond, appliquée au pointeur sur les attributs de cet événement, retourne TRUE. S’il
n’y en a pas, cette procédure retourne NIL.

PROCEDURE CancelNotice
( VAR N : EventNotice
);

Annule l’événement associé à l’avis d’événement N.

PROCEDURE NoticeType
( N : EventNotice
) : EventType;

Retourne le type de l’événement associé à l’avis d’événement N.

PROCEDURE NoticeAttrib
( N : EventNotice
) : ADDRESS;

Retourne le pointeur sur les attributs de l’événement associé à N. Ce pointeur est la valeur du
paramètre Par fourni lors de sa prévision.

PROCEDURE NoticeTime
( N : EventNotice
) : LONGREAL;

Retourne l’instant (prévu) d’occurence de l’événement associé à N.

PROCEDURE PrintEventList;

Imprime la liste des événements dans le fichier de sortie courant.

PROCEDURE EmptyEventList;

Vide complètement la liste des événements prévus, c’est-à-dire annule tous ces événements.

PROCEDURE DeleteType
( VAR E : EventType
);

Détruit le type d’événement E et récupère l’espace mémoire utilisé pour sa description.

END EVENT.

CONT

Ce module permet la simulation continue, pour laquelle l’évolution de certaines variables
en fonction du temps est régie par des équations différentielles. Dans un même modèle,
on peut mélanger des événements discrets, des processus, et des variables “continues” qui
obéissent à des équations différentielles. Il doit cependant y avoir un nombre fini de telles
variables, et pour chacune d’entre elles, on doit donner l’équation différentielle qui la régit,
sous forme d’une fonction. Ce module ne permet pas de traiter les équations aux dérivées
partielles, que l’on simule habituellement par des techniques d’éléments finis.

Pour chaque variable continue que l’on veut utiliser, on doit déclarer une variable de type
VarCont, puis la créer en appelant Create. Lors de la création de la variable, on doit donner
sa valeur initiale, et une fonction qui retourne sa dérivée en fonction du temps. L’évolution
de la valeur de la variable se fait en intégrant cette fonction. Pour chaque variable, on
peut démarrer l’intégration en appelant StartInteg, et la stopper en appelant StopInteg.
Lors du démarrage, on peut aussi choisir un type d’événement qui sera exécuté à chaque
pas d’intégration pour cette variable (juste après avoir effectué ce pas). Un tel événement
pourrait par exemple vérifier si une variable continue a atteint un certain seuil, ou encore
mettre à jour un graphique illustrant l’évolution du système. Avant de démarrer toute
intégration, on doit appeler SelectMethod pour choisir la méthode et la longueur h du pas
d’intégration. Le même h et la même méthode seront utilisés pour toutes les variables. Le
pas d’intégration qui fait passer la variable de la valeur qu’elle a à l’instant t − h, à celle
qu’elle doit avoir à l’instant t, se fait lorsque l’horloge de la simulation atteint l’instant t.

Lors de la création d’une variable, on peut aussi lui associer un paramètre (en général
un pointeur), que l’on pourra ensuite récupérer en appelant VarAttrib. La fonction
CurrentVar, lorsqu’appelée à l’intérieur de la fonction de type Deriv associée à une variable
continue V, durant l’intégration, retourne V. En particulier, une combinaison de ces deux
fonctions permet d’examiner le paramètre de la variable que l’on est en train d’intégrer. On
peut réinitialiser la valeur d’une variable continue en appelant Init, et examiner sa valeur
courante en appelant Value. Enfin, pour la détruire et récupérer l’espace qu’elle occupait,
il suffit d’appeler Delete.

DEFINITION MODULE CONT;

FROM EVENT IMPORT EventType;
FROM SYSTEM IMPORT ADDRESS;

TYPE

VarCont;

Pour chaque variable continue que l’on veut utiliser, on doit déclarer une variable de ce type,
puis appeler Create pour la créer.

Deriv = PROCEDURE (LONGREAL) : LONGREAL;

Type de procédure à passer en paramètre dans Create.

Method = (Euler, RungeKutta2, RungeKutta4);

Une méthode d’intégration. Euler: la méthode d’Euler; RungeKutta2: la méthode d’Euler
modifiée (Runge Kutta d’ordre 2); RungeKutta4: Runge Kutta d’ordre 4.

PROCEDURE Create
( VAR V : VarCont;

Val : LONGREAL;
D : Deriv;
Par : ADDRESS

);

Crée la variable continue V, l’initialise à la valeur Val, et lui associe la procédure D. Cette
procédure, appelée avec le paramètre t, doit retourner la dérivée de la valeur de la variable
en fonction du temps, au temps t, en fonction des valeurs courantes des variables (continues
ou autres) du modèle. Le paramètre Par est en général un pointeur sur un bloc d’information
quelconque associé à cette variable (ou toute autre variable compatible avec ADDRESS). On
pourra récupérer sa valeur en appelant VarAttrib.

PROCEDURE Init
( V : VarCont;

Val : LONGREAL
);

Réinitialise la valeur de la variable V à Val.

PROCEDURE Value
( V : VarCont
) : LONGREAL;

Retourne la valeur courante de la variable V.

PROCEDURE VarAttrib
( V : VarCont
) : ADDRESS;

Retourne la valeur du paramètre associé à V, c’est-à-dire la valeur de Par fournie lors de sa
création.

PROCEDURE CurrentVar
() : VarCont;

Lorsqu’appelée à l’intérieur d’une procédure de type Deriv associée à une variable V, durant
l’intégration, cette fonction retourne V. Sinon, retourne NIL.

PROCEDURE SelectMethod
( M : Method;

h : LONGREAL
);

Choisit la méthode d’intégration qui fait évoluer l’état des variables continues, et la longueur
du pas d’intégration. Pour toutes les variables pour lesquelles l’intégration a été démarrée, la
méthode d’intégration sera M, avec un pas d’intégration de h unités de temps.

PROCEDURE StartInteg
( V : VarCont;

E : EventType;
Par : ADDRESS

);

Démarre le processus d’intégration qui fait évoluer l’état de la variable V. Juste après chaque
pas d’intégration, un événement de type E, avec Par comme pointeur sur ses paramètres, sera
exécuté. Si on ne veut pas d’un tel événement, on passe EventType (NIL) pour valeur de E.

PROCEDURE StopInteg
( V : VarCont
);

Stoppe l’intégration qui avait été démarrée par StartInteg pour la variable V. La valeur de V

demeurera celle calculée lors du dernier pas d’intégration effectué avant l’appel à StopInteg.

PROCEDURE Delete
( VAR V : VarCont
);

Détruit la variable continue V, et récupère l’espace qu’elle occupait.

END CONT.

STAT

Le module STAT permet de recueillir des statistiques, d’obtenir des rapports et de calculer
des intervalles de confiance. Il fournit un type prédéfini appelé Block, qui correspond à un
bloc d’information pour le recueil de statistiques. Pour chaque quantité pour laquelle on
désire recueillir des statistiques, on déclare une variable de type Block, puis on crée le bloc
d’information correspondant en appelant Create. On doit distinguer deux types de blocs
statistiques. On a un bloc de type Tally lorsqu’on s’intéresse à une séquence d’observations
(de valeurs numériques) X1, X2, X3, . . . d’une variable. Par ailleurs, lorsqu’on s’intéresse
à l’évolution dans le temps de la valeur d’une variable, X(t), t ≥ 0, on a un bloc de type
Accumulate. Par exemple, pour estimer la longueur moyenne d’une file d’attente en fonction
du temps, on utilisera un bloc de type Accumulate, tandis que pour estimer la durée moyenne
d’attente par client, on utilisera un bloc de type Tally. Le type d’un bloc doit être déclaré
lors de sa création.

Dans SIMOD , les blocs statistiques déclarés par l’usager ne sont pas mis à jour automa-
tiquement par le système. L’usager doit appeler la procédure Update pour fournir la nouvelle
valeur de Xi chaque fois qu’une nouvelle valeur est observée, dans le cas d’un bloc de type
Tally, et pour donner la nouvelle valeur de X(t) chaque fois que la valeur de la variable est
modifiée, dans le cas d’un bloc de type Accumulate.

La procédure Init permet de réinitialiser, à n’importe quel moment, tous les compteurs
associés à un bloc statistique. On peut donc réinitialiser sélectivement ces derniers au besoin.
Lorsqu’on effectue plusieurs répétitions, par exemple, on peut utiliser un bloc statistique pour
recueillir la distribution des moyennes des différentes répétitions, et réinitialiser seulement
les autres blocs au début de chaque répétition. Delete permet d’éliminer un bloc statistique
et d’en récupérer l’espace.

La procédure Report permet d’obtenir un rapport complet sur un bloc statistique.
Lorsqu’on veut connâıtre seulement la valeur d’une statistique particulière, on peut faire ap-
pel aux fonctions NumberObs, Minimum, Maximum, Sum, Average, Variance et StandardDev.
A noter cependant que le nombre d’observations, la variance et l’écart-type ne sont disponibles
que pour les blocs de type Tally. La variance et l’écart-type n’ont d’utilité, en général, que
lorsque les valeurs observées sont indépendantes (par exemple les moyennes de plusieurs
répétitions). On peut aussi, sous ces mêmes restrictions, calculer un intervalle de confi-
ance, de niveau spécifié, pour la moyenne d’une variable. Il suffit d’appeler la procédure
ConfidenceInterval ou ReportConfidenceInterval. Ces procédures supposent cepen-
dant que les valeurs observées sont indépendantes et identiquement distribuées selon une loi
normale. Les intervalles de confiance obtenus ne seront valides que si cette hypothèse est
satisfaite, ou n’est pas trop loin de l’être. Pour un bloc de type Accumulate, tout appel à
l’une des procédures mentionnées dans ce paragraphe provoque une mise-à-jour automatique
(par un appel à Update).

DEFINITION MODULE STAT;

TYPE
Block;

Bloc statistique renfermant toutes les informations nécessaires pour le recueil des statis-
tiques concernant une variable donnée. Chaque bloc statistique que l’on veut utiliser doit
être identifié par une variable de type Block, dont l’identificateur servira de nom au bloc
statistique en question, puis créé à l’aide de la procédure Create.

BlockType = (Tally, Accumulate);

Sorte de bloc statistique.

PROCEDURE Create
( VAR X : Block;

Kind : BlockType;
Name : ARRAY OF CHAR

);

Crée le bloc statistique X, dont le type est spécifié par Kind. Alloue l’espace et initialise tous
les compteurs associés à ce bloc. Le paramètre Name permet de donner un nom à X sous forme
d’une châıne de caractères. Ce nom sera utilisé lorsqu’on fera imprimer un rapport pour X. On
recommande un maximum de 32 caractères.

PROCEDURE Init
( X : Block
);

Initialise ou réinitialise tous les compteurs associés au bloc statistique X. Le bloc X doit avoir
été créé auparavant. A noter que dans le cas où X est de type Accumulate, Init ne modifie pas
la valeur courante de la variable. Cette valeur courante est toujours celle du dernier appel à
Update. Pour l’initialiser, la modifier, ou la remettre à 0, il faut appeler Update. Habituellement,
un appel à Init pour un bloc de type Accumulate devrait être immédiatement suivi d’un appel
à Update.

PROCEDURE Update
( X : Block;

V : LONGREAL
);

Met à jour tous les compteurs associés au bloc statistique X. Si X est de type Tally, une nouvelle
observation de la variable, de valeur V, est ajoutée. Si X est de type Accumulate, cette mise a
jour indique qu’à partir de l’instant courant, la variable associée au bloc X prend la valeur V.

PROCEDURE Report
( X : Block
);

Imprime, dans le fichier de sortie (habituellement à l’écran), un rapport complet sur la variable
associée au bloc statistique X, depuis la dernière initialisation de ce bloc.

PROCEDURE NumberObs
( X : Block
): LONGCARD ;

Pour une statistique de type Tally, fournit le nombre de valeurs observées pour la variable
associée au bloc statitique X depuis la dernière initialisation de ce bloc. Pour une statistique de
type Accumulate, imprime un message d’erreur.

PROCEDURE Minimum
( X : Block
): LONGREAL;

Fournit la plus petite valeur qu’a prise la variable associée au bloc statistique X depuis la dernière
initialisation de ce bloc.

PROCEDURE Maximum
( X : Block
): LONGREAL;

Fournit la plus grande valeur qu’a prise la variable associée au bloc statistique X depuis la
dernière initialisation de ce bloc.

PROCEDURE Sum
( X : Block
): LONGREAL;

Pour une statistique de type Tally, fournit la somme des valeurs prises par la variable associée
au bloc X depuis sa dernière initialisation. Pour une statistique de type Accumulate, fournit
l’intégrale de la valeur de la variable, par rapport au temps, depuis la dernière initialisation du
bloc X jusqu’à l’instant courant.

PROCEDURE Average
( X : Block
): LONGREAL;

Pour une statistique de type Tally, fournit la moyenne des observations de la variable associée au
bloc X (somme des valeurs divisée par le nombre d’observations) depuis sa dernière initialisation.
Pour une statistique de type Accumulate, fournit la valeur moyenne de la variable, par unité
de temps, depuis la dernière initialisation du bloc.

PROCEDURE Variance
( X : Block
): LONGREAL;

Pour une statistique de type Tally, fournit la variance des observations de la variable associée
au bloc X (somme des carrés des écarts à la moyenne, divisée par le nombre d’observations moins
1) depuis sa dernière initialisation. Si le nombre d’observations est inférieur à 2, ou si X est
de type Accumulate, un message d’erreur sera imprimé. Attention: L’implantation utilise la
double précision, mais il peut y avoir une erreur numérique significative lorsque la variance est
très petite par rapport à la moyenne.

PROCEDURE StandardDev
( X : Block
): LONGREAL;

Pour une statistique de type Tally, fournit l’écart-type des observations. Appelle la fonction
Variance et extrait la racine carrée.

PROCEDURE ConfidenceInterval
( X : Block;

Level : LONGREAL; (* Niveau de confiance. *)
VAR Center, Radius : LONGREAL (* Milieu et rayon de l’intervalle. *)

);

Pour une statistique de type Tally, fournit le point milieu et le rayon d’un intervalle de confi-
ance, de niveau spécifié par l’usager, pour la vraie moyenne µ de la variable associée à X. Cet
intervalle est calculé en utilisant la statistique :

T =
X − µ

Sx/
√
n

où n est le nombre d’observations de la variable, X = Average(X) est la moyenne échantillonna-
le, et Sx = StandardDev(X) est l’écart-type échantillonnal. Si on suppose que les observations
de la variable sont indépendantes et identiquement distribuées suivant une loi normale, alors la
statistique T suit la loi de Student à n− 1 degrés de liberté. Lorsque cette hypothèse n’est pas
vérifiée, les valeurs fournies peuvent avoir une valeur statistique doûteuse. Si n < 2, ou si la
statistique X est de type Accumulate, un message d’erreur sera imprimé.

PROCEDURE ReportConfidenceInterval
( X : Block;

Level : LONGREAL (* Niveau de confiance. *)
);

Pour une statistique de type Tally, imprime un intervalle de confiance, de niveau spécifié par
l’usager, pour la vraie moyenne µ de la variable associée à X. Cet intervalle est calculé en utilisant
la procédure ConfidenceInterval.

PROCEDURE Delete
( VAR X : Block
);

Détruit le bloc statistique X et récupère l’espace mémoire qu’il utilisait.

END STAT.

PROCS

Ce module fournit les outils de base pour la simulation avec vision par processus. Le type
ProcessType correspond à un type de processus. Pour chaque type de processus, l’usager
doit déclarer une variable de type ProcessType, puis appeler Create, pour lui associer une
procédure (sans paramètres) et déclarer la taille de l’espace mémoire requis à l’exécution de
sa coroutine. Un type de processus correspond habituellement à un type d’objet dans le
modèle, et la procédure (coroutine) associée décrit le comportement de cet objet.

La procédure Schedule permet de prévoir l’activation d’un processus d’un type donné,
dans un délai donné. Cette procédure place un avis d’événemment (EventNotice) dans la
liste d’événements, et c’est cet événement qui déclenchera l’exécution du processus. Lors
de la prévision, on donne le type du processus, le délai, et un pointeur sur les attributs
(les paramètres) que l’on veut associer à ce processus. La valeur de ce pointeur peut être
récupérée, durant l’exécution du processus en question, en appelant la fonction Attrib.
Habituellement, les processus sont anonymes, c’est-à-dire que les processus d’un même type
(il peut y en avoir plusieurs qui évoluent simultanément) n’ont pas de noms qui les dis-
tinguent. Lorsqu’on veut donner un nom spécifique (un identificateur) à un processus, on
doit d’abord déclarer un identificateur de type ProcessInstance, puis prévoir l’activation
de ce processus en appelant ScheduleNamed au lieu de Schedule. Ce processus devient alors
baptisé, et on pourra y référer par l’identificateur en question. Lorsqu’on veut qu’un proces-
sus débute son exécution à l’instant courant, et que l’événement déclenchant cette exécution
soit placé au tout début de la liste d’événements, avant les autres déjà prévus pour le même
instant s’il y a lieu, on appelle plutôt ScheduleNext ou ScheduleNamedNext pour le prévoir
(l’effet équivalent peut être obtenu en appelant ScheduleNamed suivi de Resume pour le pro-
cessus en question). Chaque processus reçoit aussi un numéro de série, qui lui est donné
lors de sa prévision: le n-ième processus prévu reçoit le numéro n. On peut connâıtre le
numéro de série d’un processus en appelant InstanceSerialNumber. L’intérêt est que ce
numéro est unique: deux processus ne peuvent pas avoir le même numéro (à moins que
l’on génère un plus grand nombre de processus que le plus grand entier représentable sur
l’ordinateur). Par contre, l’identificateur de type ProcessInstance est en fait un pointeur
sur le bloc d’information associé à un processus. Deux processus qui existent simultanément
ne peuvent pas avoir des identificateurs de même valeur, mais un processus prévu après
qu’un autre processus se soit terminé peut avoir un identificateur de même valeur que ce
dernier.

Un processus existant peut se trouver dans l’un des trois états suivants: Executing,
Delayed ou Suspended (voir le diagramme ci-haut). Il n’y a jamais plus d’un seul processus
à la fois qui se trouve dans l’état Executing, et lorsqu’il y en a un, c’est celui-ci qui est
en exécution (qui est en train d’exécuter ses instructions). On dit aussi qu’il a le contrôle.
Lorsque ce n’est pas un processus de l’usager qui a le contrôle (par exemple lorsqu’on exécute
un événement), on dit que c’est le “processus principal” (ou encore l’exécutif de la simulation)
qui l’a. Un processus est dans l’état Delayed s’il y a un événement de prévu pour lui redonner
le contrôle. Il est dans l’état Suspended s’il est en attente qu’on lui redonne le contrôle, sans
qu’il y ait d’événement de prévu à cet effet.

Lorsqu’on prévoit son activation, un processus passe à l’état Delayed. Il passe à l’état
Executing lorsqu’il débute ou reprend son exécution. Le processus en exécution peut passer
à l’état Suspended en appelant Suspend (ou en appelant une procédure qui appelle elle-même
Suspend, comme c’est le cas dans les modules RES, BIN, etc.). Dans ce cas, on exécutera
ensuite le prochain événement de la liste (souvent, cet événement redonnera le contrôle à un
autre processus). Le processus en exécution peut aussi appeler Delay afin de passer à l’état
Delayed, et attendre que l’horloge de la simulation ait avancé d’un certain nombre d’unités
de temps avant de reprendre son exécution. Cette procédure place dans la liste d’événements
un avis d’événement qui redonnera le contrôle à ce processus. On peut modifier par la suite
l’instant d’occurence de cet événement en appelant ResetDelay. On peut aussi appeler
ResetDelay pour faire passer un processus de l’état Suspended à l’état Delayed, et prévoir
un événement qui lui redonnera éventuellement le contrôle. Pour faire passer un processus de
l’état Delayed à l’état Suspended, il suffit d’appeler CancelInstanceNotice. Par ailleurs,
si on veut que l’événement redonnant le contrôle à un processus soit placé au tout début de
la liste, on peut appeler Resume.

D’autres procédures permettent de connâıtre le type d’un processus, son état, le poin-
teur sur ses attributs, l’avis d’événement prévu qui lui donnera ou redonnera le contrôle
(s’il y a lieu), et le temps qui reste avant l’exécution de cet l’événement. La procédure
CurrentProcess permet aussi de connâıtre le processus présentement en cours d’exécution.

La vie d’un processus se termine lorsqu’il appelle la procédure Terminate, ou encore
lorsqu’on appelle Kill pour le détruire de l’extérieur. On peut aussi détruire d’un seul coup
tous les processus. A noter qu’un processus doit toujours se terminer avant d’atteindre le
END final de sa procédure associée, sinon le programme s’arrêtera lorsque ce END sera atteint.

DEFINITION MODULE PROCS;

FROM EVENT IMPORT EventNotice;
FROM SYSTEM IMPORT ADDRESS;

TYPE
ProcessType;

Un type de processus. Pour chaque type de processus qu’il veut utiliser, l’usager doit déclarer
une variable de ce type, puis appeler la procédure Create.

ProcessInstance;

Lorsqu’on veut donner un nom spécifique (identificateur) à un processus, c’est-à-dire à un
“exemplaire” d’un type de processus, on utilise une variable de ce type comme identifi-
cateur. On associe cet identificateur au processus lorsqu’on prévoit son activation par
ScheduleNamed ou ScheduleNamedNext. A noter que le type ProcessInstance est différent
du type PROCESS prédéfini dans Modula-2.

ProcessState = (Executing, Delayed, Suspended);

Liste des différents états dans lesquels un processus peut se trouver : en exécution
(Executing), retardé pour une durée connue, avec un événement prévu pour lui redonner
le contrôle (Delayed), en attente qu’on lui redonne le contrôle, sans qu’il y ait d’événement
de prévu à cet effet (Suspended).

PROCEDURE Create
( VAR T : ProcessType;

P : PROC;
Size : CARDINAL;
Name : ARRAY OF CHAR

);

Associe la procédure P au type de processus T, et spécifie l’espace mémoire (Size) requis pour
chaque processus de ce type. Cette procédure est obligatoire pour chaque type de processus. Elle
ne crée pas de processus, mais seulement un type de processus. La taille de l’espace mémoire est
spécifiée en octets, et doit être suffisante pour l’exécution de la procédure P, pour ses variables
locales, de même que pour tout l’espace mémoire (pile d’exécution) requis par les procédures
qu’elle appelle directement ou indirectement. Name est une châıne de caractères identifiant ce
type de processus. Seuls les 32 premiers caractères sont retenus.

PROCEDURE Schedule
( T : ProcessType;

D : LONGREAL;
Par : ADDRESS

);

Crée un nouveau processus (un nouvel exemplaire) de type T, et prévoit (par un appel à
EVENT.Schedule) un événement qui donnera le contrôle à ce processus, dans D unités de temps.

Ce processus débutera donc son exécution à l’instant “SIM.Time() + D”. En attendant, il est
dans l’état Delayed. L’usager peut associer à ce processus un bloc de paramètres (un “at-
tribut”), sous forme d’un pointeur dont la valeur est donnée au paramètre Par. Il pourra
récupérer la valeur de ce pointeur lorsque ce processus sera en exécution, en utilisant la fonction
Attrib(). Lorsqu’on n’a pas besoin de passer une telle information, on passe NIL pour la valeur
de Par. La valeur de D doit être non négative, sinon une erreur est signalée et rien d’autre n’est
effectué.

PROCEDURE ScheduleNamed
( T : ProcessType;

D : LONGREAL;
Par : ADDRESS;
VAR P : ProcessInstance

);

Identique à Schedule, à la seule différence près que l’usager donne ici le nom spécifique P

au processus prévu. Ce processus devient donc baptisé et pourra être référencé par son nom
(c’est-à-dire par la valeur du paramètre P) ailleurs dans le programme.

PROCEDURE ScheduleNext
( T : ProcessType;

Par : ADDRESS
);

Tout comme Schedule, sauf que l’avis d’événement pour démarrer le processus est placé au
tout début de la liste d’événements. Ce processus doit démarrer à l’instant courant. S’il y a
d’autres événements prévus pour l’instant courant, l’avis de démarrage du processus sera placé
avant eux.

PROCEDURE ScheduleNamedNext
( T : ProcessType;

Par : ADDRESS;
VAR P : ProcessInstance

);

Identique à ScheduleNext, à la seule différence près que l’usager donne ici le nom spécifique P

au processus prévu, tout comme dans ScheduleNamed.

PROCEDURE Attrib () : ADDRESS;

Retourne le pointeur sur les attributs du processus en cours d’exécution. Ce pointeur est la
valeur du paramètre Par fourni lors de sa prévision. C’est à l’usager de s’assurer que ce résultat
est affecté à une variable du même type (c’est-à-dire un pointeur sur le même type de structure)
que la variable passée en paramètre lors de la prévision de ce processus.

PROCEDURE Delay
( D : LONGREAL
);

Indique que l’horloge de la simulation (la valeur de SIM.Time) doit avancer de D unités de
temps avant que la prochaine instruction du processus courant s’exécute. Cette procédure
arrête l’exécution du processus courant, le place dans l’état Delayed, et prévoit (par un avis
d’événement) son redémarrage dans D unités de temps.

PROCEDURE ResetDelay
( P : ProcessInstance;

D : LONGREAL
);

L’événement prévu qui doit redonner le contrôle au processus P (voir la procédure InstanceNotice,
plus bas) est déplacé dans le temps de façon à ce que son exécution soit maintenant prévue dans
D unités de temps à partir de l’instant courant. Si un tel événement n’existe pas, on en crée un
et le processus passe dans l’état Delayed, sauf s’il s’agit du processus en exécution, auquel cas
on imprime un message d’erreur. Par exemple, si le processus P a appelé la procédure Delay à
l’instant t = 10.0 avec D = 5.0, puis à t = 13.5, un autre processus appelle ResetDelay (P,

6.2), le processus P ne reprendra son exécution qu’à l’instant 19.7.

PROCEDURE CancelInstanceNotice
( P : ProcessInstance
);

Annule l’avis d’événement qui doit redonner le contrôle au processus P, et place ce dernier dans
l’état Suspended. Si un tel avis n’existe pas, c’est-à-dire si le processus n’est pas dans l’état
Delayed, imprime un message d’erreur.

PROCEDURE Suspend;

Le processus appelant cette procédure laisse le contrôle au processus principal, et passe dans
l’état Suspended.

PROCEDURE Resume
( P : ProcessInstance
);

Place un événement, au tout début de la liste d’événements (voir ScheduleNext), qui fera passer
le contôle au processus P. Ce dernier doit être dans l’état Delayed ou Suspended.

PROCEDURE Terminate;

Indique la fin normale de la vie d’un processus. Cette procédure doit être obligatoirement
appelée à la fin de la procédure associée à un type de processus, à moins que les processus de
ce type soient détruits de l’extérieur avant la fin de l’exécution de leur procédure.

PROCEDURE CurrentProcess
() : ProcessInstance;

Retourne le processus qui est en train de s’exécuter.

PROCEDURE InstanceType
( P : ProcessInstance
) : ProcessType;

Retourne le type du processus P.

PROCEDURE InstanceState
( P : ProcessInstance
) : ProcessState;

Retourne l’état dans lequel se trouve actuellement le processus P.

PROCEDURE InstanceAttrib
( P : ProcessInstance
) : ADDRESS;

Retourne le pointeur sur les attributs du processus P.

PROCEDURE InstanceNotice
( P : ProcessInstance
) : EventNotice;

Retourne l’avis d’événement prévu qui devrait redonner le contrôle au processus P (c’est-à-dire
qui marquera le début ou la poursuite de son exécution). Si un tel événement n’existe pas pour
ce processus, retourne NIL.

PROCEDURE InstanceDelay
( P : ProcessInstance
) : LONGREAL;

Retourne le temps qu’il reste avant l’exécution de l’avis d’événement retourné par la procédure
InstanceNotice (voir la procédure précédente). Si cet événement n’existe pas, imprime un
message d’erreur.

PROCEDURE InstanceSerialNumber
( P : ProcessInstance
) : LONGCARD;

Retourne le numéro de série du processus P.

PROCEDURE Kill
( VAR P : ProcessInstance
);

Détruit le processus P. Si P est le processus en cours d’exécution, cela équivaut à appeler
Terminate. A noter que dans le cas ou le processus possède des paramètres (valeur de Par lors
de la prevision), et que Par est un pointeur sur une structure de données quelconque, l’espace
occupé par cette structure n’est pas recupéré par le système, car celui-ci ne la connâıt pas. De
plus, si P se trouve dans une ou plusieurs listes lorsqu’on le détruit (à part la liste d’événements
et certaines listes gérées par le système), le contenu des listes en question n’est pas modifié. En
fait le système ne mémorise par les noms des listes dans lesquelles se trouve un processus, et
c’est à l’usager de s’en occuper.

PROCEDURE KillAll;

Détruit tous les processus existants (par des appels à Kill. A noter qu’il faut éviter d’appeler
cette procédure à l’intérieur d’un module qui risque de faire partie d’un système contenant
plusieurs modules, car si les autres modules contiennent des processus, ceux-ci seront aussi
détruits.

PROCEDURE DeleteType
( VAR T : ProcessType
);

Détruit le type de processus T et récupère l’espace mémoire utilisé pour sa description.

END PROCS.

GENEP

Ce module fournit des outils permettant de créer, démarrer ou stopper des mécanismes
automatiques pour générer une suite d’événements ou de processus d’un certain type. Cette
génération peut se faire selon un processus de Poisson, ou plus généralement selon un pro-
cessus de renouvellement quelconque.

Pour définir un tel mécanisme d’activation, l’usager doit déclarer une variable de type
ArrivalProcess, puis appeler Create. Les procédures StartArrivalsE, StartPoisson-
ArrivalsE, StartArrivalsP, StartPoissonArrivalsP et StopArrivals permettent de
démarrer ou de stopper ce mécanisme d’arrivées. Delete permet de le détruire et d’en
récupérer l’espace.

La fonction NbArriv permet de connâıtre le nombre d’arrivées qui se sont produites
depuis le dernier démarrage du mécanisme. Si on veut que des statistiques soient recueillies
automatiquement sur les durées entre les arrivées successives, il suffit d’appeler CollectStat
avant d’activer le mécanisme. La fonction StatDelay retournera alors le bloc statistique sur
ces durées. La fonction Report permet d’obtenir un rapport statistique.

DEFINITION MODULE GENEP;

FROM EVENT IMPORT EventType;
FROM PROCS IMPORT ProcessType;
FROM RAND IMPORT RngStream;
FROM STAT IMPORT Block;
FROM SYSTEM IMPORT ADDRESS;
TYPE

ArrivalProcess;

Un mécanisme de génération d’arrivées ou de processus. Pour chaque mécanisme qu’il veut
utiliser, l’usager doit déclarer une variable de ce type, puis appeler Create.

DelayProc = PROCEDURE () : LONGREAL;

Type utilisé en paramètre dans StartArrivals.

PROCEDURE Create
( VAR A : ArrivalProcess;

Name : ARRAY OF CHAR
);

Crée le mécanisme A de génération d’événements ou de processus, et lui associe le descripteur
Name.

PROCEDURE StartArrivalsE
( A : ArrivalProcess;

E : EventType;
Par : ADDRESS;
D : DelayProc;
Nmax : LONGCARD

);

Démarre le mécanisme d’arrivées A. Le temps jusqu’à la première arrivée, de même que les laps de
temps entre les arrivées successives, sont des variables aléatoires indépendantes, identiquement

distribuées, dont les valeurs sont générées en utilisant la fonction D, fournie par l’usager. Chaque
“arrivée” correspond à l’exécution d’un événement de type E. Tous ces événements auront
la valeur Par comme pointeur sur leur bloc de paramètres (c’est-à-dire leurs attributs). Si
Nmax > 0, le générateur s’arrêtera après avoir généré Nmax arrivées, à moins que l’on appelle
StopArrivals avant que ces Nmax arrivées ne se soient produites. Si Nmax = 0, c’est comme si
on avait Nmax égal à l’infini.

PROCEDURE StartPoissonArrivalsE
( g : RngStream;

A : ArrivalProcess;
E : EventType;
Par : ADDRESS;
Mean : LONGREAL;
Nmax : LONGCARD

);

Identique à StartArrivalsE, sauf que les durées entre les arrivées sont générées selon la loi
exponentielle de moyenne Mean, en utilisant le générateur numéro g. Ainsi, les arrivées se font
selon un processus de Poisson de taux 1/Mean.

PROCEDURE StartArrivalsP
( A : ArrivalProcess;

T : ProcessType;
Par : ADDRESS;
D : DelayProc;
Nmax : LONGCARD

);

Identique à StartArrivalsE, sauf que chaque “arrivée” active une nouvelle instance d’un pro-
cessus de type T.

PROCEDURE StartPoissonArrivalsP
( g : RngStream;

A : ArrivalProcess;
T : ProcessType;
Par : ADDRESS;
Mean : LONGREAL;
Nmax : LONGCARD

);

Identique à StartPoissonArrivalsE, sauf que chaque “arrivée” active une nouvelle instance
d’un processus de type T.

PROCEDURE StopArrivals
( A : ArrivalProcess
);

Arrête le mécanisme de génération A. Ce mécanisme devait avoir été démarré auparavant en
utilisant l’une des quatre procédures précédentes. On pourra le redémarrer en utilisant aussi
l’une de ces procédures.

PROCEDURE CollectStat
( A : ArrivalProcess
);

Déclenche un mécanisme automatique de recueil de statistiques pour le processus d’arrivées A.

Un bloc statistique de type Tally, qui échantillonne les durées entre les arrivées successives,
est créé et initialisé. C’est ce bloc qui est retourné par la fonction StatDelay. Il est réinitialisé
automatiquement lorsqu’on redémarre le processus d’arrivées. La durée entre deux arrivées est
recueillie au moment où la seconde arrivée se produit, et seulement les durées qui se terminent
après l’appel à CollectStat seront recueillies.

PROCEDURE StatDelay
( A : ArrivalProcess
) : Block;

Retourne le bloc statistique qui échantillonne les durées entre les arrivées successives pour le
processus d’arrivées A. Il s’agit d’un bloc de type Tally, qui n’existe que si on a déjà appelé
CollectStat pour A.

PROCEDURE NbArriv
( A : ArrivalProcess
) : LONGCARD;

Retourne le nombre d’arrivées qui se sont produites pour A depuis son dernier démarrage.

PROCEDURE Report
( A : ArrivalProcess
);

Fournit un rapport statistique sur A. Il faut d’abord avoir appelé CollectStat pour pouvoir
utiliser cette fonction.

PROCEDURE Delete
( VAR A : ArrivalProcess
);

Détruit le processus d’arrivées A et récupère l’espace qu’il utilisait.

END GENEP.

LIST

LIST est un module de gestion de listes doublement châınées. Le type List est prédéfini
dans le module. Un objet de ce type est constitué d’une tête de liste, contenant l’information
générale sur cette liste, et peut contenir zéro ou plusieurs objets.

Ce module gère chaque liste sans connâıtre le type des objets qui s’y trouvent. Les
“objets” insérés ou retirés des listes sont de type ADDRESS, et LIST ne manipule que ces
pointeurs. On peut construire des listes contenant à peu près n’importe quoi, comme par
exemple des processus, des blocs statistiques, des ressources, ou même d’autres listes. Il
peut y avoir plusieurs types d’objets dans une même liste, et un objet peut se trouver dans
plusieurs listes à la fois. En pratique, les types des objets d’une liste sont déclarés dans le
programme de l’usager et sont en général des pointeurs sur des structures de données de
types biens spécifiques. L’usager doit donc être prudent, et s’assurer que chaque fois qu’un
objet est retiré d’une liste, il est placé dans une variable du même type (c’est-à-dire le même
type de pointeur) que celle utilisée lors de son insertion dans la liste.

Pour chaque liste qu’il veut utiliser, l’usager doit déclarer une variable de type List, puis
créer cette liste. Une liste peut être ordonnée ou bien selon la façon dont les objets y sont
insérés, ou encore selon une procédure d’ordonnancement des objets fournie par l’usager lors
de sa création. Le choix de l’un ou l’autre de ces deux modes d’ordonnancement est fixé
une fois pour toutes lors de la création de la liste, et dépend de la procédure appelée pour
effectuer cette création: Create et CreateOrd respectivement. Dans le premier cas, on dit
(par un léger abus de langage) qu’on a une liste de type non ordonné, alors que dans le
second cas, on dit que la liste est de type ordonné.

Pour insérer des objets dans une liste, on utilise l’une des procédures Insert ou
InsertOrd, selon le type de la liste. La fonction Size permet de connâıtre le nombre
d’objets qui se trouvent dans une liste, et la fonction Belongs permet de savoir si un objet
particulier s’y trouve.

Pour chaque liste, le système mémorise la position du dernier objet référencé ou inséré.
Cet objet s’appelle l’objet courant. Les procédures Remove et View permettent de retirer un
objet d’une liste, ou de consulter (observer) un objet d’une liste sans le retirer de la liste. On
peut retirer ou consulter l’objet courant, son successeur ou son prédécesseur, de même que le
premier ou le dernier objet de la liste. Cela permet, par exemple, de parcourir une liste pour
y trouver un objet ayant une propriété particulière. On peut aussi, lorsqu’on connâıt le nom
d’un objet (c’est-à-dire la valeur du pointeur associé), utiliser la procédure RemoveObject

pour le retirer d’une liste.

D’autres procédures permettent d’unir deux listes, de trier une liste, ou de la diviser en
deux. Concatenate met bout-à-bout deux listes non ordonnées pour en faire une seule liste,
tandis que Merge fusionne en une seule liste deux listes ordonnées selon la même fonction
d’ordonnacement. Une liste ordonnée peut aussi être réordonnée, par la procédure Sort,
selon une nouvelle fonction fournie par l’usager. Enfin, il est aussi possible, par la procédure
Split, de diviser une liste en deux nouvelles listes au niveau de l’objet courant.

SIMOD recueille des statistiques automatiquement sur les listes, mais seulement si on
le lui demande (cela permet d’accélérer l’exécution dans les cas où ce n’est pas nécessaire).
Pour chaque liste pour laquelle on le désire, il suffit d’appeler la procédure CollectStat, et
le système lui associe deux blocs statistiques, mis-à-jour automatiquement. L’un de ces blocs
est de type Accumulate, et sert à mesurer l’évolution de la taille de la liste en fonction du
temps, tandis que l’autre est de type Tally, et échantillonne les durées de séjour des objets
dans la liste. Les fonctions StatSize et StatSojourn retournent ces blocs statistiques,
ce qui permet d’appeler les fonctions du module STAT pour examiner certaines statistiques
particulières. La procédure Report fournit par ailleurs un rapport statistique complet sur
une liste donnée, et InitStat permet de réinitialiser les deux blocs statistiques associés à
une liste.

La procédure Init vide une liste de tous ses objets et réinitialise sa tête (y compris
les blocs statistiques associés, s’il y a lieu), et Delete détruit une liste et récupère l’espace
occupé par la tête de liste. A noter cependant que ces procédures ne détruisent pas les objets
créés et gérés par l’usager, c’est-à-dire ceux qui correspondent aux pointeurs qui se trouvent
dans la liste créée ou détruite. L’usager peut donc réutiliser ces objets à d’autres fins. En
général, lorsqu’on effectue plusieurs répétitions d’une simulation, on doit vider la plupart
des listes après chaque répétition. On peut utiliser Init si les objets qui s’y trouvent n’ont
pas à être détruits parce qu’ils peuvent être réutilisés d’une répétition à l’autre. Lorsque les
objets qui s’y trouvent doivent être détruits pour en récupérer l’espace, l’usager doit alors
les retirer de la liste un à un pour les détruire.

DEFINITION MODULE LIST;

FROM STAT IMPORT Block;
FROM SYSTEM IMPORT ADDRESS;

TYPE
List;

Pour chaque liste qu’il veut utiliser, l’usager doit déclarer une variable de type List, puis
créer cette liste.

OutPosition = (First, Last, Previous, Next, Current);

Type de position où l’on peut observer ou retirer un objet dans une liste.

InPosition = [First..Next];

Type de position où l’on peut insérer un objet dans une liste non ordonnée.

OrdProc = PROCEDURE (ADDRESS, ADDRESS) : BOOLEAN;

Type de procedure utilisé comme paramètre dans les procédures CreateOrd et Sort.

CondProc = PROCEDURE (ADDRESS) : BOOLEAN;

Type de procedure utilisé comme paramètre dans la procédure FindFirstSuchThat.

ActionProc = PROCEDURE (ADDRESS);

Type de procedure utilisé comme paramètre dans la procédure ForEachInListDo.

PROCEDURE Create
( VAR L : List;

Name : ARRAY OF CHAR
);

Crée une liste L, de type non ordonné, initialement vide. On pourra y insérer des objets en
utilisant la procédure Insert. Le paramètre Name sert à identifier la liste lors des traces et des
rapports. On recommande que le nombre de caractères ne dépasse pas 32.

PROCEDURE CreateOrd
( VAR L : List;

Precede : OrdProc;
Name : ARRAY OF CHAR

);

Crée une liste L, de type ordonné, initialement vide, et dans laquelle on pourra insérer des objets
en utilisant la procédure InsertOrd. Cette liste sera maintenue en ordre automatiquement, selon
la fonction Precede fournie par l’usager. On peut passer directement l’identificateur de notre
fonction comme argument en autant qu’elle soit compatible avec la définition du type OrdProc.
Cette fonction doit retourner la valeur TRUE si et seulement si l’objet A doit précéder l’objet B
dans la liste. Pour le paramètre Name, voir Create.

PROCEDURE CollectStat
( L : List
);

Déclenche un recueil automatique de statistiques pour la liste L (déjà créée). Lorsqu’on appelle
cette procédure, deux blocs statistiques sont créés et initialisés. L’un (de type Accumulate) sert
à mesurer l’évolution de la taille de la liste en fonction du temps, et l’autre (de type Tally)
échantillonne les durées de séjour des objets dans la liste. Ce dernier ne recueille comme obser-
vations que les durées de séjour des objets insérés dans la liste après l’appel à CollectStat, et
qui quittent la liste pendant la période d’observation, c’est-à-dire entre la dernière initialisation
de ce bloc et l’instant courant. Les fonctions StatSize et StatSojourn retournent les blocs
statistiques en question. CollectStat appelle aussi automatiquement InitStat pour initialiser
ces deux blocs et faire une mise à jour pour le premier. Lorsqu’on appelle cette fonction, il est
recommandé de le faire tout de suite après la création de la liste.

PROCEDURE InitStat
( L : List
);

Réinitialise les deux blocs statistiques associés à la liste L, par des appels à STAT.Init, et fait
une mise à jour pour celui qui mesure l’évolution de la longueur de la liste. Ces deux blocs
doivent avoir été créés auparavant par un appel à CollectStat.

PROCEDURE Init
( L : List
);

Vide la liste L de tous ses objets et réinitialise ses blocs statistiques (s’il y a lieu). Ne détruit

pas, cependant, les objets qui s’y trouvent. Ces objets peuvent donc servir ultérieurement à
d’autres fins. Seul l’usager peut détruire, lorsqu’il le désire et afin de récupérer l’espace qu’ils
occupent, les objets qu’il a lui-même créés.

PROCEDURE Insert
( P : ADDRESS;

Where : InPosition;
L : List

);

Insère l’objet P dans la liste L. Selon la valeur du paramètre Where, l’objet est inséré au début
de la liste (First), à la fin (Last), avant l’objet courant (Previous) ou après l’objet courant
(Next). L’objet P devient l’objet courant. La liste L doit être de type non ordonné, c’est-à-dire
avoir été créée à l’aide de la procédure Create.

PROCEDURE InsertOrd
( P : ADDRESS;

L : List
);

Insère l’objet P dans la liste L. Cette liste doit être de type ordonné (créée par la procédure
CreateOrd), et est donc maintenue en ordre selon la fonction d’ordonnancement fournie par
l’usager.

PROCEDURE Belongs
( P : ADDRESS;

L : List
) : BOOLEAN;

Prend la valeur TRUE si l’objet P est dans la liste L, sinon prend la valeur FALSE. Lorsque la
fonction prend la valeur TRUE, P devient l’objet courant, sinon l’objet courant demeure inchangé.

PROCEDURE View
( VAR P : ADDRESS;

Which : OutPosition;
L : List

);

Permet d’observer un objet P qui se trouve dans la liste L. Selon la valeur du paramètre Which,
cet objet sera le premier objet de la liste (First), le dernier (Last), l’objet courant (Current),
celui qui précède l’objet courant (Previous) ou celui qui le suit (Next). La valeur NIL sera
retournée pour P si on tente d’observer l’objet précédent le premier de la liste, ou celui suivant
le dernier de la liste, ou encore si on tente d’observer un objet dans une liste vide. Si P est
différent de NIL, alors il devient l’objet courant.

PROCEDURE Remove
( VAR P : ADDRESS;

Where : OutPosition;
L : List

);

Retire un objet de la liste L et retourne cet objet dans la variable P. Selon la valeur du paramètre
Where, cet objet sera le premier objet de la liste (First), le dernier (Last), l’objet courant
(Current), celui qui précède l’objet courant (Previous) ou celui qui le suit (Next). La valeur

NIL sera retournée pour P si on tente de retirer l’objet précédent le premier de la liste, ou celui
suivant le dernier de la liste, ou encore si on tente de retirer un objet dans une liste vide. Si
l’objet retiré est l’objet courant et que la liste n’est pas vide, le nouvel objet courant sera le
suivant de celui retiré (ou le dernier de la liste, si l’objet enlevé était le dernier).

PROCEDURE RemoveObject
( P : ADDRESS;

L : List
);

Retire de la liste L un objet particulier P, spécifié par l’usager. Une erreur sera signalée si on
tente de retirer un objet d’une liste vide, ou un objet ne faisant pas partie de la liste. Cette
procédure regarde d’abord si l’objet courant est l’objet cherché, sinon elle parcourt la liste pour
le retrouver. Lorsqu’on retire l’objet courant, le nouvel objet courant est choisi de la même
façon que dans Remove.

PROCEDURE Size
( L : List
) : CARDINAL;

Retourne le nombre d’objets se trouvant dans la liste L.

PROCEDURE Concatenate
( VAR L1, L2 : List
);

Concatène la liste L2 à la fin de la liste L1, puis détruit L2. L1 et L2 doivent avoir été créées
par Create (c’est-à-dire être non ordonnées). L’objet courant de L1 n’est pas modifié. Si des
statistiques étaient recueillies sur L1, elles sont réinitialisées.

PROCEDURE Sort
( L : List;

Precede : OrdProc
);

Trie la liste L selon la fonction d’ordonnancement Precede fournie par l’usager. On peut passer
directement l’identificateur de notre fonction comme argument en autant qu’elle soit compatible
avec la définition du type OrdProc. Cette fonction doit retourner la valeur TRUE si l’objet A
doit précéder l’objet B dans la liste L. La liste L doit être de type ordonné (créée à l’aide de
la procédure CreateOrd). La fonction Precede utilisée lors du tri peut être différente de celle
spécifiée lors de la création. Dans ce cas, la liste est tout simplement retriée selon la nouvelle
fonction, qui sera utilisée par la suite.

PROCEDURE Merge
( VAR L1, L2 : List
);

Fusionne les listes ordonnées L1 et L2 en une seule liste ordonnée L1, puis détruit L2. L1 et L2
doivent être toutes les deux ordonnées selon la même procédure Precede fournie par l’usager.
La nouvelle liste s’appellera L1 et sera triée selon l’ordre des listes de départ. L’objet courant
de L1 n’est pas modifié. Si des statistiques étaient recueillies sur L1, elles sont réinitialisées.

PROCEDURE Split
( VAR L, L1, L2 : List;

Name1, Name2 : ARRAY OF CHAR
);

Divise la liste L en deux nouvelles listes L1 et L2, à partir de l’objet courant de L, puis détruit
la liste L. Ces deux nouvelles listes auront le même type d’ordonnancement que la liste L. La
liste L2 contiendra l’objet courant de L et tous ses suivants, dans le même ordre, et la liste L1

contiendra tous les objets précédant l’objet courant de L, dans le même ordre que dans L. Les
objets courants de L1 et L2 deviennent leurs premiers objets respectifs. Les paramètres Name1
et Name2 servent à donner des descripteurs aux nouvelles listes, pour les identifier lors des traces
à l’écran. Si des statistiques étaient recueillies sur L, elles sont réinitialisées et on en recueillera
sur L1 et L2.

PROCEDURE FindFirstSuchThat
( L : List;

VAR P : ADDRESS;
Cond : CondProc

);

Parcourt la liste L linéairement à partir de l’objet courant (inclusivement), et retourne le premier
objet P dans la liste tel que Cond (P) = TRUE. Si aucun des objets à partir de l’objet courant
jusqu’à la fin de la liste ne satisfait la condition, la valeur de P retournée est NIL. Si on veut
que toute la liste soit parcourue, à partir du début, on peut appeler View juste avant l’appel de
cette procédure. A noter que la procédure Cond ne doit pas modifier (par des effets de bord,
par exemple) la valeur du pointeur sur l’objet courant de la liste L, sinon les résultats obtenus
sont imprévisibles.

PROCEDURE ForEachInListDo
( L : List;

Action : ActionProc
);

Parcourt la liste L linéairement à partir de l’objet courant (inclusivement), et appelle la
procédure Action (P) pour chacun des objets P de la liste, à partir de l’objet courant jusqu’à
la fin de la liste. Si on veut que toute la liste soit parcourue, à partir du début, on peut appeler
View juste avant l’appel de cette procédure. A noter que la procédure Action ne doit pas mod-
ifier (par des effets de bord, par exemple) la valeur du pointeur sur l’objet courant de la liste L,
sinon les résultats obtenus sont imprévisibles.

PROCEDURE StatSize
( L : List
) : Block;

Retourne le bloc statistique qui mesure l’évolution de la longueur de la liste en fonction du temps.
Il s’agit d’un bloc de type Accumulate. Ce bloc n’existe que si on a déjà appelé CollectStat

pour cette liste. Sinon, cette fonction imprime un message d’erreur.

PROCEDURE StatSojourn
( L : List
) : Block;

Retourne le bloc statistique qui mesure les durées de séjour des objets dans la liste L. Il s’agit

d’un bloc de type Tally. Ce bloc n’existe que si on a déjà appelé CollectStat pour cette liste.
Sinon, cette fonction imprime un message d’erreur.

PROCEDURE Report
( L : List
);

Fournit un rapport statistique complet sur la liste L. On doit avoir auparavant appelé CollectStat
pour cette liste.

PROCEDURE Delete
( VAR L : List
);

Vide la liste L de tous ses objets et détruit cette liste. Tout comme Init, cette procédure ne
détruit pas les objets se trouvant dans cette liste.

END LIST.

RES

Le module RES fournit le type Resource, et un ensemble de procédures permettant la
synchronisation des processus qui utilisent des ressources à capacité limitée. Un objet de
type Resource représente une unité de service (une ressource), dont la capacité correspond
au nombre de serveurs. Ces serveurs sont tous identiques.

Un processus (voir le module PROCS) doit demander (Request) un certain nombre d’unités
de la ressource, et obtenir ces unités, avant de pouvoir les utiliser. Lorsqu’il en a terminé, il
libère ce nombre d’unités (Release). Il n’est pas nécessaire de demander ou de libérer d’un
seul coup toutes les unités de ressource que l’on veut utiliser. Un processus peut aussi détenir
simultanément des unités de plusieurs ressources différentes. Lorsqu’un processus demande
une ressource et que le nombre demandé d’unités est disponible, il obtient immédiatement
ces unités. Sinon, il est placé dans une file d’attente et doit attendre qu’un nombre suffisant
d’unités soit libéré avant d’obtenir la ressource en question et poursuivre son exécution.

Pour chaque ressource qu’il veut utiliser, l’usager doit déclarer une variable de type
Resource, puis créer cette ressource en appelant la procédure Create. Chaque ressource
possède une file d’attente unique, de capacité infinie, et dont la politique de service est
fixée lors de la création. Cette politique peut être Fifo (premier arrivé, premier servi), Lifo
(dernier arrivé, premier servi) ou Prior (avec priorités). Dans le cas d’une file d’attente avec
priorités, chaque processus qui demande la ressource doit le faire en utilisant la procédure
RequestWithPrior, en indiquant sa priorité. Les priorités sont des valeurs réelles, et les
priorités plus élevées passent avant les priorités plus faibles. La file d’attente est ordonnée
d’abord selon les priorités, et les processus ayant des priorités égales sont placés selon leur
ordre d’arrivée.

Lorsque des unités de ressource se libèrent, et que le premier processus dans la file de-
mande plus d’unités que ce qui est disponible, la ressource est allouée au prochain processus
dans la file dont la requête peut être satisfaite par les unités disponibles, si un tel processus
existe. De même, si un processus qui a besoin de plus d’unités que ce qui est disponible attend
dans la file, et si un second processus arrive et demande un nombre d’unités qui ne dépasse pas
ce qui est disponible, la requête du second processus sera satisfaite immédiatement (même
si ce dernier a une priorité inférieure). Si on veut qu’un processus s’empare des unités de
ressource au fur et à mesure que celles-ci deviennent disponibles, même si elles ne sont pas
en nombre suffisant pour ses besoins, on peut lui faire demander une à une les unités de
ressource désirées. Du point de vue des statistiques, toutefois, ce sera comme si toutes ces
requêtes venaient de processus différents.

La fonction Avail permet d’obtenir le nombre d’unités disponibles, pour une ressource
donnée, et la procédure ChangeCapacity permet de modifier la capacité d’une ressource.

A chaque ressource sont associées deux listes (ce sont des objets de type List, du mod-
ule LIST), soit la liste des processus en attente pour la ressource (la file d’attente) et la
liste des processus en service. Les fonctions WaitList et ServList retournent ces deux
listes, respectivement, et permettent donc d’observer ce qu’il y a dans ces listes, d’accéder

à leurs blocs statistiques, etc. A noter que la taille de ServList correspond au nombre
de processus qui occupent la ressource, mais pas nécessairement au nombre d’unités de la
ressource qui sont utilisées. Les objets manipulés par ces deux listes ne sont pas vraiment
des objets de type ProcessInstance du module PROCS. En fait, chaque fois qu’un processus
demande une ressource, on lui ouvre un dossier contenant un pointeur sur ce processus, en
plus d’information additionnelle. Le type UserDossier, prédéfini dans le module, représente
un pointeur sur un tel dossier. Les objets manipulés par WaitList et ServList sont des
variables de type UserDossier. Les fonctions User et NbUnits retournent respectivement le
nom du processus pour lequel un dossier donné a été créé, et le nombre demandé d’unités de
la ressource. Ces fonctions peuvent permettre, par exemple, de parcourir l’une des deux listes
pour retrouver un processus particulier, de savoir combien d’unités un tel processus demande
ou occupe, etc. Il est possible d’insérer ou de retirer directement de ces listes des objets de
type UserDossier, mais il n’est pas recommandé de le faire à moins que ce soit vraiment
nécessaire (par exemple si on veut simuler des préemptions). En particulier, lorsqu’on ef-
fectue directement de telles insertions ou de tels retraits, le bloc statistique retourné par
StatUtil (voir ci-bas) n’est pas mis à jour automatiquement.

Le progiciel peut recueillir des statistiques automatiquement sur les ressources lorsqu’on
le lui demande. Lorsqu’on appelle CollectStat pour une ressource donnée, SIMOD appelle
LIST.CollectStat pour les deux listes associées, et associe aussi à cette ressource trois
autres blocs statistiques, qui servent à mesurer respectivement l’évolution de la capacité et
l’utilisation de la ressource en fonction du temps, et les durées de séjour des processus sur
cette ressource. Les fonctions StatCapacity, StatUtil et StatSojourn, respectivement,
retournent ces blocs statistiques, ce qui permet par exemple d’appeler les fonctions du mod-
ule STAT pour observer certaines statistiques particulières. La procédure Report permet
par ailleurs d’obtenir un rapport statistique complet sur une ressource donnée, tandis que
InitStat permet de réinitialiser toutes les statistiques associées à une ressource donnée, y
compris celles des listes des processus en attente et en service. Lorsqu’on ne veut recueil-
lir des statistiques que sur les listes WaitList ou ServList, on peut appeler CollectStat
seulement pour la ou les liste(s) qui nous intéresse(nt). Cela peut permettre d’accélérer
l’exécution du programme.

Enfin, la procédure Init permet d’initialiser une ressource, de vider les deux listes qui
sont associées, et de réinitialiser ses statistiques, tandis que Delete permet de détruire la
ressource.

DEFINITION MODULE RES;

FROM STAT IMPORT Block;
FROM LIST IMPORT List;
FROM PROCS IMPORT ProcessInstance;

TYPE
Resource;

Pour chaque type de ressource qu’il veut utiliser, l’usager doit déclarer une variable de ce
type, puis appeler la procédure Create pour créer cette ressource.

ServicePolicy = (Fifo, Lifo, Prior);

Politique de service pour une ressource, détermine l’ordonnancement de la file d’attente.
Fifo: les processus sont placés dans la file selon leur ordre d’arrivée, les premiers arrivés
ayant priorité. Lifo: les processus sont placés dans l’ordre inverse de leur arrivée, les
derniers arrivés ayant priorité. Prior: Les processus doivent donner leur priorité lorsqu’ils
demandent la ressource (RequestWithPrior), et les priorités les plus élevées passent en
premier.

UserDossier;

Chaque fois qu’un processus demande pour une ressource, le module crée un dossier
contenant un pointeur sur ce processus, le nombre d’unités demandées, etc. Le type
UserDossier est un pointeur sur un tel dossier. Les listes retournées par les fonctions
WaitList et ServList sont des listes d’objets de ce type.

PROCEDURE Create
( VAR R : Resource;

Policy : ServicePolicy;
Capac : CARDINAL;
Name : ARRAY OF CHAR

);

Crée la ressource R. Le paramètre Policy indique la politique de service, et Capac indique la
capacité initiale de cette ressource. Cette capacité peut être modifiée par la suite en appelant
ChangeCapacity. Le paramètre Name sert à identifier la ressource dans les rapports statistiques.

PROCEDURE CollectStat
( R : Resource
);

Déclenche un recueil automatique de statistiques pour la ressource R. Cette ressource doit avoir
été créée auparavant. Cette procédure appelle les procédures LIST.CollectStat(WaitList(R))
et LIST.CollectStat(ServList(R)), afin que des statistiques soient recueillies automatique-
ment sur les listes des processus en attente et en service. Elle crée et initialise aussi trois
autres blocs statistiques associés à cette ressource, qui seront mis à jour automatiquement. Les
fonctions StatCapacity, StatUtil et StatSojourn retournent ces blocs statistiques, respec-
tivement. Le premier sert à mesurer l’évolution de la capacité de la ressource en fonction du
temps (il est mis à jour seulement lorsqu’on modifie la capacité de la ressource), le second sert
à mesurer l’utilisation de la ressource (le nombre d’unités occupées) en fonction du temps, et
le troisième recueille des statistiques sur les durées de séjour (attente + service) des processus;
il échantillonne une valeur pour chaque dossier qui quitte R après l’appel à CollectStat. Les
deux premiers blocs sont de type Accumulate et le troisième est de type Tally (voir le module
STAT).

Lorsqu’on appelle cette procédure, il est recommandé de le faire juste après la création de la
ressource. Note : Lorsqu’on tue un processus qui occupe une ressource, cela peut avoir des
conséquences imprévisibles sur les statistiques.

PROCEDURE InitStat
( R : Resource
);

Réinitialise tous les blocs statistiques associés à la ressource R, par des appels à LIST.InitStat

et STAT.Init, et effectue une mise-à-jour pour les blocs de type Accumulate. On doit avoir
appelé CollectStat auparavant pour cette ressource.

PROCEDURE Init
( R : Resource
);

Vide les deux listes associées à la ressource R, en récupérant l’espace occupé par les dossiers. Les
processus qui s’y trouvent demeurent suspendus. Réinitialise aussi tous les blocs statistiques
associés à cette ressource, s’il y a lieu.

PROCEDURE Capacity (R : Resource) : CARDINAL;

PROCEDURE ChangeCapacity
( R : Resource;

N : INTEGER
);

Modifie de N unités (augmente si N > 0, diminue si N < 0) la capacité de la ressource R, c’est-
à-dire le nombre d’unités de cette ressource. Si N > 0 et s’il y a des processus en attente dont
la requête peut maintenant être satisfaite, on leur accorde la ressource. Si N < 0, le nombre
d’unités disponibles (la capacité moins le nombre d’unités occupées) pour la ressource R doit
être supérieur ou égal à −N, sinon une erreur sera signalée. En particulier, la capacité d’une
ressource ne peut jamais devenir négative.

PROCEDURE SetCapacity
( R : Resource;

N : CARDINAL
);

Fixe à N unités la capacité de la ressource R. Similaire à ChangeCapacity.

PROCEDURE Avail
( R : Resource
) : CARDINAL;

Fonction qui retourne le nombre d’unités de la ressource R qui sont présentement disponibles,
c’est-à-dire la capacité de R moins le nombre d’unités occupées.

PROCEDURE Request
( N : CARDINAL;

R : Resource
);

Le processus qui appelle cette procédure demande N unités de la ressource R. Si le nombre
d’unités disponibles suffit pour répondre à sa requête, on lui accorde les unités demandées et il
les détient jusqu’à ce qu’il appelle Release pour les libérer. Ce processus est aussi inséré dans la
liste des processus en service pour cette ressource. Par contre, si ce nombre n’est pas suffisant,
le processus est placé dans la file d’attente pour R, dans l’état Suspended, et son exécution est
bloquée jusqu’à ce qu’il obtienne les unités de ressource demandées.

PROCEDURE RequestWithPrior
( N : CARDINAL;

R : Resource;
Pri : LONGREAL

);

Similaire à la procédure Request, sauf que le processus appelant demande la ressource avec
une priorité. On doit utiliser cette procédure dans le cas de ressources dont la politique de
service est de type Prior. Si le nombre d’unités disponibles est suffisant pour répondre à sa
requête, on lui accorde les unités demandées et il les détient jusqu’à ce qu’il appelle Release

pour les libérer. Le paramètre Pri indique la priorité, et les priorités plus élevés passent avant
les priorités plus faibles. Les processus ayant des valeurs de priorités égales sont classés selon
leur ordre d’arrivée.

PROCEDURE Release
( N : CARDINAL;

R : Resource
);

Le processus appelant libère N unités de la ressource R. Si ce processus occupait exactement N
unités de la ressource R, il est retiré de la liste des processus en service pour cette ressource.
S’il occupait moins de N unités, un message d’erreur est imprimé. S’il y a d’autres processus
en attente pour cette ressource, les unités libérées leur sont maintenant disponibles et peuvent
leur être allouées.

PROCEDURE WaitList
( R : Resource
) : List;

Retourne la liste des dossiers sur les processus en attente pour la ressource R. Les objets de
cette liste sont de type UserDossier.

PROCEDURE ServList
( R : Resource
) : List;

Retourne la liste des dossiers sur les processus en service pour la ressource R, c’est-à-dire les
processus qui occupent au moins une unité de cette ressource. Les objets de cette liste sont de
type UserDossier.

PROCEDURE User
( U : UserDossier
) : ProcessInstance;

Retourne le processus auquel correspond le dossier U.

PROCEDURE NbUnits
( U : UserDossier
) : CARDINAL;

Retourne le nombre d’unités de la ressource qui avait été demandé par le processus auquel
correspond le dossier U, lors de la création de ce dossier.

PROCEDURE StatCapacity
( R : Resource
) : Block;

Retourne le bloc statistique qui mesure l’évolution de la capacité de la ressource R en fonction
du temps. Il s’agit d’un bloc de type Accumulate. Ce bloc n’existe que si on a déjà appelé
CollectStat pour cette ressource. Sinon, cette fonction imprime un message d’erreur.

PROCEDURE StatUtil
( R : Resource
) : Block;

Retourne le bloc statistique qui mesure l’évolution du nombre d’unités de la ressource R qui
sont occupées, en fonction du temps. La moyenne associée à ce bloc, divisée par la capacité de
cette ressource, correspond au taux d’utilisation. Il s’agit d’un bloc de type Accumulate. Ce
bloc n’existe que si on a déjà appelé CollectStat pour cette ressource. Sinon, cette fonction
imprime un message d’erreur. A noter que le bloc retourné par StatSize (ServList (R))

s’intéresse au nombre de dossiers en service, ce qui n’est pas toujours la même chose que le
nombre d’unités de la ressource occupées, car un dossier peut occuper plusieurs unités à la fois.

PROCEDURE StatSojourn
( R : Resource
) : Block;

Retourne le bloc statistique qui mesure les durées de séjour des “UserDossier”’s pour la
ressource R. Il s’agit d’un bloc de type Tally, auquel une observation est ajoutée chaque fois
qu’un processus libère toutes les unités de ressource R dont il dispose. Ce bloc n’existe que si
on a déjà appelé CollectStat pour cette ressource. Sinon, cette fonction imprime un message
d’erreur.

PROCEDURE Report
( R : Resource
);

Fournit un rapport statistique complet sur la ressource R, dans le fichier de sortie courant.
On doit avoir auparavant appelé CollectStat pour cette ressource. Ce rapport contient des
statistiques sur les temps passés en attente et en service pour cette ressource, sur les durées de
séjour, sur l’évolution dans le temps de la capacité, du nombre d’unités occupées, du nombre de
processus utilisant la ressource, du nombre de processus en attente, et sur le taux d’utilisation
de cette ressource.

PROCEDURE Delete
( VAR R : Resource
);

Détruit la ressource R et récupère l’espace mémoire qu’elle occupait.

END RES.

BIN

Le module BIN fournit le type Bin, qui permet d’établir des relations de type produc-
teur/consommateur entre les processus. Un Bin correspond essentiellement à une bôıte (ou
une pile) de jetons indifférentiables, et une liste de processus qui attendent pour des jetons.
On peut ajouter des jetons à la pile, en appelant Give, ou demander des jetons, en appelant
Take. Dans le premier cas, on joue le rôle de producteur, tandis que dans le second cas, on
joue le rôle de consommateur. Un producteur peut ajouter des jetons à la pile sans les avoir
demandés auparavant, et un consommateur peut en demander sans jamais les remettre dans
la pile. En fait, un appel à Give peut être vu comme une création de jetons, et un appel
à Take comme une demande de destruction de jetons. Un appel à Give peut se faire de
n’importe où, mais un appel à Take ne peut se faire qu’à partir d’un processus. Le nombre
de jetons dans la pile doit toujours être supérieur ou égal à zéro, de sorte qu’un processus qui
demande un nombre de jetons plus grand que ce qui est disponible se voit bloqué et placé
dans une file d’attente.

L’utilisation du module BIN se fait de façon assez semblable à celle du module RES. Pour
chaque pile de jetons que l’on veut utiliser, on doit déclarer une variable de type Bin, puis
l’initialiser en appelant Create. Chaque pile possède une file d’attente unique, de capacité
infinie, et dont la politique de service est fixée lors de la création. Cette politique peut être
Fifo (premier arrivé, premier servi), Lifo (dernier arrivé, premier servi) ou Prior (avec
priorités). Dans le cas d’une file d’attente avec priorités, chaque processus qui demande des
jetons doit le faire en utilisant la procédure TakeWithPrior, en indiquant sa priorité. Les
priorités sont des valeurs réelles, et les priorités plus élevées passent avant les priorités plus
faibles. La file d’attente est ordonnée d’abord selon les priorités, et les processus ayant des
priorités égales sont placés selon leur ordre d’arrivée.

Lorsque des nouveaux jetons sont placés dans la pile, et que le premier processus dans
la file demande plus de jetons que ce qui est disponible, les jetons sont alloués au prochain
processus dans la file dont la requête peut être satisfaite par les jetons disponibles, et ne
sont alloués que si un tel processus existe. De même, si un processus qui a besoin de plus
de jetons que ce qui est disponible attend dans la file, et si un second processus demande un
nombre de jetons qui ne dépasse pas ce qui est disponible, la requête du second processus
sera satisfaite immédiatement (même si ce dernier a une priorité inférieure). Si on veut qu’un
processus s’empare des jetons au fur et à mesure que ceux-ci deviennent disponibles, même
s’ils ne sont pas en nombre suffisant pour ses besoins, on peut lui faire demander un à un
les jetons désirés. Toutefois, du point de vue des statistiques sur cette pile, ce sera comme
si toutes ces requêtes venaient de processus différents. La fonction Avail permet d’obtenir
le nombre de jetons disponibles, pour une pile donnée.

A chaque pile est associé un objet de type List, du module LIST, qui correspond à la liste
des processus qui attendent pour avoir des jetons de cette pile. La fonction WaitList retourne
cette liste, ce qui permet d’accéder à ses blocs statistiques, d’observer les dossiers sur les
processus en attente, etc. A noter que les objets manipulés par cette liste ne sont pas vraiment
des objets de type ProcessInstance du module PROCS. En fait, chaque fois qu’un processus



BIN 97

est placé dans la file d’attente pour une pile, on lui ouvre un dossier contenant un pointeur
sur ce processus, en plus d’informations additionnelles. Le type UserDossier, prédéfini dans
le module, représente un pointeur sur un tel dossier. Les objets manipulés par WaitList

sont de type UserDossier. Les fonctions User et NbTokens retournent respectivement le
nom du processus pour lequel un dossier donné a été créé, et le nombre de jetons demandés.
Il est possible d’insérer ou des retirer directement de la liste des objets de type UserDossier,
mais cela n’est pas vraiment recommandé. Enfin, la procédure Init permet d’initialiser une
pile, et Delete permet de la détruire.

DEFINITION MODULE BIN;

FROM LIST IMPORT List;
FROM PROCS IMPORT ProcessInstance;

TYPE
Bin;

Pour chaque pile de jetons qu’il veut utiliser, l’usager doit déclarer une variable de ce type,
puis appeler Create pour initialiser cette pile.

ServicePolicy = (Fifo, Lifo, Prior);

Types de politiques de service déterminant l’ordonnancement de la file d’attente pour une
pile de jetons. Fifo: les processus sont placés dans la file selon leur ordre d’arrivée, les
premiers arrivés ayant priorité. Lifo: les processus sont placés dans l’ordre inverse de leur
arrivée, les derniers arrivés ayant priorité. Prior: Les processus doivent donner leur priorité
lorsqu’ils demandent des jetons (TakeWithPrior), et les priorités les plus élevées passent en
premier.

UserDossier;

Chaque fois qu’un processus demande des jetons, le module crée un dossier contenant un
pointeur sur ce processus, le nombre de jetons demandés, etc. Le type UserDossier est un
pointeur sur un tel dossier. La liste retournée par la fonction WaitList est une liste d’objets
de ce type.

PROCEDURE Create
( VAR B : Bin;

Policy : ServicePolicy;
Name : ARRAY OF CHAR

);

Crée la pile de jetons B, initialement vide. Le paramètre Policy indique la politique de service.
Le paramètre Name sert à identifier cette pile dans les rapports ou les messages d’erreurs. Seuls
les 32 premiers caractères sont considérés.

PROCEDURE Init
( B : Bin
);

Vide la liste associée à B (file d’attente), en récupérant l’espace occupé par les dossiers. Les
processus qui s’y trouvent demeurent suspendus. Remet le nombre de jetons à zéro.

PROCEDURE Avail
( B : Bin
) : CARDINAL;

Retourne le nombre de jetons disponibles dans B.

PROCEDURE Take
( N : CARDINAL;

B : Bin
);

Cette procédure ne peut s’appeler qu’à partir d’un processus. Celui qui l’appelle demande N

jetons de la pile B. Si le nombre de jetons disponibles suffit pour répondre à sa requête, on lui
accorde les jetons demandés. Sinon, le processus est placé dans la file d’attente pour B (dans
l’état Suspended), et son exécution est bloquée jusqu’à ce qu’il obtienne les jetons demandés.

PROCEDURE TakeWithPrior
( N : CARDINAL;

B : Bin;
Priority : LONGREAL

);

Similaire à la procédure Take, sauf que le processus appelant demande N jetons avec une priorité.
On doit utiliser cette procédure dans le cas d’une pile dont la politique de service est de type
Prior. Le paramètre Priority indique la priorité, et les priorités plus élevées passent (dans la
file d’attente) avant les priorités plus faibles. Les processus ayant des valeurs de priorités égales
sont classés selon leur ordre d’arrivée.

PROCEDURE Give
( N : CARDINAL;

B : Bin
);

Place N jetons dans la pile B. S’il y a des processus en attente de jetons pour cette pile, ces nou-
veaux jetons maintenant disponibles peuvent leur être alloués et ces processus peuvent reprendre
leur exécution.

PROCEDURE WaitList
( B : Bin
) : List;

Retourne la liste des dossiers sur les processus en attente pour B. Les objets de cette liste sont
de type UserDossier.

PROCEDURE User
( U : UserDossier
) : ProcessInstance;

Retourne le processus auquel correspond le dossier U.

PROCEDURE NbTokens
( U : UserDossier
) : CARDINAL;

Retourne le nombre de jetons demandés par le processus en attente qui correspond au dossier
U.

PROCEDURE Delete
( VAR B : Bin
);

Détruit la pile B et récupère son espace mémoire.

END BIN.

COND

Ce module offre le type Condition, qui correspond essentiellement à un indicateur
booléen et à une liste de processus qui attendent que la condition devienne vraie. Un pro-
cessus qui appelle Wait pour une condition donnée peut poursuivre son exécution seulement
si la condition est vraie. Sinon, il se bloque et doit attendre que la condition soit mise à
“vrai” avant de reprendre son exécution. Les procédures SetTrue et SetFalse permettent
de mettre à vrai ou à faux l’état de la Condition, et CondState permet de connâıtre son
état.

Pour chaque condition que l’on veut utiliser, on doit déclarer une variable de type
Condition, puis créer cette condition en appelant la procédure Create. Chaque condi-
tion possède une liste de processus en attente, et la fonction WaitList retourne cette liste.
Les objets manipulés par cette liste sont de type UserDossier, tout comme dans RES et
dans BIN. La fonction User retourne le nom du processus correspondant à un dossier donné.
La procédure Init permet de réinitialiser une condition, et Delete permet de la détruire.

DEFINITION MODULE COND;

FROM LIST IMPORT List;
FROM PROCS IMPORT ProcessInstance;

TYPE

Condition;

Pour chaque condition que l’on veut utiliser, on doit déclarer une variable de ce type, puis
appeler Create pour l’initialiser.

UserDossier;

Chaque fois qu’un processus appelle Wait pour une condition qui n’est pas vraie, le module
crée un dossier contenant un pointeur sur ce processus, et place le pointeur sur ce dossier
(qui est de type UserDossier) dans la liste des processus en attente pour cette condition
(c’est-à-dire la liste retournée par WaitList).

PROCEDURE Create
( VAR C : Condition;

V : BOOLEAN;
Name : ARRAY OF CHAR

);

Crée la condition C et l’initialise à la valeur V. Le paramètre Name sert à identifier cette
Condition. Seuls les 32 premiers caractères sont considérés.

PROCEDURE Init
( C : Condition;

V : BOOLEAN
);

Vide la liste associée à C (file d’attente), en récupérant l’espace occupé par les dossiers. Les
processus qui s’y trouvent demeurent suspendus. Réinitialise la Condition à V.

PROCEDURE Wait
( C : Condition
);

Le processus qui appelle cette procédure ne peut pousuivre son exécution que si la condition C

est vraie. Sinon, le processus est placé dans la liste d’attente pour C, dans l’état Suspended, et
son exécution est bloquée jusqu’à ce que cette Condition soit mise à vrai.

PROCEDURE SetTrue
( C : Condition
);

Met la condition C à vrai. Tous les processus en attente peuvent reprendre leur exécution.

PROCEDURE SetFalse
( C : Condition
);

Met la condition C à faux.

PROCEDURE CondState
( C : Condition
) : BOOLEAN;

Retourne l’état de la condition C.

PROCEDURE WaitList
( C : Condition
) : List;

Retourne la liste des dossiers sur les processus en attente pour C. Les objets de cette liste sont
de type UserDossier.

PROCEDURE User
( U : UserDossier
) : ProcessInstance;

Retourne le processus auquel correspond le dossier U.

PROCEDURE Delete
( VAR C : Condition
);

Détruit la condition C et récupère l’espace mémoire qu’elle utilisait.

END COND.

MASLA

Le module MASLA permet de modéliser des relations de type mâıtre/esclave entre des
processus. Un objet de type MasterSlave, un type prédéfini dans MASLA, correspond à
un point de rencontre entre des processus “mâıtres” et des processus “esclaves”. Chaque
MasterSlave possède deux files d’attente: une pour les mâıtres et une pour les esclaves.
Un processus (agissant comme mâıtre) peut demander un esclave en appelant la fonction
Request, qui retourne le processus qui agira comme esclave. Si aucun esclave n’est disponible
à ce point de rencontre au moment de la demande, le processus demandeur est placé dans la
file d’attente des mâıtres, et ne peut poursuivre son exécution que lorsqu’un esclave devient
disponible. Un processus devient un esclave en appelant Wait pour un point de rencontre
donné. Si la file d’attente des mâıtres est vide, ce processus est placé dans la file d’attente
des esclaves et doit attendre qu’un mâıtre le demande. Sinon, le premier mâıtre de la file
obtient le nouvel esclave et peut maintenant poursuivre son exécution. Dans les deux cas,
l’esclave doit attendre que son mâıtre le libère en appelant Release avant de poursuivre son
exécution. A noter qu’un mâıtre peut détenir simultanément plusieurs esclaves, obtenus un
à un par le biais du même ou de différents point(s) de rencontre.

Pour chaque point de rencontre mâıtre/esclave qu’il veut utiliser, l’usager doit déclarer
une variable de type MasterSlave, puis créer ce point de rencontre en appelant la procédure
Create. Chaque point de rencontre possède deux files d’attente, de capacité infinie, et dont
la politique de service est fixée lors de la création. Cette politique peut être Fifo (premier
arrivé, premier servi), Lifo, (dernier arrivé, premier servi) ou Prior (avec priorités). Dans
le cas d’une file d’attente avec priorités, chaque processus qui demande à devenir un mâıtre
ou un esclave doit le faire en appelant WaitWithPrior ou RequestWithPrior, selon le cas,
en indiquant sa priorité. Les priorités sont des valeurs réelles, et les priorités plus élevées
passent avant les priorités plus faibles. Les files d’attente sont ordonnées d’abord selon les
priorités, et les processus ayant des priorités égales sont placés selon leur ordre d’arrivée.

A chaque point de rencontre sont associées deux listes, soit la file d’attente des mâıtres et
la file d’attente des esclaves. Les fonctions MastersList et SlavesList retournent ces
deux listes, respectivement, et permettent donc d’observer ce qu’il y a dans ces listes,
d’accéder à leurs blocs statistiques, etc. Les objets manipulés par ces deux listes sont de
type UserDossier. Ce sont des pointeurs sur des blocs d’information contenant, entre autres
choses, des pointeurs sur les processus qu’ils représentent. La fonction User retourne le nom
du processus pour lequel un dossier donné a été créé.

Enfin, la procédure Init permet d’initialiser un point de rencontre, en vidant les deux
listes qui lui sont associées, et Delete permet de le détruire.

DEFINITION MODULE MASLA;

FROM LIST IMPORT List;
FROM PROCS IMPORT ProcessInstance;

TYPE

MasterSlave;

Pour chaque point de rencontre qu’il veut utiliser, l’usager doit déclarer une variable de ce
type, puis appeler la procédure Create pour la créer.

ServicePolicy = (Fifo, Lifo, Prior);

Politique de service pour l’une ou l’autre des files d’attente d’un point de rencontre. Fifo: les
processus sont placés dans la file selon leur ordre d’arrivée, les premiers arrivés ayant priorité.
Lifo: les processus sont placés dans l’ordre inverse de leur arrivée, les derniers arrivés ayant
priorité. Prior: Les processus doivent donner leur priorité lorsqu’ils demandent à devenir
des mâıtres ou des esclaves (WaitWithPrior ou RequestWithPrior), et les priorités les plus
élevées passent en premier.

UserDossier;

Chaque fois qu’un processus est placé dans une file d’attente pour un point de rencontre,
le module crée un dossier contenant un pointeur sur ce processus, et quelques informations
additionnelles. Le type UserDossier est un pointeur sur un tel dossier. Les listes retournées
par les fonctions MastersList et SlavesList sont des listes d’objets de ce type.

PROCEDURE Create
( VAR M : MasterSlave;

MastersPolicy : ServicePolicy;
SlavesPolicy : ServicePolicy;
Name : ARRAY OF CHAR

);

Crée le point de rencontre M. Les paramètres MastersPolicy et SlavesPolicy indiquent les
politiques de service respectives des files d’attente des mâıtres et des esclaves. Le paramètre
Name sert à identifier ce point de rencontre dans d’éventuels rapports. On ne retient que les 32
premiers caractères.

PROCEDURE Init
( M : MasterSlave
);

Vide les deux listes associées à M, en récupérant l’espace occupé par les dossiers.

PROCEDURE Wait
( M : MasterSlave
);

Le processus qui appelle cette procédure devient un esclave au point de rencontre M. Si la file
d’attente des mâıtres est vide, ce processus est placé dans la file d’attente des esclaves, dans
l’état Suspended, et doit attendre qu’un mâıtre le demande. Sinon, le premier mâıtre de la file
obtient cet esclave et peut maintenant poursuivre son exécution. Dans les deux cas, cet esclave

doit attendre que son mâıtre le libère en appelant Release avant de poursuivre son exécution.
En attendant, il demeure dans l’état Suspended.

PROCEDURE WaitWithPrior
( M : MasterSlave;

Priority : LONGREAL
);

Similaire à Wait, sauf que le processus appellant possède une priorité. On doit utiliser cette
procédure lorsque la file d’attente des esclaves a une politique de service de type Prior. Le
paramètre Priority indique la priorité, et les priorités plus élevées passent avant les priorités
plus faibles. Les processus ayant des priorités égales sont placés selon leur ordre d’arrivée.

PROCEDURE Request
( M : MasterSlave
) : ProcessInstance;

Un processus appelant cette fonction devient un mâıtre au point de rencontre M. La fonction
retourne le processus qui agira comme esclave. Si aucun esclave n’est disponible à ce point de
rencontre, le processus (mâıtre) demandeur est placé dans la file d’attente des mâıtres, dans
l’état Suspended, et ne peut poursuivre son exécution que lorsqu’un esclave devient disponible.
Le processus esclave ne pourra poursuivre son exécution que lorsqu’on l’aura libéré en appelant
Release.

PROCEDURE RequestWithPrior
( M : MasterSlave;

Priority : LONGREAL
) : ProcessInstance;

Similaire à Request, sauf que le processus appellant possède une priorité. On doit utiliser cette
procédure lorsque la file d’attente des mâıtres a une politique de service de type Prior. Le
paramètre Priority indique la priorité, et le priorités plus élevées passent avant les priorités
plus faibles. Les processus ayant des priorités égales sont placés selon leur ordre d’arrivée.

PROCEDURE Release
( P : ProcessInstance
);

Permet au processus P, qui était un esclave au point de rencontre M, de poursuivre son exécution.
Equivaut à ResetDelay (0.0). Si on veut que le processus P ne reprenne son exécution que
dans D unités de temps, on peut plutôt appeler ResetDelay (D).

PROCEDURE MastersList
( M : MasterSlave
) : List;

Retourne la liste des dossiers sur les processus en attente dans la file des mâıtres au point de
rencontre M. Les objets de cette liste sont de type UserDossier.

PROCEDURE SlavesList
( M : MasterSlave
) : List;

Retourne la liste des dossiers sur les processus en attente dans la file des esclaves au point de
rencontre M. Les objets de cette liste sont de type UserDossier.

PROCEDURE User
( U : UserDossier
) : ProcessInstance;

Retourne le processus auquel correspond le dossier U.

PROCEDURE Delete
( VAR M : MasterSlave
);

Détruit le point de rencontre M et récupère l’espace mémoire qu’il utilisait.

END MASLA.

SIMOD

Les fichiers SIMOD.h et SIMOD.c existent uniquement pour permettre à un utilisateur de
programmer directement en C. Dans ce cas, il lui suffira d’inclure la directive

#include "SIMOD.h"

au début de son fichier, et d’ajouter

SIMOD_BEGIN ();

comme première instruction dans sa fonction main. Il pourra consulter les fichiers *.h

appropriés de $XDSINC pour voir les prototypes des fonctions externes C.

<* +M2EXTENSIONS *>
<* +NOHEADER *>
<* +CSTDLIB *>

DEFINITION MODULE ["C"] SIMOD;

PROCEDURE SIMOD_BEGIN();

END SIMOD.

